
Κεφάλαιο 1 Εισαγωγή στην Visual Basic Net 2005 Σελίδα 4 

Εισαγωγή στην Visual Basic Net 2005 

1.1 Το περιβάλλον της Visual Basic 2005 Express Edition  

 

Στο µάθηµα του Προγραµµατισµού διδάσκεται η γλώσσα Visual Basic και χρησιµοποιείται το 

περιβάλλον ανάπτυξης εφαρµογών Visual Basic 2005 Express Edition. Η έκδοση αυτή της Visual 

Basic διατίθεται ελεύθερα στο διαδίκτυο στην ηλεκτρονική διεύθυνση: 

http://msdn.microsoft.com/vstudio/express/vb/download/  

Εκκίνηση της Visual Basic 

 

Η εκκίνηση της Visual Basic γίνεται από το εικονίδιο «Microsoft Visual Basic 2005 Express Edition» 

της επιφάνειας εργασίας ή την επιλογή από το µενού «Έναρξη» 
«Έναρξη» 

 «Όλα τα προγράµµατα» 

     «Microsoft Visual Basic 2005 Express Edition» 

∆ηµιουργία νέας εφαρµογής  

 

Η δηµιουργία µια νέας εφαρµογής της Visual Basic, γίνεται µε την επιλογή «New Project» του µενού 

«File» (εικόνα 1.1). Εναλλακτικά µπορούµε να χρησιµοποιηθεί η συντόµευση Ctrl-N. 

 

 
 

Εικόνα 1.1 Η αρχική σελίδα της Visual Basic 

 
Στην συνέχεια εµφανίζεται ένα πλαίσιο διαλόγου (εικόνα 1.2) µε τον Τίτλο «New Project». Με την 

επιλογή «Console Application» στο πάνω παράθυρο του πλαισίου διαλόγου µε τον τίτλο «Template» 

δηµιουργείται µια νέα εφαρµογή τύπου «Console Application».  Ο τύπος της εφαρµογής αυτής είναι ο 

απλούστερος τύπος εφαρµογής και χρησιµοποιείται όταν δεν χρειάζεται το γραφικό περιβάλλον και οι 

φόρµες των Windows για την επικοινωνία (διεπαφή) µε τον χρήστη.  

Στο κάτω τµήµα του πλαισίου διαλόγου γίνεται η εισαγωγή του ονόµατος της εφαρµογής.  (π.χ. 

lab_1_askisi_1 για την πρώτη άσκηση του πρώτου εργαστηρίου) και στην συνέχεια µε το πάτηµα του 

πλήκτρο ΟΚ ολοκληρώνεται η δηµιουργία της εφαρµογής (εικόνα 1.2). 

 

  
Εικόνα 1.2 Καθορισµός ονόµατος και τύπου εφαρµογής 

 

 
 

Εικόνα 1.3 Η  νέα εφαρµογή 

Το πρώτο απλό πρόγραµµα 

 

Οι εφαρµογές της Visual Basic αποτελούνται από ένα ή περισσότερα αντικείµενα της Visual Basic 

(όπως modules, classes, windows forms, dialogs, user controls κλπ). Τα αντικείµενα αυτά µπορούν να 

επιλεγούν και να εµφανισθούν από το παράθυρο του “Solution Explorer” ο οποίος εµφανίζεται 

αρχικά στο δεξιό τµήµα της οθόνης. 

 
Το παράθυρο του “Solution Explorer” εµφανίζεται µε την επιλογή από το µενού “View” της επιλογής 

“Solution Explorer” (συντόµευση Ctrl-Alt-L). 

 

Με  την δηµιουργία µιας εφαρµογής τύπου «Console Application» δηµιουργούνται αυτόµατα δύο 

αντικείµενα: Το «My Project» το οποίο χρησιµοποιείται για την αποθήκευση των ρυθµίσεων της 

εφαρµογής και η λειτουργική µονάδα της Visual Basic «Module1» την οποία µπορούµε να 

χρησιµοποιήσουµε για να ορίσουµε µεταβλητές, διαδικασίες, και συναρτήσεις. 

 
Με την επιλογή “Properties window” του µενού “View” (συντόµευση F4) µπορούµε να εµφανίσουµε 

το παράθυρο ιδιοτήτων των αντικειµένων της Visual Basic (εικόνα 1.4). 

 

 
 

Εικόνα 1.4  Το παράθυρο ιδιοτήτων 

 

Με το παράθυρο ιδιοτήτων µπορούµε να εµφανίσουµε και να αλλάξουµε το όνοµα και τις ιδιότητες 

των αντικειµένων της Visual Basic. Για παράδειγµα εάν θέλουµε να µετονοµάσουµε το αντικειµένου 

“Module1” σε «FirstProgram», επιλέγουµε το αντικείµενο «Module1» και στην συνέχεια αλλάζουµε 

την τιµή της ιδιότητας «FileName» από «Module1.vb» σε «FirstProgram.vb»  

 

Παράδειγµα ενός απλού προγράµµατος Visual Basic το οποίο εκτυπώνει ένα µήνυµα είναι το 

παρακάτω: 

Module FirstProgram 
 
    Sub Main() 
        Console.WriteLine("Welcome") 
    End Sub 
 
End Module 
 

Παράδειγµα 1.1 
 

Το πρόγραµµα αυτό χρησιµοποιεί την εντολή WriteLine της κλάσης αντικειµένων Console για να 

εκτυπώσει ένα µήνυµα στην οθόνη. 

Αποθήκευση και εκτέλεση εφαρµογών 

 

Με την  επιλογή «Save All» του µενού «File γίνεται η αποθήκευση των εφαρµογών της Visual Basic 

(συντόµευση Ctrl-Shift-S).  

 

Στο παράθυρο διαλόγου µε το τίτλο «Save Project» (εικόνα 3) το οποίο εµφανίζεται γίνεται ο ορισµός 

του ονόµατος της εφαρµογής, της διαδροµής και του ονόµατος του καταλόγου όπου θα αποθηκευθεί 

η εφαρµογή. 

 

 
 

Εικόνα 1.5 ∆ηµιουργία νέας εφαρµογής 

 

Η εκτέλεση της εφαρµογής γίνεται µε την επιλογή «Start Debugging»  από το µενού «Debug» 

(συντόµευση F5). Η επιλογή «Start Debugging» χρησιµοποιείται για να ελέγξουµε την λειτουργία 

ενός προγράµµατος και να διορθώσουµε τα λάθη που εµφανίζονται κατά την εκτέλεση του 

προγράµµατος. Εναλλακτικά µπορεί να χρησιµοποιηθεί η επιλογή «Start Without Debugging»  από το 

µενού «Debug» (συντόµευση Ctrl-F5). 

 

Πριν από την εκτέλεση του προγράµµατος, γίνεται ο συντακτικός έλεγχος του. Εάν υπάρχουν 

συντακτικά λάθη, τότε εµφανίζεται ένα πλαίσιο διαλόγου το οποίο ενηµερώνει τον χρήστη ότι 

υπάρχουν λάθη. Με την επιλογή «Νο»  του πλαισίου διαλόγου αναιρείται η εκτέλεση του 

προγράµµατος ώστε να γίνει πρώτα η διόρθωση των συντακτικών λαθών τα οποία εµφανίζονται στο 

κάτω τµήµα του παραθύρου της Visual Basic µε τον τίτλο «Error List». 

 

Το παράθυρο «Error List» εµφανίζει λάθη, προειδοποιήσεις και µηνύµατα σχετικά µε την σύνταξη 

του προγράµµατος. Με την χρήση του αντίστοιχου εικονιδίου (errors, warnings and messages) γίνεται 

η επιλογή της εµφάνισης ή της απόκρυψης τους. 

 

Ο έλεγχος της σύνταξης ενός προγράµµατος ταυτόχρονα µε την σύνταξη του προγράµµατος. Τα 

συντακτικά λάθη υπογραµµίζονται και  εµφανίζονται ταυτόχρονα και στο παράθυρο  «Error List» 

(εικόνα 1.6).  

Εικόνα 1.6 Συντακτικά λάθη 

 

Με την εκτέλεση του προγράµµατος του παραδείγµατος 1.1 εµφανίζεται το παρακάτω παράθυρο 

τερµατικού (Console Application) µε το µήνυµα «Welcome». 

  

 
 

 Εικόνα 1.7 Το αποτέλεσµα της εκτέλεσης του παραδείγµατος 1.1 

 

Η εκτέλεση ενός προγράµµατος µπορεί να γίνει βήµα προς βήµα µε την επιλογή «Step Into» 

(συντόµευση F8) του µενού «Debug».  

 

Με την επιλογή «Toggle Breakpoint» του µενού «Debug» ορίζονται  ή να αναιρούνται σηµεία όπου η 

εκτέλεση του προγράµµατος διακόπτεται. Με την διακοπή του προγράµµατος στα συγκεκριµένα 

σηµεία µπορεί γίνει ο έλεγχος της ροή του προγράµµατος,  των τιµών των µεταβλητών και των 

ιδιοτήτων των διαφόρων αντικειµένων του προγράµµατος. Η εκτέλεση του προγράµµατος 

συνεχίζεται µε την επιλογή «Continue»  του µενού «Debug» (συντόµευση F5) ή µε την επιλογή «Step 

Into» (συντόµευση F8). 

Κατά την διάρκεια της διακοπής της εκτέλεσης του προγράµµατος δίνεται η δυνατότητα επιλογής και 

εµφάνισης των παραθύρων  “Locals”, “Call Stack”, “Watch” και “Immediate”.  

 

 
 

Εικόνα 1.8 ∆ιακοπή εκτέλεσης προγράµµατος  

 

Τα παράθυρα αυτά µπορούν να επιλεγούν µε τις επιλογές «Locals», «Call Stack», «Watch» και 

«Immediate» του µενού «Debug»-«Windows»  ή µε την επιλογή του τίτλου τους ο οποίος 

εµφανίζεται στο κάτω αριστερό τµήµα του παραθύρου του περιβάλλοντος της Visual Basic.  

 

Το παράθυρο «Locals» εµφανίζει τις τιµές των µεταβλητών και τα ιδιοτήτων των αντικειµένων του 

τµήµατος του προγράµµατος το οποίο εκτελείται.  

 

Το παράθυρο «Call Stack» εµφανίζει τις διαδοχικές κλήσεις µεθόδων και υποπρογραµµάτων οι 

οποίες έχουν γίνει.  

 

Το παράθυρο «Watch» εµφανίζει τις τιµές των µεταβλητών και τις ιδιότητες των αντικειµένων του 

υποπρογράµµατος το οποίο έχει επιλεγεί µε την επιλογή «Add Watch» του µενού «Debug».  

 

Το παράθυρο «Immediate» επιτρέπει την εισαγωγή και εκτέλεση εντολών. Για παράδειγµα (εικόνα 

1.8) µπορεί να γίνει η εισαγωγή της παρακάτω εντολής 

 
Console.WriteLine("Welcome again") 
 

και η εκτέλεση της µε το πάτηµα του πλήκτρου {Enter}. 

 

Κατά την διάρκεια της διακοπής ενός προγράµµατος είναι δυνατόν να γίνουν αλλαγές και διορθώσεις 

στο πρόγραµµα. Με τον τρόπο αυτό µπορεί να γίνει ταυτόχρονα ο έλεγχος και η διόρθωση των 

λογικών λαθών ενός προγράµµατος χωρίς να χρειάζεται η επανεκκίνηση του.  

Άλλες επιλογές του Μενού File 

 

Με  την εντολή “Close” του µενού “File” κλείνει η τρέχουσα εφαρµογή. 

 

Με  την εντολή “Open” του µενού “File” ανοίγει µια εφαρµογή που είχε δηµιουργηθεί και 

αποθηκευθεί παλαιότερα. 

Συνέχιση εντολής σε επόµενη γραµµή 

 

Κατά την σύνταξη ενός προγράµµατος η κάθε δήλωση ή εντολή καταλαµβάνει µια γραµµή κειµένου. 

Για να χρησιµοποιηθούν περισσότερες από µια γραµµές θα πρέπει να εισαχθεί ο χαρακτήρας '_ ' µε 

τον οποίο δηλώνεται ότι η σύνταξη της δήλωσης συνεχίζει στην επόµενη γραµµή.  

 
  Console.WriteLine( _ 
          "Εργαστήριο Προγραµµατισµού") 

Συνέχιση εντολής στην ίδια γραµµή 

 

Για την εισαγωγή περισσότερων από µία εντολών στην ίδια γραµµή κειµένου χρησιµοποιείται ο 

χαρακτήρας ': ' 

 
  Console.Write("Εργαστήριο "):Console.WriteLine("Προγραµµατισµού") 

Σχόλια 

 

Τα σχόλια στην Visual Basic εισάγονται µε το σύµβολο « ' » ή την λέξη REM. 

 
  Console.Write("Εργαστήριο ") ' Εκτύπωση µηνύµατος  

 

1.2 Λειτουργικές µονάδες (Modules) 

 

Τα βασικά στοιχεία µιας λειτουργικής µονάδα (module) της Visual Basic είναι: 

 

• Η δήλωση του ονόµατος της. (πχ Module FisrtProgram). Το όνοµα µιας λειτουργικής 

µονάδας πρέπει να είναι το ίδιο µε το όνοµα του αρχείου το οποίο χρησιµοποιείται για την 

αποθήκευση της.  

• ∆ηλώσεις µεταβλητών και σταθερών της µονάδος 

• ∆ηλώσεις διαδικασιών. Η δήλωση µιας διαδικασίας µε την δεσµευµένη λέξη «Sub» και το 

όνοµα της διαδικασίας. Το τέλος µιας διαδικασίας ορίζεται µε την δήλωση «End Sub”. 

• ∆ηλώσεις συναρτήσεων. Η δήλωση µιας συνάρτησης γίνεται µε την δεσµευµένη λέξη 

«Function», την δήλωση του ονόµατος της και τον τύπο της µεταβλητής την οποία 

επιστρέφει. Το τέλος µιας συνάρτησης ορίζεται µε την δήλωση «End Function». 

• Η δήλωση «End module» η οποία ορίζει το τέλος της λειτουργικής µονάδας. 

 
 
Module FirstProgram                     '∆ήλωση ονόµατος µονάδας 
 
    Sub Main()                          '∆ιαδικασία main 
 
        Console.WriteLine("Welcome")  
 
    End Sub                             'Τέλος διαδικασίας main 
 
End Module                              'Τέλος module 

Τα ονόµατα των προγραµµάτων, διαδικασιών, µεταβλητών και άλλων αντικειµένων αρχίζουν µε 

λατινικούς χαρακτήρες ή το χαρακτήρα «_», µπορούν να περιέχουν αριθµητικούς χαρακτήρες (αλλά 

όχι στην αρχή), και δεν πρέπει περιέχουν κενά ή άλλους ειδικούς χαρακτήρες.  Το ονόµατα επίσης 

δεν πρέπει να είναι δεσµευµένες λέξεις της Visual Basic. 

 

Ένα πρόγραµµα µπορεί να έχει πολλές λειτουργικές µονάδες. Η πρόσθεση νέων λειτουργικών 

µονάδων γίνεται µε την επιλογή «Add Module» του µενού «Project». Στο παρακάτω πλαίσιο 

διαλόγου επιλέγεται ο τύπο του αντικειµένου ο οποίος θα προστεθεί στην εφαρµογή (στην 

συγκεκριµένη περίπτωση ο προεπιλεγµένος τύπος είναι Module) και το όνοµα του. 

 

 
 

Εικόνα 1.9 ∆ιακοπή εκτέλεσης προγράµµατος  

 

Οι εφαρµογές τύπου Console Application έχουν µια αρχική λειτουργική µονάδα µε την οποία ξεκινά 

η εκτέλεση τους. Η αρχική λειτουργική µονάδα πρέπει να περιέχει µια διαδικασία µε το όνοµα 

«Main» η οποία είναι η πρώτη διαδικασία η οποία θα εκτελεσθεί. 

 

Ο ορισµός της αρχικής µονάδας µιας εφαρµογής γίνεται µε την επιλογή (διπλό πάτηµα του αριστερού 

πλήκτρου του ποντικιού) του αντικειµένου “MyProject” στο παράθυρο “Solution Explorer”. Στη 

συνέχεια εµφανίζεται το παράθυρο ρύθµισης των ιδιοτήτων της εφαρµογής (εικόνα 1.9) µε το οποίο 

µπορούµε να επιλέξουµε το αρχικό αντικείµενο εκκίνησης της εφαρµογής.  

Εικόνα 1.10 Επιλογή τύπου εφαρµογής και αρχικού αντικειµένου εκκίνησης 

1.3 Η κλάση αντικειµένων Console 

 

Σε εφαρµογές τύπου ConsoleApplication είναι διαθέσιµες οι εντολές και οι ιδιότητες της κλάσης 

αντικειµένων Console. Η κλάση Console αντιπροσωπεύει ένα παράθυρο τερµατικού στην οθόνη του 

Η/Υ. Με την χρήση των µεθόδων και των ιδιοτήτων της κλάσης αυτής µπορεί να γίνει η διαχείριση 

του παράθυρο τερµατικού της εφαρµογής. 

 

Η πρόσβαση στις µεθόδους και τις ιδιότητες αντικειµένων ή κλάσης αντικειµένων γίνεται µε την 

εισαγωγή πρώτα του ονόµατος του αντικειµένου ή της κλάσης ακολουθούµενο από µια τελεία και 

στην συνέχεια το όνοµα της µεθόδου ή της ιδιότητας. ( πχ Console.WriteLine(“Wellcome”) )   

 

Οι κυριότερες ιδιότητες και µεθόδοι της κλάσης Console είναι:  

 

• Η µέθοδος Clear()  η οποία καθαρίζει τα περιεχόµενα του παραθύρου. 

 

• Η ιδιότητα BackgroundColor η οποία ορίζει το χρώµα υπόβαθρου του παραθύρου και 

ιδιότητα ForegroundColor η οποία καθορίζει το χρώµα µε το οποίο εκτυπώνονται οι 

χαρακτήρες. 

 

• Οι ιδιότητες CursorLeft και CursorTop οι οποίες τοποθετούν τον δείκτη κειµένου (cursor) 

του παραθύρου σε συγκεκριµένη στήλη και γραµµή αντίστοιχα.  

 

• Η ιδιότητα Title µε την οποία ορίζεται ο τίτλος του παραθύρου. 

 

Το παρακάτω παράδειγµα αλλάζει τον τίτλο, το χρώµα υπόβαθρου και το χρώµα των χαρακτήρων, 

καθαρίζει το παράθυρο, εκτυπώνει το µήνυµα "Πατήστε οποιοδήποτε πλήκτρο για τερµατισµό" και 

περιµένει µε την χρήση της µεθόδου ReadLine να πατηθεί ένας οποιοσδήποτε χαρακτήρας από το 

πληκτρολόγιο.   

Sub main() 
 
     Console.Title = "Παράδειγµα 1.2" 
     Console.BackgroundColor = System.ConsoleColor.White 
     Console.ForegroundColor = System.ConsoleColor.Black 
     Console.Clear() 
     Console.CursorLeft = 30 
     Console.CursorTop = 10 
     Console.Write("Πατήστε οποιοδήποτε πλήκτρο για τερµατισµό") 
     Console.Read() 
 
End Sub 

 

Παράδειγµα 1.2 

  

Οι µέθοδοι Write και WriteLine 

 

Οι µέθοδοι Wriτe και WriteLine της κλάσης αντικειµένων Console χρησιµοποιούνται για την 

εκτύπωση µηνυµάτων στην οθόνη. Η διαφορά µεταξύ τους είναι ότι η WriteLine αλλάζει γραµµή 

κειµένου µετά την εκτύπωση του µηνύµατος. 

 

Οι µέθοδοι Write και WriteLine εκτυπώνουν σχεδόν όλους τους βασικούς τύπους δεδοµένων όπως 

ακέραιους, πραγµατικούς, λογικές µεταβλητές, χαρακτήρες και αλφαριθµητικές τιµές (Παράδειγµα 

1.3) 

 
 
Module Module1 
    Sub Main() 
        Console.WriteLine("Welcome") 
        Console.WriteLine(12.45) 
        Console.WriteLine(1) 
        Console.WriteLine(True) 
    End Sub 
End Module 
 

 

Παράδειγµα 1.3  

 

Οι µέθοδοι αυτοί µπορούν να χρησιµοποιηθούν και για την µορφοποίηση της εκτύπωσης των 

δεδοµένων. Εάν η σύνταξη της µεθόδου γίνει µε περισσότερες από µία παραµέτρους εισόδου, τότε η 

πρώτη παράµετρος ορίζει την µορφή µε την οποία θα εκτυπωθούν τα δεδοµένα τα οποία ακολουθούν. 

Τα δεδοµένα τα οποία θα εκτυπωθούν αριθµούνται ξεκινώντας από το αριθµό 0. Η παράµετρος 

µορφοποίησης είναι αλφαριθµητική τιµή η οποία χρησιµοποιεί αγκύλες για να δηλώσει την θέση στην 

οποία θα εκτυπωθούν τα δεδοµένα.  

 

Για παράδειγµα η εντολή 

 

   Console.WriteLine("{0}{1}{2}", 12, 1.32, 45) 

 

ορίζει ότι θα εκτυπωθούν οι αριθµοί 12, 1.32 και 25 στις θέσεις {0} , {1} και {2} της παραµέτρου 

µορφοποίησης. Το αποτέλεσµα της εκτύπωσης της µεθόδου είναι: 

 

121,3225 

 

Η παράµετρος µορφοποίησης µπορεί να εµπεριέχει και κείµενο εκτός των αγκυλών που ορίζουν τα 

δεδοµένα τα οποία θα εκτυπωθούν. Για παράδειγµα η εντολή 

 

   Console.WriteLine("∆1={0} ∆2={1} ∆3={2}", 12, 1.32, 45) 

εκτυπώνει το παρακάτω µήνυµα στην οθόνη: 

 

∆1=12 ∆2=1,32 ∆3=45 
 

Μπορούµε επίσης να καθορισθεί το εύρος εκτύπωσης των δεδοµένων και η στοίχισή τους (δεξιά εάν 

το εύρος είναι θετικός αριθµός και αριστερά εάν το εύρος είναι αρνητικός αριθµός). Για παράδειγµα η 

εντολή 

 

   Console.WriteLine("{0,10}{1,10}{2,10}", 12, 1.32, 45) 
 

εκτυπώνει το παρακάτω µήνυµα στην οθόνη: 

 

12        1,32      45 

  

Μπορεί επίσης να ορισθεί το πλήθος των ψηφίων ακεραίων αριθµών, το πλήθος των δεκαδικών 

ψηφίων και να εκτυπωθούν ακέραιοι σε δεκαεξαδική µορφή. Για παράδειγµα η εντολή 

 
   Console.WriteLine("{0:D4} {1:N5} {2 :X}", 12, 1.32, 25)  
 

εκτυπώνει το παρακάτω µήνυµα στην οθόνη: 
 

 0012 1,32000 19 

 

Τα σύµβολα µορφοποίησης D4, Ν5 και X καθορίζουν ότι: 

 

• {0:D4} ο ακέραιος αριθµός 12 θα εκτυπωθεί µε τουλάχιστον 4 ψηφία (αν τα ψηφία του 

αριθµού δεν επαρκούν θα προστεθούν µηδενικά ψηφία µπροστά από τον αριθµό) 

• {1:Ν5} ο πραγµατικός αριθµός 1.32 θα εκτυπωθεί µε τουλάχιστον 5 δεκαδικά ψηφία (αν τα 

δεκαδικά ψηφία του αριθµού δεν επαρκούν θα προστεθούν µηδενικά ψηφία στο τέλος του 

αριθµού) 

• {0:Χ} ο ακέραιος αριθµός 25 θα εκτυπωθεί σε δεκαεξαδική µορφή. 

 

Οι µέθοδος ReadLine 

 

Η µέθοδος ReadLine διαβάζει µία γραµµή κειµένου η οποία εισάγεται από το πληκτρολόγιο.  

 
Sub Main() 
 
     Dim x As Double 
    Console.Write ("∆ώστε ένα πραγµατικό αριθµό") 
     x = Console.ReadLine() 
     Console.WriteLine("Ο αριθµός είναι ο {0}", x) 
 
End Sub 

Παράδειγµα 1.4 

 

Το παραπάνω πρόγραµµα εκτυπώνει ένα µήνυµα το οποίο ζητά από τον χρήστη να εισάγει ένα 

πραγµατικό αριθµό, διαβάζει τον αριθµό από το πληκτρολόγιο και τον εκτυπώνει.  

 

Το κείµενο το οποίο εισάγεται από το πληκτρολόγιο, διαβάζεται µε την µέθοδο ReadLine, 

µετατρέπεται αυτόµατα σε πραγµατικό αριθµό και αποθηκεύεται στην µεταβλητή x.  

 

Περισσότερα παραδείγµατα χρήσης της ReadLine δίνονται στην επόµενη ενότητα. 

1.4 Ασκήσεις 

 

Άσκηση 1.1 

∆ηµιουργήστε εφαρµογή µε το όνοµα Lab_1_Ask_1, γράψτε το πρόγραµµα τους 

παραδείγµατος 1.1, αποθηκεύστε την εφαρµογή και εκτελέστε τη.  

 

Προσθέστε την εντολή Console.Readline() πριν από το τέλος του προγράµµατος. Τι 

παρατηρείτε; 

 

Άσκηση 1.2 

∆ηµιουργήστε εφαρµογή µε το όνοµα Lab_1_Ask_2, γράψτε το πρόγραµµα τους 

παραδείγµατος 1.2 και εκτελέστε το. Τροποποιήστε το τίτλο του παραθύρου και το χρώµα 

υπόβαθρου και εκτελέστε ξανά την εφαρµογή.  

 

Άσκηση 1.3 

∆ηµιουργήστε εφαρµογή µε το όνοµα Lab_1_Ask_3, η οποία ζητά από τον χρήστη να 

πληκτρολογήσει ένα πραγµατικό αριθµό και στην στοιχεία εκτυπώστε τον αριθµό που 

πληκτρολογήθηκε στην οθόνη. Κάντε χρήση του συµβόλου µορφοποίησης {0:Ν2} ώστε να 

εκτυπώσετε τον πραγµατικό αριθµό µε δύο δεκαδικά ψηφία. 

 

Άσκηση 1.4 

∆οκιµάστε να εκτελέσετε την εφαρµογή της προηγούµενης άσκησης βήµα προς βήµα. 

(πλήκτρο F8). 

Επιλέξτε το παράθυρο Locals (Επιλογή Locals του µενού Debug  - Windows). Τι εµφανίζεται 

στο παράθυρο αυτό; 

 Πηγαίνετε στην τελευταία εντολή του προγράµµατος και επιλέξτε «Toggle breakpoint» από 

το µενού «Debug» (πλήκτρο F9).  Τι παρατηρείτε να γίνεται κατά την εκτέλεση του 

προγράµµατος;  

Γράψτε στο παράθυρο «Immediate» την µέθοδο Console.WriteLine(“Welcome”) και 

εκτελέστε τη. 

2 Mεταβλητές και σταθερές 

2.1 ∆ηλώσεις Μεταβλητών 

 

Με τις δηλώσεις των µεταβλητών ορίζονται τα ονόµατα και οι τύποι των δεδοµένων µιας εφαρµογής 

και γίνεται η δέσµευση απαραίτητης µνήµης του υπολογιστή για την αποθήκευση και την ανάκτηση 

των δεδοµένων. 

 

Οι δηλώσεις µεταβλητών γίνονται µε την δεσµευµένη λέξη Dim, το όνοµα της µεταβλητής, την λέξη 

As και τον τύπο των δεδοµένων. Σε µία δήλωση µπορεί να ορίζονται περισσότερες από µία 

µεταβλητές όπως φαίνεται και στο παρακάτω παράδειγµα: 

 
Dim i, j as Integer 
Dim d as Double 

2.2 Τύποι ∆εδοµένων 

 

Στον πίνακα 1 εµφανίζονται οι βασικότεροι τύποι µεταβλητών που υποστηρίζονται από την Visual 

Basic. 

 
Τύπος Περιγραφή Απαιτούµενη 

Μνήµη 

Εύρος Τιµών 

Boolean Λογική Μεταβλητη 2 bytes True ή False 

Byte Ακέραιος 1 byte 0 έως 255 

Sbyte Ακέραιος 1 byte -128 έως 127 (signed byte) 

Char  Χαρακτήρας 2 bytes 0 έως 65,535 (unsigned character) 

Short Ακέραιος 2 bytes -32,768 έως 32,767 

Ushort Ακέραιος 2 bytes 0 έως 65,535 (unsigned short) 

Integer Ακέραιος 4 bytes -2,147,483,648 έως 2,147,483,647 

Uinteger Ακέραιος 4 bytes  0 έως 4,294,967,295 (unsigned integer) 

Long Ακέραιος 8 bytes  9,223,372,036,854,775,808  έως 

 9,223,372,036,854,775,807 

Ulong Ακέραιος 8 bytes 0 through 18,446,744,073,709,551,615 (unsigned long) 

Decimal Αριθµός 16 bytes 0 έως +/-79,228,162,514,264,337,593,543,950,335 για 

ακέραιους. 

 

0 +/-7.9228162514264337593543950335 

 µε δεκαδικά ψηφία 

Single Πραγµατικός απλής 

ακρίβειας 

4 bytes -3.4028235E+38 έως -1.401298E-45  

(αρνητικές τιµές) 

1.401298E-45 έως 3.4028235E+38  

(θετικές τιµές) 

Double Πραγµατικός 

διπλής ακρίβειας 

8 bytes Από -1.79769313486231570E+308 

έως  -4.94065645841246544E-324  

για αρνητικούς αριθµούς 

 

Από 4.94065645841246544E-324  

Έως 1.79769313486231570E+308 

 για θετικούς αριθµούς 

String  Αλφαριθµητική 

µεταβλητή 

Μεταβλητή από 0 έως 2.000.000.000 χαρακτήρες Unicode  

Date Ηµεροµηνία και 

ώρα 

8 bytes Από  1 Ιανουαρίου 0001 0:0:00 πµ  

έως 31 ∆εκεµβρίου 9999 11:59:59 µµ 

Object Αντικείµενο 4 bytes ∆είκτης σε αντικείµενα 

Structure ∆οµή Μεταβλητή Αναλόγως της δοµής που έχει ορισθεί 

 

Πίνακας 2.1 Βασικοί τύποι µεταβλητών 

Ανάθεση τιµής 

 

Η ανάθεση τιµής (αποθήκευση τιµής) σε µεταβλητής γίνεται µε τον τελεστή “=”. 
i=12 
d=1.34 

 

Η ανάθεση τιµής σε µεταβλητές τύπου Date γίνεται µε την χρήση των ειδικών χαρακτήρων # 
Dim startDate as Date 
startDate = #1/5/2005 16:55# 
 

Η ανάθεση τιµής σε µια µεταβλητή µπορεί να γίνει ταυτόχρονα µε τον ορισµό της. 
Dim startDate as Date = #1/5/2005 16:55# 
 

Η ανάθεση τιµής σε µια ακέραια µεταβλητή µπορεί να γίνει και µε την χρήση δεκαεξαδικών ή 

οκταδικών αριθµών. Πρίν από την τιµή θα πρέπει να προηγηθεί η κατάλληλη ακολουθία συµβόλων: 

&Η για δεκαεξαδικούς και &Ο για οκταδικούς. 
Dim num1 as Integer = &H64  '∆εκαεξαδικός αριθµός 
Dim num2 as Integer = &O144 'Οκταδικόςαριθµός 

 

Παράδειγµα προγράµµατος το οποίο ορίζει µεταβλητές, αποθηκεύει τιµές σε µεταβλητές και τις 

εκτυπώνει. 

 
 Module Module1 
    Sub Main() 
        '∆ηλώσεις µεταβλητών 
         Dim i As Integer 
        Dim l As Long 
        Dim s As Short 
        Dim b As Byte 
        Dim d As Double 
        Dim sn As Single 
        Dim msg1 As String 
        Dim c As Char 
        Dim flg As Boolean 
        Dim birthDay as Date  
        'Ανάθεση τιµών στις µεταβλητές 
        i = 12 
        l = 3456789 
        s = 45 
        b = 213 
        d = 1.0000000000045 
        sn = 1.32 
        msg1 = "Welcome" 
        c = "e" 
        flg = True 
        birthDay = #1/5/1994# 
        'Εκτύπωση των µεταβλητών 
        Console.WriteLine(i) 
        Console.WriteLine(l) 
        Console.WriteLine(s) 
        Console.WriteLine(b) 
        Console.WriteLine(d) 
        Console.WriteLine(sn) 
        Console.WriteLine(msg1) 
        Console.WriteLine(c) 
        Console.WriteLine(flg) 
        Console.WriteLine(birthDay) 
    End Sub 
End Module 

Παράδειγµα 2.1 

Το αποτέλεσµα της  εκτέλεσης του παραπάνω προγράµµατος είναι: 

 
12 
3456789 
45 
213 
1,0000000000045 
1,32 
Welcome 
e 
True 
5/1/1994 12:00:00 πµ 
 

Είσοδος δεδοµένων από το πληκτρολόγιο 

 

Η εισαγωγή δεδοµένων σε εφαρµογές τύπου ConsoleApplication γίνεται µε την µέθοδο ReadLine της 

κλάσης αντικειµένων Console.  

 

Στο παρακάτω παράδειγµα, ορίζεται µια ακέραια µεταβλητή i , και εισάγεται η τιµή της από το 

πληκτρολόγιο. 

 
Dim i as Integer 
i =Console.ReadLine() 
 

Συνήθως όταν σε µια εφαρµογή χρειάζεται να γίνει η εισαγωγή δεδοµένων, εκτυπώνονται πρώτα τα 

κατάλληλα µηνύµατα το οποία ενηµερώνουν τον χρήστη για τα στοιχεία τα οποία πρέπει να εισάγει.  

 

Για παράδειγµα, το παρακάτω πρόγραµµα εκτυπώνει το µήνυµα «∆ώστε ένα ακέραιο»,  χρησιµοποιεί 

την µέθοδο ReadLine για να διαβάσει έναν ακέραιο από το πληκτρολόγιο και στην συνέχεια τον 

εκτυπώνει στην οθόνη.  

 
Sub Main() 
 
     Dim i As Long 
    Console.Write ("∆ώστε ένα ακέραιο") 
     i = Console.ReadLine() 
     Console.WriteLine("Ο ακέραιος είναι ο {0}", i) 
 
End Sub 

Παράδειγµα 2.2 

 

Αν υποθέσουµε ότι πληκτρολογείτζαι ο αριθµός 45, τότε το αποτέλεσµα της εκτέλεσης του 

προγράµµατος είναι το παρακάτω: 

 
∆ώστε ένα ακέραιο 
45 
O ακέραιος εiναι ο 45 

2.3 Μετατροπή δεδοµένων 

Η µετατροπή τιµών µεταξύ διαφόρων τύπων δεδοµένων γίνεται είτε µε την χρήση του τελεστή 

ανάθεσης τιµής ή µε την χρήση ειδικών συναρτήσεων µετατροπής τύπων δεδοµένων.  

 
Module varConversions 
    Public Sub Main() 
        Dim i As Integer 
        Dim d As Double 
        d = 14.55 
        i = d 'Μετατροπή πραγµατικού αριθµού σε ακέραιο 
        Console.WriteLine(i) 
    End Sub 
End Module 

Παράδειγµα 2.3 

 

Στο παραπάνω παράδειγµα η µετατροπή της πραγµατικής µεταβλητής σε ακέραιο (µε 

στρογγυλοποίηση) γίνεται µε τον τελεστή ανάθεσης τιµής.  

 

Με την επιλογή «Option Strict On» η οποία µπορεί να δηλωθεί στην  αρχή µιας λειτουργικής µονάδας 

της Visual Basic απενεργοποιούνται οι αυτόµατες µετατροπές µεταξύ τύπων δεδοµένων (narrowing 

conversions) όπου τµήµα της µετατρεπόµενης τιµής (π.χ. από πραγµατικό σε ακέραιο) χάνεται ή το 

εύρος τιµών της µετατροπής είναι µικρότερο από το αρχικό εύρος (πχ. από πραγµατικό διπλής 

ακρίβειας σε πραγµατικό απλής ακρίβειας). 

 
Option Strict On 'Περιορισµός µετατροπών 
Module varConversions 
    Public Sub Main() 
        Dim i,j As Integer 
        Dim d As Double 
        Dim s As Single 
        d = 14.55 
        i = d 'Μη επιτρεπτή µετατροπή (Πραγµατικός σε ακέραιο) 
        s = d 'Μη επιτρεπτή µετατροπή (∆ιπλής ακρίβειας σε απλή ακρίβεια) 
        j = "45" 'Μη επιτρεπτή µετατροπή (Αλφαριθµητική τιµή σε ακέραιο) 
        Console.WriteLine("{0} {1:N2} {2}", i, s, j) 
    End Sub 
End Module 

Παράδειγµα 2.4 

 

Στο παραπάνω παράδειγµα οι µετατροπές που γίνονται µε τις εντολές i=d, s=d και i=”45” δεν 

επιτρέπονται και αντιµετωπίζονται ως συντακτικά λάθη του προγράµµατος. Εάν δεν υπήρχε η 

δήλωση «Option Strict On», οι µετατροπές αυτές θα ήταν αποδεκτές. 

 

Η επιλογή «Option Strict On» ή «Off» καθορίζεται καθολικά (για ολόκληρη την εφαρµογή) στις 

ρυθµίσεις Compile της εφαρµογής (σχήµα 2.1) . 

 

 Με την επιλογή «Option Explicit Off» δεν είναι απαραίτητη η δήλωση των  µεταβλητών. Το 

παρακάτω παράδειγµα χρησιµοποιεί την µεταβλητή s χωρίς να δηλώσει τον τύπο της µεταβλητής. 

 
Option Strict Off 
Option Explicit Off 
Module varConversions 
    Public Sub Main() 
        s = "45" 
        Console.WriteLine(s) 
    End Sub 
End Module   

Παράδειγµα 2.5 

Εικόνα 2.1 Επιλογή ρυθµίσεων Option Explicit και Option Strict 

Συναρτήσεις µετατροπής µεταβλητών 

 

Στον πίνακα 2.1 φαίνονται τα ονόµατα συναρτήσεων για την µετατροπή των κυριοτέρων τύπων 

µεταβλητών. 

 

Όνοµα συνάρτησης Τύπος µεταβλητής ο οποίος επιστρέφεται 

Cbool Boolean 

Cbyte Byte 

Cchar Char 

Cdate Date 

CDbl Double 

CDec Decimal 

CInt Integer 

CLng Long 

CObj Object 

CSByte SByte 

CShort Short 

CSng Single 

CStr String 

CUInt UInteger 

CULng ULong 

CUShort UShort 

 

Πίνακας 2.2 Συναρτήσεις µετατροπής δεδοµένων 

 

Στο παρακάτω παράδειγµα γίνεται η µετατροπή τιµών µε την χρήση των συναρτήσεων Cint και 

CSng. 

Option Strict On 'Περιορισµός µετατροπών 
Module varConversions 
    Public Sub Main() 
        Dim i, j As Integer 
        Dim d As Double 
        Dim s As Single 
        d = 14.55 
        i = CInt(d) '(Πραγµατικός σε ακέραιο) 
        s = CSng(d) '(∆ιπλής ακρίβειας σε απλή ακρίβεια) 
        j = CInt("45") '(Αλφαριθµητική τιµή σε ακέραιο) 
        Console.WriteLine("{0} {1:N2} {2}", i, s, j) 
    End Sub 
End Module 

Παράδειγµα 2.6 

Η συνάρτηση Ctype 

 

Εναλλακτικά αντί των συναρτήσεων του πίνακα 2.2 µπορεί να χρησιµοποιηθεί η συνάρτηση Ctype η 

οποία δέχεται ως πρώτη παράµετρο τον τιµή η οποία θα µετατραπεί και ως δεύτερη παράµετρο τον 

τύπο δεδοµένων όπου θα γίνει η µετατροπή 

 
    Public Sub Main() 
        Dim i As Integer 
        Dim d As Date 
        Dim s As Single 
        i = CType("456", Integer) 
        s = CType("4,434", Single) 
        d = CType("1/1/2005", Date) 
        Console.WriteLine("{0} {1:N2} {2}", i, s, d) 
    End Sub 

Παράδειγµα 2.7 

H µέθοδος Parse 

 

Όλοι οι βασικοί τύποι δεδοµένων (εκτός του αλφαριθµητικού τύπου String) διαθέτουν την µέθοδο 

parse η οποία µετατρέπει µια αλφαριθµητική τιµή στον αντίστοιχο τύπο δεδοµένων. Στο παρακάτω 

παράδειγµα φαίνεται ή σύνταξη της µεθόδου Parse. 

 
    Public Sub Main() 
        Dim i As Integer 
        Dim d As Date 
        Dim s As Single 
        i = Integer.Parse("456") 
        s = Single.Parse("4,434") 
        d = Date.Parse("1/1/2005") 
        Console.WriteLine("{0} {1:N2} {2}", i, s, d) 
    End Sub 

Παράδειγµα 2.8 

H µέθοδος ToString 

 

Όλοι οι βασικοί τύποι δεδοµένων (εκτός του αλφαριθµητικού τύπου String) διαθέτουν επίσης την 

µέθοδο ToString η οποία µετατρέπει την τιµή της µεταβλητής σε αλφαριθµητική τιµή (αντίστροφος 

της µεθόδου Parse). Η µέθοδος αυτή µπορεί προαιρετικά να πάρει ως παράµετρο όρισµα 

µορφοποίησης (παρόµοιο της µεθόδου WriteLine) 

 

 

 



Κεφάλαιο 2 Μεταβλητές και σταθερές Σελίδα  22 

Σηµειώσεις Προγραµµατισµού  Γρηγόρης Τζιάλλας Τµήµα Ηλεκτρονικής Τ.Ε.Ι. Λαµίας  

    Public Sub Main() 
        Dim i As Integer 
        Dim d As Date 
        Dim s As Single 
        i = 456 
        s = 4.434 
        d = #25/2/2005# 
        'Μετατροπή ακέραιου χωρίς παράµετρο µορφοποίησης 
        Console.WriteLine("i={0,30}", i.ToString()) 
        'Μετατροπή ακέραιου µε εύρος 9 ψηφίων 
        Console.WriteLine("i={0,30}", i.ToString("D9")) 
        'Μετατροπή πραγµατικού µε 4 δεκαδικά ψηφία 
        Console.WriteLine("s={0,30}", s.ToString("N4")) 
        'Μετατροπή ηµεροµηνίας σε πλήρως ανεπτυγµένη µορφή 
        Console.WriteLine("d={0,30}", d.ToString("D")) 
    End Sub 

Παράδειγµα 2.9 

 

Για την µετατροπή ηµεροµηνιών µπορούν να χρησιµοποιηθούν προκαθορισµένες ή προσαρµοσµένες 

µορφές ηµεροµηνίας. Το παρακάτω πρόγραµµα εκτυπώνει την τρέχουσα ηµέρα και ώρα στις 

κυριότερες προκαθορισµένες µορφές καθώς και σε 4 προσαρµοσµένες µορφές. Οι προσαρµοσµένες 

µορφές ηµεροµηνιών ορίζονται µε την χρήση των συµβόλων: 

• d, dd, ddd, dddd για την ηµέρα (πχ 2, 02, Κυρ, Κυριακή) 

• M, MM, MMM, MMMMM για τους µήνες (πχ 2, 02, Φεβ, Φεβρουάριος) 

• yy, yyyy για το έτος (πχ 07, 2007) 

• h, hh για τις ώρες (πχ 6, 06)  

• m, mm για τα λεπτά (πχ 5, 05) 

• s, ss για τα δευτερόλεπτα (πχ 6, 06) 

 
Public Sub Main() 
   Dim dd As Date 
   dd = Date.Now 'Επιστρέφει την τρέχουσα ηµεροµηνία 
   Console.WriteLine("Προκαθορισµένες µορφές ηµεροµηνίας") 
   Console.WriteLine("********************************************************************") 
   Console.WriteLine("Long Date Pattern           D = {0}", dd.ToString("D")) 
   Console.WriteLine("Short Date Pattern (short)  d = {0}", dd.ToString("d")) 
   Console.WriteLine("Full Date Pattern           F = {0}", dd.ToString("F")) 
   Console.WriteLine("Full Date Pattern (short)   f = {0}", dd.ToString("f")) 
   Console.WriteLine("General Date Pattern        G = {0}", dd.ToString("G")) 
   Console.WriteLine("General Date Pattern (short)g = {0}", dd.ToString("g")) 
   Console.WriteLine("Long Time Pattern           T = {0}", dd.ToString("T")) 
   Console.WriteLine("Short Time Pattern          t = {0}", dd.ToString("t")) 
   Console.WriteLine("Προσαρµοσµένες µορφές ηµεροµηνίας") 
   Console.WriteLine("********************************************************************") 
   Console.WriteLine("            dd/MM/yy hh:mm:ss = {0}",  dd.ToString("dd/MM/yy hh:mm:ss")) 
   Console.WriteLine("             dd/MM/yyyy hh:mm = {0}", dd.ToString("dd/MM/yyyy hh:mm")) 
   Console.WriteLine("         dd-MMM-yyyy hh:mm:ss = {0}", dd.ToString("dd-MMM-yyyy hh:mm:ss")) 
   Console.WriteLine("  dddd, dd MMMMM yyyy h:mm:ss = {0}", _ 
                               dd.ToString("dddd, dd MMMMM yyyy h:mm:ss")) 
End Sub 

Παράδειγµα 2.10 

 

Το αποτέλεσµα της εκτέλεσης του προγράµµατος είναι: 

 
Προκαθορισµένες µορφές ηµεροµηνίας 
*********************************************************************** 
Long Date Pattern           D = Κυριακή, 25 Φεβρουαρίου 2007 
Short Date Pattern (short)  d = 25/2/2007 
Full Date Pattern           F = Κυριακή, 25 Φεβρουαρίου 2007 1:47:17 µµ 
Full Date Pattern (short)   f = Κυριακή, 25 Φεβρουαρίου 2007 1:47 µµ 
General Date Pattern        G = 25/2/2007 1:47:17 µµ 
General Date Pattern (short)g = 25/2/2007 1:47 µµ 
Long Time Pattern           T = 1:47:17 µµ 
Short Time Pattern          t = 1:47 µµ 
Προσαρµοσµένες µορφές ηµεροµηνίας 
*********************************************************************** 
            dd/MM/yy hh:mm:ss = 25/02/07 01:47:17 
             dd/MM/yyyy hh:mm = 25/02/2007 01:47 
         dd-MMM-yyyy hh:mm:ss = 25-Φεβ-2007 01:47:17 
  dddd, dd MMMMM yyyy h:mm:ss = Κυριακή, 25 Φεβρουαρίου 2007 1:47:17 

2.4 Σταθερές 

 

Οι δηλώσεις σταθερών σε ένα πρόγραµµα χρησιµοποιούνται για να ορίσουν αριθµούς ή άλλες τιµές 

δεδοµένων οι οποίες παραµένουν σταθερές και αµετάβλητες κατά την διάρκεια εκτέλεσης ενός 

προγράµµατος. 

 

Η δήλωση σταθερών γίνεται µε την δεσµευµένη λέξη Const, το όνοµα της σταθερής, τον τελεστή 

ισότητας και την τιµή της σταθερής   

 
Const DaysInYear = 365 
Const WorkDays = 250 
conPi = 3.14159265358979 
Const conMaxPlanets As Integer = 9 
Const conReleaseDate = #1/1/1995# 
 

Προαιρετικά µπορούµε να δηλωθεί και ο τύπος της σταθερής 

 
Const conΜsg as String = "Hello" 
 

2.5 Ασκήσεις κεφαλαίου 2 

 

Άσκηση 2.1 

∆ηµιουργήστε εφαρµογή µε όνοµα Lab_2_Ask_1 η οποία ζητά από τον χρήστη να 

πληκτρολογήσει το όνοµα, το επώνυµο, την ηλικία , το ύψος και την ηµεροµηνία γέννησης 

του. Στην συνέχεια η εφαρµογή εκτυπώνει τα στοιχεία του χρήστη στην οθόνη. Το ύψος 

εκτυπώνεται µε δύο δεκαδικά στοιχεία. Η ηµεροµηνία εκτυπώνεται µε την χρήση των 

προκαθορισµένων µορφών ηµεροµηνίας (όπως στο παράδειγµα 2.10) 

 

Ορίστε σηµείο διακοπής («Toggle breakpoint» του µενού Debug) στην πρώτη εντολή 

εκτύπωσης των στοιχείων. Χρησιµοποιήστε το παράθυρο Locals για να δείτε τις τιµές των 

µεταβλητών πριν την εκτύπωση και το παράθυρο Immediate για να αλλάξετε τις τιµές των 

δεδοµένων. 

 

Άσκηση 2.2 

∆ηµιουργήστε εφαρµογή µε όνοµα Lab_2_Ask_2 η οποία ζητά από τον χρήστη να εισάγει 6 

πραγµατικούς αριθµούς και στην συνέχεια τους εκτυπώνει σε 3 γραµµές (δύο αριθµοί σε κάθε 

γραµµή). Οι αριθµοί εκτυπώνονται µε ακρίβεια  4 δεκαδικών στοιχείων και εύρος εκτύπωσης 

10 χαρακτήρες.  

 

Χρησιµοποιήστε ως παράµετρο µορφοποίησης:  {0,10:N4}{1,10:N4} 

 

Άσκηση 2.3 
∆ηµιουργήστε εφαρµογή µε όνοµα Lab_2_Ask_3 η οποία ζητά από τον χρήστη να εισάγει µια 

ηµεροµηνία και στην συνέχεια εκτυπώνει την ηµέρα της εβδοµάδος (Κυριακή, ∆ευτέρα κλπ) η 

οποία αντιστοιχεί στην ηµεροµηνία αυτή. 

 

Άσκηση 2.4 

∆ηµιουργήστε εφαρµογή µε όνοµα Lab_2_Ask_4 η οποία ζητά από τον χρήστη να εισάγει 

δεδοµένα, κάνει χρήση των συναρτήσεων µετατροπής δεδοµένων του πίνακα 2.2, της 

συνάρτησης ctype, και των µεθόδων parse και ToString και εκτυπώνει τα αποτελέσµατα των 

µετατροπών. 

3  Τελεστές – Μαθηµατικές συναρτήσεις 

H Visual Basic υποστηρίζει τους παρακάτω τύπους τελεστών: 

• Αριθµητικοί τελεστές 

• Αλφαριθµητικοί τελεστές 

• Τελεστές σύγκρισης 

• Λογικοί τελεστές 

3.1 Αριθµητικοί τελεστές 

 

 

Τελεστής Περιγραφή Παράδειγµα Αποτέλεσµα 

^ ∆ύναµη 3 ^ 2 9 

- Αρνητικό πρόσηµο -5  5 

* Πολλαπλασιασµός 2 * 3 6 

/ ∆ιαίρεση 5 / 2 2.5 

\ Ακέραια διαίρεση 7 \ 3 2 

Mod Υπόλοιπο διαίρεσης 7 Mod 3 1 

+ Πρόσθεση 2 + 4 6 

- Αφαίρεση 4 – 2 2 

Πίνακας 3.1 Αριθµητικοί τελεστές 

 

Παράδειγµα χρήσης αριθµητικών τελεστών µε πραγµατικούς αριθµούς για τον υπολογισµό της 

παράστασης x 
3
 + 2 x y + 4  y + 24: 

 
Public Sub main() 
    Dim x, y, z As Double 
    Console.Write("∆ώσε ένα πραγµατικό αριθµό x ") 
    x = Console.ReadLine() 
    Console.Write ("∆ώσε ένα δεύτερο πραγµατικό αριθµό y ") 
    y = Console.ReadLine() 
    z = x ^ 3 + 2 * x * y + 4 * y + 24 
    Console.WriteLine(" x ^ 3 + 2 * x * y + 4 * y + 24 = {0:N4}", z) 
End Sub 

Παράδειγµα 3.1 

 

Το αποτέλεσµα του παραπάνω προγράµµατος εάν εισαχθούν οι αριθµοί 1.45 και 3,321 είναι: 

 
∆ώσε ένα πραγµατικό αριθµό x 1,45 
∆ώσε ένα δεύτερο πραγµατικό αριθµό 3,321 
x ^ 3 + 2 * x * y + 4 * y + 24 = 49,9635 
 

Παράδειγµα χρήσης αριθµητικών τελεστών µε ακέραιους αριθµούς: 

 
    Public Sub main() 
        Dim i, j As Integer 
        Console.Write ("∆ώσε ένα ακέραιο ") 
        i = Console.ReadLine() 
        Console.Write ("∆ώσε ένα δεύτερο ακέραιο ") 
        j = Console.ReadLine() 
        Console.WriteLine(" {0} + {1} = {2}", i, j, i + j) 
        Console.WriteLine(" {0} - {1} = {2}", i, j, i - j) 
        Console.WriteLine(" {0} * {1} = {2}", i, j, i * j) 
        Console.WriteLine(" {0} / {1} = {2:N4}", i, j, i / j) 
        Console.WriteLine(" {0} ^ {1} = {2}", i, j, i ^ j) 
        Console.WriteLine(" {0} \ {1} = {2}", i, j, i \ j) 
        Console.WriteLine(" {0} mod {1} = {2}", i, j, i Mod j) 
    End Sub 

Παράδειγµα 3.2 

Το αποτέλεσµα του παραπάνω προγράµµατος εάν εισαχθούν οι αριθµοί 3 και 2 είναι: 

 
∆ώσε ένα ακέραιο 3 
∆ώσε ένα δεύτερο ακέραιο  2 
 3 + 2 = 5 
 3 - 2 = 1 
 3 * 2 = 6 
 3 / 2 = 1,5000 
 3 ^ 2 = 9 
 3 \ 2 = 1 
 3 mod 2 = 1 
 

Με την χρήση του υπολοίπου και της διαίρεση ακεραίων µπορεί να υπολογισθούν τα ψηφία ενός 

αριθµού. 

 
    Public Sub main() 
        Dim i, j, k As Integer 
        Console.Write(" ∆ώστε ένα πενταψήφιο ακέραιο ") 
        i = Console.ReadLine() 
        j = i Mod 10^4 
        Console.WriteLine(" Τα 4 τελευταία ψηφία είναι {0}", j) 
        k = j \ 10^3 
        Console.WriteLine(" Τα τέταρτο ψηφίο είναι {0}", k) 
    End Sub 

Παράδειγµα 3.3 

 

Εάν πληκτρολογηθεί ο αριθµός 12345 το αποτέλεσµα της εκτέλεσης του προγράµµατος είναι: 

 
∆ώστε ένα πενταψήφιο ακέραιο 12345 
Τα 4 τελευταία ψηφία είναι 2345 
Τα τέταρτο ψηφίο είναι 2 
 

Στο παραπάνω παράδειγµα το υπόλοιπο της διαίρεσης ενός αριθµού µε τον αριθµό 10000 υπολογίζει 

τα τελευταία 4 ψηφία του αριθµού και µε την ακέραια διαίρεση των 4 τελευταίων µε το αριθµό 1000 

υπολογίζεται το τέταρτο ψηφίο του αριθµού. 

3.2 Αλφαριθµητικοί τελεστές 

 

Με την χρήση των τελεστών & ή + µπορούµε να ενώσουµε αλφαριθµητικές µεταβλητές. Στο 

παρακάτω παράδειγµα ο τελεστής & χρησιµοποιείται για τον σχηµατισµό του ονοµατεπώνυµου από 

το όνοµα και το επώνυµο και ο τελεστής + χρησιµοποιείται για να ενώσει δύο αλφαριθµητικές τιµές 

και να τις εκτυπώσει. 

  
Public Sub main() 
   Dim onoma, eponymo, onomateponymo As String 
   Console.Write("∆ώστε το όνοµα σας ") 
   onoma = Console.ReadLine() 
   Console.Write("∆ώστε το επώνυµο σας ") 
   eponymo = Console.ReadLine() 
   'Χρήση τελεστή & για τον υπολογισµό ονοµατεπώνυµο 
   onomateponymo = onoma & " " & eponymo 
   'Χρήση τελεστή + για τον υπολογισµό ονοµατεπώνυµο 
   Console.WriteLine("Tο ονοµατεπώνυµο σας είναι " + onomateponymo) 
End Sub 

Παράδειγµα 3.4 

 

Το αποτέλεσµα της εκτέλεσης του προγράµµατος είναι: 
∆ώστε το όνοµα σας Ιωάννης 
∆ώστε το επώνυµο σας Γεωργίου 
Tο ονοµατεπώνυµο σας είναι Ιωάννης Γεωργίου 

3.3 Τελεστές σύγκρισης 

 

Τελεστής Περιγραφή Παράδειγµα Αποτέλεσµα 

= Ίσο 5 = 7 False 

<> ∆ιάφορο 5 <> 7 True 

< Μικρότερο 5 < 7 True 

<= Μικρότερο ή ίσο 5 <= 5 True 

> Μεγαλύτερο 5 > 7 False 

>= Μεγαλύτερο ή ίσο 7 >= 7 True 

Is Είναι το ίδιο 

αντικείµενο 

Α Is Β  

IsNot ∆εν είναι το ίδιο 

αντικείµενο 

Α IsNot Β  

TypeOf...Is Ο τύπος του 

αντικειµένου είναι 

TypeOf(A) is B  

Like Η αλφαριθµητική τιµή 

µοιάζει µε .. 

"En*" Like "Ena"  True 

Πίνακας 3.2 Τελεστές σύγκρισης 

 

Παράδειγµα αριθµητικών συγκρίσεων µεταξύ δύο αριθµών που εισάγονται από τον χρήστη: 

 
    Public Sub main() 
        Dim i, j As Integer 
        Console.Write("∆ώσε ένα ακέραιο ") 
        i = Console.ReadLine() 
        Console.Write("∆ώσε ένα δεύτερο ακέραιο ") 
        j = Console.ReadLine() 
        Console.WriteLine("{0} =  {1} = {2}", i, j, i = j) 
        Console.WriteLine("{0} <> {1} = {2}", i, j, i <> j) 
        Console.WriteLine("{0} >  {1} = {2}", i, j, i > j) 
        Console.WriteLine("{0} >= {1} = {2}", i, j, i >= j) 
        Console.WriteLine("{0} <  {1} = {2}", i, j, i < j) 
        Console.WriteLine("{0} <= {1} = {2}", i, j, i <= j) 
    End Sub 

Παράδειγµα 3.5 

 

Εάν εισαχθούν οι αριθµοί 5 και 7 τα αποτελέσµατα των συγκρίσεων είναι: 

 
∆ώσε ένα ακέραιο 5 
∆ώσε ένα δεύτερο ακέραιο 7 
5 =  7 = False 
5 <> 7 = True 
5 >  7 = False 
5 >= 7 = False 
5 <  7 = True 
5 <= 7 = True 
 

Οι τελεστές σύγκρισης µπορούν να χρησιµοποιηθούν και για την σύγκριση αλφαριθµητικών 

µεταβλητών ή χαρακτήρων.  

 

 Με την οδηγία Option Compare Text στην αρχή της λειτουργικής µονάδος, η σύγκριση γίνεται µε 

βάση της ρυθµίσεις της επιλεγµένης γλώσσας των Windows. Με την επιλογή αυτή οι πεζοί 

χαρακτήρες θεωρούνται και ταξινοµούνται ως ίσοι µε τους αντίστοιχους κεφαλαίους χαρακτήρες.  Με 

την οδηγία Option Compare Binary η σύγκριση γίνεται µε βάση των κωδικό (αριθµό) που αντιστοιχεί 

σε κάθε χαρακτήρα. Με την επιλογή Option Compare Binary οι πεζοί χαρακτήρες δεν είναι 

ισοδύναµοι µε τους κεφαλαίους χαρακτήρες.  Η επιλογή του τρόπου σύγκρισης των αλφαριθµητικών 

τιµών µπορεί να γίνει για όλο το πρόγραµµα µε την φόρµα ρυθµίσεων της εφαρµογής (όπως και οι 

επιλογές Option Explicit και Option Strict) 

 



Κεφάλαιο 3 Τελεστές – Μαθηµατικές Συναρτήσεις Σελίδα  27 

 

Σηµειώσεις Προγραµµατισµού  Γρηγόρης Τζιάλλας Τµήµα Ηλεκτρονικής Τ.Ε.Ι. Λαµίας  

Το παρακάτω πρόγραµµα µε την επιλογή Option Compare Text εκτυπώνει ως αποτέλεσµα της 

σύγκρισης true («Welcome» = «welcome»).  Το αποτέλεσµα  της σύγκρισης θα ήταν false εάν είχαµε 

επιλέξει Option Compare Binary. 

    
Option Compare Text 
Module stringComparisonς 
    Public Sub main() 
        Console.WriteLine("WELCOME" >= "welcome") 
    End Sub 
End Module 

Παράδειγµα 3.6 

 

Οι τελεστές Is, IsNot και TypeOf .. Is χρησιµοποιούνται για τις συγκρίσεις αντικειµένων. 

Παραδείγµατα τέτοιων συγκρίσεων δίνονται σε επόµενο κεφάλαιο. 

 

Με τον τελεστή Like µπορούµε να συγκρίνουµε εάν κάποια αλφαριθµητική τιµή µοιάζει ή εµπεριέχει 

κάποια άλλη. Στις συγκρίσεις µπορούµε να χρησιµοποιήσουµε τους ειδικούς χαρακτήρες ?, *, #  ή να 

χρησιµοποιήσουµε σύνολα η περιοχές χαρακτήρων. Με τον χαρακτήρα ? η σύγκριση αληθεύει για 

οποιονδήποτε χαρακτήρα υπάρχει στη θέση αυτή. Με τον χαρακτήρα # η σύγκριση αληθεύει για 

οποιονδήποτε αριθµητικό χαρακτήρα (χαρακτήρες από ‘0’ έως ‘9’) υπάρχει στη θέση αυτή. Με τον 

χαρακτήρα * η σύγκριση αληθεύει για οποιαδήποτε ακολουθία χαρακτήρων υπάρχει στη θέση αυτή. 

Αντί του χαρακτήρα ? µπορεί να χρησιµοποιηθεί ένα σύνολο εναλλακτικών χαρακτήρων (πχ 

[αο]σηµαίνει ότι στην θέση αυτή µπορεί να υπάρχουν ή χαρακτήρες ‘α’ ή ‘ο’). Αντί του χαρακτήρα ? 

µπορεί να χρησιµοποιηθεί επίσης µία περιοχή εναλλακτικών χαρακτήρων (πχ [a-z] σηµαίνει ότι στη 

θέση αυτή µπορεί να υπάρχουν οι χαρακτήρες από a έως z, δηλαδή οι πεζοί χαρακτήρες του λατινικού 

αλφαβήτου).  Μπορούµε επίσης να αποκλείσουµε ένα σύνολο ή µια περιοχή χαρακτήρων  µε την 

χρήση του χαρακτήρα ! (πχ [!αο] σηµαίνει ότι στην θέση αυτή µπορεί να υπάρχει οποιοσδήποτε 

χαρακτήρας εκτός από τους χαρακτήρες ‘α’ ή ‘ο’) 

 

Παράδειγµα χρήσης του τελεστή like δίνεται στο παρακάτω πρόγραµµα. Το πρόγραµµα ζητά την 

εισαγωγή ενός ονόµατος και στην συνέχεια µε την χρήση του τελεστή like κάνει διάφορες συγκρίσεις 

για το όνοµα. 

 
Public Sub main() 
    Dim s As String 
    Console.Write("∆ώστε ένα όνοµα ") 
    s = Console.ReadLine() 
    'Είναι οι συγκρινόµενες τιµές είναι ίδιες;  
    Console.WriteLine("Το όνοµα " & s & " είναι Ηλίας {0}", s Like "Ηλίας") 
    'Είναι παρόµοιο µε 'Ηλ??ς 
     Console.WriteLine("Το όνοµα " & _ 
          s & " είναι παρόµοιο µε 'Ηλ??ς'{0}", s Like "Ηλ??ς") 
    'Αρχίζει µε την λέξη "Ηλ";  
    Console.WriteLine("Το όνοµα " & s & " αρχίζει από 'Ηλ' {0}", s Like "Ηλ*") 
    'Εµπεριέχει την λέξη "λία";  
    Console.WriteLine("Το όνοµα " & _ 
          s & " περιέχει την λέξη 'λία' {0}", s Like "*λία*") 
    'Τελειώνει σε τον χαρακτήρα "ς"  
    Console.WriteLine("Το όνοµα " & s & " τελειώνει µε 'ς' {0}", s Like "*ς") 
    'Η κατάληξη είναι "ας" ή "ος";  
    Console.WriteLine("Το όνοµα " & _ 
          s & " έχει κατάληξη 'ας' ή 'ος' {0}", s Like "*[αο]ς") 
    'Αρχίζει µε κεφαλαίο φωνήεν;  
     Console.WriteLine("Το όνοµα " & _  
          s & " αρχίζει µε κεφαλαίο φωνήεν {0}", s Like "[ΑΕΗΙΥΟΩ]*") 
    'Περιέχει τους αριθµητικούς χαρακτήρες 0 έως 9;   
     Console.WriteLine("Το όνοµα " & s & " περιέχει αριθµούς {0}", s Like "*#*") 
    '∆εν περιέχει τους αριθµητικούς χαρακτήρες 0 και 1;   
     Console.WriteLine("Το όνοµα " & _) 
          s & "δεν περιέχει 0 καί 1 {0}", s Like "*[!01]*") 
End Sub 

Παράδειγµα 3.7 

Το αποτέλεσµα της εκτέλεσης του προγράµµατος αν δοθεί το όνοµα «Ηλίας» ως είσοδος είναι το 

παρακάτω 

 
Το όνοµα Ηλίας είναι Ηλίας True 
Το όνοµα Ηλίας είναι παρόµοιο µε 'Ηλ??ς'True 
Το όνοµα Ηλίας αρχίζει από 'Ηλ' True 
Το όνοµα Ηλίας περιέχει την λέξη 'λία' True 
Το όνοµα Ηλίας τελειώνει µε 'ς' True 
Το όνοµα Ηλίας έχει κατάληξη 'ας' ή 'ος' True 
Το όνοµα Ηλίας αρχίζει µε κεφαλαίο φωνήεν True 
Το όνοµα Ηλίας περιέχει αριθµούς False 
Το όνοµα Ηλίας δεν περιέχει 0 και 1 True 

3.4 Λογικοί τελεστές 

 

Τελεστής Περιγραφή Παράδειγµα Αποτέλεσµα 

Not Λογική άρνηση Not A False εάν Α = True 

True εάν Α = False 

And Λογική σύζευξη A And B True εάν Α και Β True 

Or Λογική διάζευξη Α or B True εάν Α ή Β True 

Xor Αποκλειστική διάζευξη Α Xor B True εάν  

Α True και B False ή 

Α False και B True 

AndAlso Λογική σύζευξη A AndAlso B True εάν Α και Β True 

OrAlso Λογική διάζευξη Α orAlso B True εάν Α ή Β True 

Πίνακας 3.3 Λογικοί Τελεστές 

 

Ο λογικός τελεστής AndAlso είναι παρόµοιος µε τον τελεστή And όπως και ο τελεστής OrAlso µε τον 

τελεστή Or. Η διαφορά των τελεστών AndAlso και OrAlso από τους τελεστές And και Or βρίσκεται 

στον τρόπο µε τον οποίο υπολογίζουν το αποτέλεσµα της λογικής πράξης. Οι τελεστές  AndAlso και 

OrAlso εξετάζουν την πρώτη λογική τιµή και µόνον εάν χρειάζεται εξετάζουν την δεύτερη λογική 

τιµή και κάνουν την λογική πράξη (π.χ. στην λογική σύζευξη A AndAlso B εάν Α=False επιστρέφεται 

False χωρίς να υπολογισθεί η τιµή του Β). Οι τελεστές AndAlso και OrAlso είναι χρήσιµοι σε 

περιπτώσεις όπου ο υπολογισµός της δεύτερης λογικής τιµής απαιτεί σηµαντικό χρόνο (π.χ. εάν η 

τιµή αυτή επιστρέφεται από µία συνάρτηση η οποία χρειάζεται σηµαντικό χρόνο για την εκτέλεση 

της). 

 

Οι λογικοί τελεστές χρησιµοποιούνται για τον σχηµατισµό πολύπλοκων λογικών εκφράσεων και 

συγκρίσεων.  

 

Παράδειγµα 
(x >= 0)  And  (x <=10) Εξετάζει εάν το x ανήκει στο διάστηµα 0..10   
(x=1) or (x=3)      Εξετάζει εάν το x είναι ίσο µε 1 ή 3 

3.5 Τελεστές ανάθεσης τιµής 

Εκτός του τελεστή ανάθεσης τιµής = µπορεί να χρησιµοποιηθούν και οι παρακάτω τελεστές 

 

Τελεστής Παράδειγµα  Ισοδύναµο µε 

= x=2  

^= x^=2 x=x
2
 

+= x+=1 x=x+1 

-= x-=1 x=x-1 

*= x*=3 x=x*3 

/= x/= 3 x=x/3 

\= x\= 3 x=x\3 

&= x&=”test” 

(x αλφαριθµητική µεταβλητή) 

x=x&”test” 

Πίνακας 3.4 Τελεστές ανάθεσης τιµής 

3.6 Χρήση παρενθέσεων - Προτεραιότητα τελεστών 

 

Οι παρενθέσεις χρησιµοποιούνται για τον καθορισµό της προτεραιότητας των πράξεων σε 

πολύπλοκες αριθµητικές ή λογικές παραστάσεις. 

 

Οι πράξεις σε αριθµητικές ή λογικές παραστάσεις εκτελούνται σύµφωνα µε την προτεραιότητα των 

τελεστών η οποία φαίνεται στον παρακάτω πίνακα. Οι πράξεις για τελεστές που έχουν την ίδια 

προτεραιότητα γίνονται από τα αριστερά προς τα δεξιά. 

 

Τελεστής Περιγραφή 

(  ) Παρένθεση 

^ Ύψωση σε δύναµη  

- Αρνητικό πρόσηµο 

* / Πολλαπλασιασµός - ∆ιαίρεση 

\ Ακέραια διαίρεση 

Mod Υπόλοιπο διαίρεσης 

+ - Πρόσθεση, αφαίρεση 

& Αλφαριθµητική ένωση 

=, <>, <, <=, >, >=, Like, Is,  

IsNot, TypeOf...Is 

Τελεστές σύγκρισης 

Not Λογική άρνηση 

And, AndAlso Λογική σύζευξη 

Xor, Or, OrAlso Λογική και αποκλειστική διάζευξη 

Πίνακας 3.5 Προτεραιότητας Τελεστών 

3.7 Μαθηµατικές συναρτήσεις 

 

Η κλάση αντικειµένων Math παρέχει χρήσιµες µαθηµατικές συναρτήσεις. Οι κυριότερες από αυτές 

φαίνονται στον παρακάτω πίνακα.   

 

Όνοµα συνάρτησης Περιγραφή Παράδειγµα 

Abs(x)  Επιστρέφει την  απόλυτη τιµή του χ Abs(23.7) = 23.7 

Abs(0) = 0  

Abs(-23.7) = 23.7 

Ceiling(x)  Στρογγυλοποιεί τον αριθµό x στον 

µικρότερο κοντινό ακέραιο 

Ceiling(9.2) = 10.0  

Ceiling(-9.8) is -9.0 

Cos(x)  Επιστρέφει το συνηµίτονο του x (x σε 

ακτίνια) 

Cos(0.0) = 1.0 

Exp(x)  e
x
  Exp(1.0) = 2.71828182845905 

Exp(2.0) =  7.38905609893065 

Floor(x)  Στρογγυλοποιεί τον αριθµό x στον 

µεγαλύτερο κοντινό ακέραιο 

Floor(9.2) = 9.0  

Floor(-9.8) = -10.0 

Log(x) Λογάριθµος του x Log(2.7182818284590451) = 1.0 

Log(7.3890560989306504)= 2.0 

Max(x, y) Επιστρέφει την µεγαλύτερη τιµή των x 

και  y 

Max(2.3, 12.7) = 12.7  

Max(-2.3, -12.7) = -2.3 

Min(x, y)  Επιστρέφει την µικρότερη τιµή των x 

και  y 

Min(2.3, 12.7) = 2.3  

Min(-2.3, -12.7) = -12.7 

Pow(x, y)  Επιστρέφει την y δύναµη του x Pow(2.0, 7.0) = 128.0 

Pow(9.0, .5) = 3.0 

Sin(x)  Επιστρέφει το ηµίτονο του x (x σε 

ακτίνια) 

Sin(0.0) = 0.0 

Sqrt(x)  Επιστρέφει την τετραγωνική ρίζα του x Sqrt(9.0) = 3.0 

Sqrt(2.0) = 1.4142135623731 

Tan(x)  Επιστρέφει την εφαπτοµένη του x (x σε 

ακτίνια) 

 Tan(0.0)=0.0 

Πίνακας 3.6 Μαθηµατικές συναρτήσεις 

 

Η κλάση αντικειµένων Math ορίζει επίσης τις σταθερές: 

• PI  Math.PI = 3.1415926535897931 

• E   Math.E = 2.7182818284590451 

 

Για την χρήση των συναρτήσεων και των σταθερών αυτών σε αριθµητικές παραστάσεις πρέπει να 

προηγηθεί πρώτα το όνοµα της κλάσης Math ακολουθούµενο από τελεία και το όνοµα της 

συνάρτησης ή της σταθερής. 

 

Το παρακάτω παράδειγµα χρησιµοποιεί µαθηµατικές συναρτήσεις για την µετατροπή γωνίας σε 

ακτίνια και τον υπολογισµό του ηµίτονου και του συνηµίτονου της γωνίας. 

 
    Public Sub main() 
        Dim x As Double 
        Dim xRad As Double 'Γωνία σε ακτίνια 
        Dim sinX As Double 'Ηµίτονο της γωνίας 
        Dim cosX As Double 'Συνηµίτονο της γωνίας 
        Console.Write("∆ώστε µία γωνία σε µοίρες ") 
        x = Console.ReadLine() 
        xRad = Math.PI * x / 180 'Μετατροπή γωνίας σε ακτίνια 
        sinX = Math.Sin(xRad) 'Υπολογισµός ηµίτονου 
        cosX = Math.Cos(xRad) 'Υπολογισµός συνηµίτονου 
        Console.WriteLine( _ 
           "H γωνία {0:N4} µοιρών είναι {1:N4} ακτίνια ", x, xRad) 
        Console.WriteLine( _ 
           "To ηµίτονο της γωνίας {0:N4} είναι    {1:N4} ", x, sinX) 
        Console.WriteLine( _ 
           "To συνηµίτονο της γωνίας {0:N4} είναι {1:N4} ", x, CosX) 
    End Sub 

Παράδειγµα 3.8 

 

Το αποτέλεσµα της εκτέλεσης του παραπάνω προγράµµατος εάν δοθεί γωνία 60 µοιρών είναι: 

 
∆ώστε µία γωνία σε µοίρες 60 
H γωνία 60,0000 µοιρών είναι 1,0472 ακτίνια 
To ηµίτονο της γωνίας 60,0000 είναι    0,8660 
To συνηµίτονο της γωνίας 60,0000 είναι 0,5000 

3.8 Ασκήσεις κεφαλαίου 3 

 

Άσκηση 3.1 

∆ηµιουργήστε εφαρµογή lab_3_ask_1 η οποία ζητά από τον χρήστη να εισάγει την ακτίνα 

ενός κύκλου και στην συνέχεια υπολογίζει την περίµετρο και το εµβαδόν του κύκλου.  Η 

περίµετρος και το εµβαδόν να  εκτυπωθούν µε ακρίβεια 4 δεκαδικών στοιχείων. 

 

Άσκηση 3.2 

∆ηµιουργήστε εφαρµογή lab_3_ask_2 η οποία ζητά από τον χρήστη να εισάγει µία από τις 

γωνίες (σε µοίρες) και το µήκος της υποτείνουσας  ενός ορθογωνίου τριγώνου και στην 

συνέχεια υπολογίζει και εκτυπώνει στην οθόνη τα µήκη των πλευρών του τριγώνου µε 

ακρίβεια 2 δεκαδικών στοιχείων. 

 

Υπόδειξη: Η γωνία του τριγώνου δίνεται σε µοίρες και θα πρέπει να την µετατρέψετε σε 

ακτίνια (360
o
 αντιστοιχούν σε  2π ακτίνια).  

 

Άσκηση 3.3 

∆ηµιουργήστε εφαρµογή lab_3_ask_3 η οποία ζητά από τον χρήστη να εισάγει το όνοµα µιας 

Ελληνικής πόλης. Η εφαρµογή εξετάζει και εκτυπώνει εάν το όνοµα της πόλης περιέχει 

αριθµούς, πεζούς λατινικούς χαρακτήρες, εάν αρχίζει µε κεφαλαίο Ελληνικό χαρακτήρα και 

εάν  η κατάληξη του είναι «ος». 

 

Άσκηση 3.4 

∆ηµιουργήστε εφαρµογή lab_3_ask_4 η οποία ζητά από τον χρήστη να εισάγει τον µισθό ενός 

υπαλλήλου και στην συνέχεια υπολογίζει και εκτυπώνει πόσα χαρτονοµίσµατα των 500, 200, 

100, 50, 20 και 10 ευρώ θα χρειασθούν για να πληρωθεί το ποσό αυτό. 

 

Υπόδειξη: Χρησιµοποιήστε την ακέραια διαίρεση και το υπόλοιπο της διαίρεσης µε τα 

αντίστοιχα ποσά των χαρτονοµισµάτων . Αρχίστε τον υπολογισµό µε χαρτονοµίσµατα των 500 

ευρώ. 

4 Επιλογή και Επανάληψη 

4.1 Η Εντολή Επιλογής  if ..  then 

 

Η εντολή  If .. Then .. χρησιµοποιείται για την λήψη λογικών αποφάσεων σε ένα πρόγραµµα. Η 

εντολή αυτή  έχει διάφορες µορφές σύνταξης οι οποίες περιγράφονται στις παρακάτω ενότητες. 

Η µορφή If .. Then .. End If 

 

Η εντολή If  <λογική συνθήκη> Then <τµήµα προγράµµατος> εκτελεί ένα τµήµα προγράµµατος 

εφόσον ικανοποιείται κάποια λογική συνθήκη. 

 

Στο παρακάτω παράδειγµα η εντολή αυτή χρησιµοποιείται για να ελέγξει εάν ο αριθµός ο οποίος 

εισάγεται από τον χρήστη είναι θετικός. Στην περίπτωση που ο αριθµός είναι αρνητικός, τότε 

εκτυπώνεται το µήνυµα ότι ο αριθµός είναι αρνητικός και σταµατά η εκτέλεση της διαδικασίας µε την 

εντολή exit sub. 

 
    Public Sub main() 
        Dim i As Integer 
        Console.Write("∆ώστε ένα θετικό ακέραιο") 
        i = Console.ReadLine() 
        'Έλεγχος αν ο αριθµός είναι θετικός   
        If i <= 0 Then 
            Console.WriteLine("Λάθος, ο αριθµός είναι αρνητικός") 
            Exit Sub  ' Τέλος της διαδικασίας   
        End If 
        'Εκτύπωση τετραγωνικής ρίζας 
        Console.WriteLine( _ 
             "Η τετραγωνική του ρίζα είναι {0:N4}", Math.Sqrt(i)) 
    End Sub  

Παράδειγµα 4.1 

 

Στην περίπτωση όπου το τµήµα προγράµµατος το οποίο εκτελείται αποτελείται από µία µόνο εντολή, 

η εντολή µπορεί να συνταχθεί σε µία µόνο γραµµή. 

 

π.χ. 
if i <= 0 then Exit Sub 

Η µορφή If .. Then .. Else ..  End If 

 

Η εντολή If  <λογική συνθήκη> Then <τµήµα 1> Else <τµήµα 2> End If  εκτελεί το πρώτο τµήµα 

προγράµµατος εφόσον η λογική συνθήκη της εντολής είναι αληθής διαφορετικά εκτελεί το δεύτερο 

τµήµα προγράµµατος το οποίο ορίζεται µεταξύ των λέξεων else και end if. 

 
    Public Sub main() 
        Dim i As Integer 
        Console.Write("∆ώστε ένα ακέραιο") 
        i = Console.ReadLine() 
        'Έλεγχος αν ο αριθµός είναι θετικός ή αρνητικός  
        If i <= 0 Then 
            Console.WriteLine("O αριθµός είναι αρνητικός") 
        Else 
            Console.WriteLine("O αριθµός είναι θετικός") 
        End If 
    End Sub 

Παράδειγµα 4.2 

Στο παραπάνω παράδειγµα ελέγχεται εάν ο αριθµός ο οποίος εισάγεται από τον χρήστη είναι 

µεγαλύτερος ή ίσος του µηδενός και αναλόγως εκτυπώνεται το αντίστοιχο µήνυµα. 

 

Το αποτέλεσµα της εκτέλεσης του προγράµµατος εάν εισαχθεί ο αριθµός –45 είναι: 

 
∆ώστε ένα ακέραιο -45 
O αριθµός είναι αρνητικός 

Ένθεση εντολών If .. Then 

 

Η εντολή επιλογής If .. Then µπορεί να εµπεριέχει ή ενθέτει και άλλες εντολές επιλογής. 

 

Η χρήση ένθετων (εµφωλευµένων) εντολών If .. Then γίνεται όταν υπάρχουν περισσότερες από δύο 

εναλλακτικές περιπτώσεις διακλάδωσης. Εάν για παράδειγµα χρειάζεται να γίνει έλεγχος ενός 

αριθµού εάν είναι αρνητικός, µηδενικός ή θετικός, τότε γίνεται χρήση ένθετων εντολών επιλογής if .. 

then. 

 
    Public Sub main() 
        Dim i As Integer 
        Console.Write("∆ώστε ένα ακέραιο ") 
        i = Console.ReadLine() 
        'Έλεγχος του αριθµού  
        If i < 0 Then 
            Console.WriteLine("O αριθµός είναι αρνητικός") 
        Else 
            If i = 0 Then 
                Console.WriteLine("O αριθµός είναι µηδενικός") 
            Else 
                Console.WriteLine("O αριθµός είναι θετικός") 
            End If 
        End If 
    End Sub 

Παράδειγµα 4.3 

 

Το αποτέλεσµα της εκτέλεσης του παραπάνω προγράµµατος εάν δοθεί ο αριθµός 0 είναι: 
∆ώστε ένα ακέραιο 0 
O αριθµός είναι µηδενικός 

Ένθεση µε τη µορφή If .. then .. elseif .. else ..  end if 

 

Η σύνταξη των ένθετων εντολών if .. then  µπορεί να απλοποιηθεί µε την χρήση της λέξης elseIf. 

Με την χρήση της ένθετης µορφής if .. then .. elseIf .. else .. end if το προηγούµενο παράδειγµα 

µπορεί να απλοποιηθεί όπως φαίνεται στο παρακάτω πρόγραµµα: 

 
    Public Sub main() 
        Dim i As Integer 
        Console.Write("∆ώστε ένα ακέραιο ") 
        i = Console.ReadLine() 
        'Έλεγχος του αριθµού  
        If i < 0 Then 
            Console.WriteLine("O αριθµός είναι αρνητικός") 
        ElseIf i = 0 Then 
            Console.WriteLine("O αριθµός είναι µηδενικός") 
        Else 
            Console.WriteLine("O αριθµός είναι θετικός") 
        End If 
    End Sub 

Παράδειγµα 4.4 

4.2 Η εντολή πολλαπλής επιλογής Select 

 

Η πολλαπλή επιλογή χρησιµοποιείται για την λήψη λογικών αποφάσεων όταν οι εναλλακτικές 

περιπτώσεις είναι περισσότερες από δύο. Η εντολή πολλαπλής επιλογής Select εξετάζει την τιµή 

κάποιας µεταβλητής (Integer, String κλπ) και αναλόγως της τιµής της µεταβλητής εκτελεί 

διαφορετικό τµήµα προγράµµατος. 

 

Το παρακάτω παράδειγµα µε την εντολή Select εξετάζει την τιµή ενός µονοψήφιου θετικού ακέραιου 

ο οποίος εισάγεται από τον χρήστη, υπολογίζει την κατάλληλη αλφαριθµητική τιµή που αντιστοιχεί 

στον αριθµό και τον εκτυπώνει ολογράφως. 

 
    Public Sub main() 
        'Πρόγραµµα το οποίο ζητά έναν αριθµό από 0 έως 9 
        'και τον τυπώνει ολογράφως 
        Dim i As Integer 'Ο αριθµός που θα µετατραπεί 
        Dim s As String  'Ο αριθµός ολογράφως 
        Console.Write("∆ώστε ένα ακέραιο από 0 έως 9 : ") 
        i = Console.ReadLine() 
        'Έλεγχος του αριθµού  
        Select Case i 
            Case 0 
                s = "µηδέν" 
            Case 1 
                s = "ένα" 
            Case 2 
                s = "δύο" 
            Case 3 
                s = "τρία" 
            Case 4 
                s = "τέσσερα" 
            Case 5 
                s = "πέντε" 
            Case 6 
                s = "έξι" 
            Case 7 
                s = "επτά" 
            Case 8 
                s = "οκτώ" 
            Case 9 
                s = "εννέα" 
            Case Else   'Εάν ο αριθµός δεν είναι από 0 εώς 9 
                s = "*****" 
        End Select 
        Console.WriteLine("Πληκτρολογήθηκε ο αριθµός {0}", s) 
    End Sub 

Παράδειγµα 4.5 

 

Εάν δοθεί ο αριθµός 4, τότε το αποτέλεσµα της εκτέλεσης του προγράµµατος είναι: 

 
∆ώστε ένα ακέραιο από 0 έως 9 : 4 
Πληκτρολογήθηκε ο αριθµός τέσσερα  
 

Στην περίπτωση όπου η ίδια επιλογή γίνεται για δύο ή περισσότερες τιµές, µπορούµε να παραθέσουµε 

τις τιµές µετά την λέξη case µε την χρήση κόµµατος. 

 

Π.χ. αν i είναι η µεταβλητή πολλαπλής επιλογής 
Case 1,3,5  'ισοδυναµεί µε i=1 or i=2 or i=3 
 

Εάν η ίδια επιλογή γίνεται για ένα εύρος τιµών, το εύρος τιµών µπορεί να ορισθεί µε την χρήση της 

αρχικής τιµής, της λέξης To και της τελική τιµής του εύρους 

πχ αν i είναι η µεταβλητή πολλαπλής επιλογής 
Case 0 To 5  'ισοδυναµεί µε i>=0 and i<=5 
 

Επίσης µε την χρήση του τελεστή Is µπορούµε να συγκρίνουµε την τιµή της µεταβλητής πολλαπλής 

επιλογής µε άλλες τιµές ή µεταβλητές. 

 

πχ αν i είναι η µεταβλητή πολλαπλής επιλογής 
Case Is < 0  'ισοδυναµεί µε i<0    
 

Η µεταβλητή η οποία χρησιµοποιείται στην  πολλαπλή επιλογή µπορεί να είναι και άλλος τύπος 

δεδοµένων, όπως πραγµατικός αριθµός, χαρακτήρας ή αλφαριθµητική µεταβλητή. 

 
    Public Sub main() 
        'Πρόγραµµα το οποίο διαβάζει ένα χαρακτήρα 
        'και εκτυπώνει την περιγραφή του 
        Dim c As Char 'Χαρακτήρας ο οποίος εισάγεται 
        Dim s As String 'Περιγραφή του χαρακτήρα  
        Console.Write("∆ώστε ένα ελληνικό χαρακτήρα : ") 
        c = Console.ReadLine() 
        'Πολλαπλή επιλογή ανάλογα µε το χαρακτήρα 
        Select Case c 
            Case "α" To "ω" 
                s = "ελληνικός πεζός χαρακτήρας" 
            Case "ά", "έ", "ό", "ή", "ύ", "ί", "ώ" 
                s = "ελληνικός πεζός χαρακτήρας µε τόνο" 
            Case "A" To "Ω" 
                s = "ελληνικός κεφαλαίος χαρακτήρας" 
            Case Is < " " 
                s = "ειδικός χαρακτήρας" 
            Case Else 
                s = "κάποιος άλλος χαρακτήρας" 
        End Select 
        Console.WriteLine("Πληκτρολογήθηκε {0}", s) 
    End Sub 

Παράδειγµα 4.6 

 

Το παραπάνω παράδειγµα χρησιµοποιεί ως µεταβλητή πολλαπλής επιλογής ένα χαρακτήρα ο οποίος 

εισάγεται από το πληκτρολόγιο και κάνοντας χρήση πολλαπλών τιµών, εύρους τιµών και συγκρίσεων 

τιµών για την κάθε επιλογή εκτυπώνει την περιγραφή του χαρακτήρα που εισάγεται.  

 

Το αποτέλεσµα της εκτέλεσης του προγράµµατος εάν δοθεί ο χαρακτήρας «δ» είναι: 

 
∆ώστε ένα ελληνικό χαρακτήρα : δ 
Πληκτρολογήθηκε ελληνικός πεζός χαρακτήρας 

4.3 Η εντολή επιλογής IIF 

 Η εντολή IIF υπολογίζει µια λογική συνθήκη και αναλόγως επιστρέφει µια τιµή. Η σύνταξη της 

είναι: 

 
<µεταβλητή> = IIF(<συνθήκη>, <τιµή 1>, <τιµή 2>) 
 

Στο παρακάτω παράδειγµα η εντολή IIF χρησιµοποιείται για να ελέγξει εάν ένας αριθµός είναι άρτιος 

ή περιττός. Εάν το υπόλοιπο της διαίρεσης µε τον αριθµό 2 είναι 0, τότε ο αριθµός είναι άρτιος 

διαφορετικά είναι περιττός. 

    Public Sub main() 
        'Πρόγραµµα το οποίο ελέγχει 
        'εάν ένας αριθµός είναι περιττός 
        Dim i As Integer 
        Dim s As String  
        Console.Write("∆ώστε ένα ακέραιο : ") 
        i = Console.ReadLine() 
        s = IIf(i Mod 2 = 0, "άρτιος", "περιττός") 
        Console.WriteLine("Ο αριθµός {0} είναι {1}", i, s) 
    End Sub 

Παράδειγµα 4.7 

 

Το αποτέλεσµα της εκτέλεσης του προγράµµατος εάν δοθεί ο αριθµός 35 είναι: 

 
∆ώστε ένα ακέραιο : 35 
Ο αριθµός 35 είναι περιττός 
 

Η εντολή IIF επιστρέφει µια τιµή η οποία είναι τύπου Object. Εάν χρησιµοποιείται η επιλογή Option  

Strict On, τότε πρέπει να χρησιµοποιείται και η κατάλληλη συνάρτηση ή µέθοδος για την µετατροπή 

της επιστρεφόµενης τιµής. 

 

Στο προηγούµενο παράδειγµα, εάν ορισθεί η επιλογή Option Strict On, πρέπει να χρησιµοποιηθεί η 

συνάρτηση µετατροπής Cstr. 

 
        s = CStr(IIf(i Mod 2 = 0, "άρτιος", "περιττός")) 
 

4.4 Η εντολή πολλαπλής επιλογής Choose 

 

Η εντολή Choose επιστέφει µια τιµή ανάλογα µε την τιµή κάποιας µεταβλητής η οποία 

χρησιµοποιείται ως δείκτης. Η σύνταξη της εντολής αυτής είναι: 

 
<µεταβλητή> = Choose (δείκτης, <τιµή 1>, <τιµή 2>, .. , <τιµή Ν>) 
 

Εάν ο δείκτης έχει την τιµή 1, τότε επιστρέφεται η πρώτη τιµή, εάν έχει τιµή 2 επιστρέφεται η 

δεύτερη τιµή κ.ο.κ. 

 

Το παρακάτω παράδειγµα χρησιµοποιεί την εντολή choose για να εκτυπώσει ολογράφως την τιµή 

ενός ακεραίου αριθµού από 0 έως 9.  

 
Public Sub main() 
     'Πρόγραµµα το οποίο µετατρέπει αριθµό από 0 έως 9 
     'και τον τυπώνει ολογράφως µε την εντολή choose 
     Dim i As Integer 'Ο αριθµός που θα µετατραπεί 
     Dim s As String  'Ο αριθµός ολογράφως 
     Console.Write("∆ώστε ένα ακέραιο από 0 έως 9 : ") 
     i = Console.ReadLine() 
     'Έλεγχος του αριθµού  
     s = Choose(i + 1, "µηδέν", "ένα", "δύο", "τρία", "τέσσερα", _ 
                       "πέντε", "έξι", "επτά", "οκτώ", "εννέα") 
     Console.WriteLine("Πληκτρολογήθηκε ο αριθµός {0}", s) 
 End Sub 

Παράδειγµα 4.8 

 

Το αποτέλεσµα της εκτέλεσης του προγράµµατος εάν δοθεί ο αριθµός 0 είναι: 

 
∆ώστε ένα ακέραιο από 0 έως 9 : 0 
Πληκτρολογήθηκε ο αριθµός µηδέν 

4.5 Η εντολή επανάληψης For 

 

Η εντολή επανάληψης For επαναλαµβάνει την εκτέλεση ενός τµήµατος του προγράµµατος µε την 

χρήση µιας µεταβλητής η οποία χρησιµοποιείται ως δείκτης των επαναλήψεων. Η γενική σύνταξη της 

εντολής είναι:  

 
For <µεταβλητή> [As <Τύπος>]=  <αρχική τιµή> Το <τελική τιµή> [Step <βήµα>] 
      [if <συνθήκη> then Continue For] 
      [if <συνθήκη> then Exit For] 
      <Τµήµα του προγράµµατος το οποίο επαναλαµβάνεται> 

Next 

 

Ο ορισµός του βήµατος των επαναλήψεων είναι προαιρετικός και µπορεί να παραλειφθεί. Σε 

περίπτωση που δεν έχει ορισθεί βήµα, τότε το βήµα είναι ίσο µε την µονάδα.  

 

Ο δείκτης των επαναλήψεων στην πρώτη επανάληψη παίρνει την αρχική τιµή και αυξάνεται 

διαδοχικά µε το βήµα των επαναλήψεων µέχρι να φτάσει την τελική τιµή.  

 
    Public Sub main() 
        Dim i As Integer 
        For i = 1 To 5 
            Console.WriteLine(i) 
        Next 
    End Sub 

Παράδειγµα 4.9 

 

Στο παραπάνω παράδειγµα η µεταβλητή i χρησιµοποιείται ως δείκτης των επαναλήψεων µε αρχική 

τιµή 1 και τελική τιµή 5. Το αποτέλεσµα της εκτέλεσης του προγράµµατος είναι: 

 
1 
2 
3 
4 
5 
 

Το βήµα των επαναλήψεων µπορεί να είναι αρνητικό. Στην περίπτωση αυτή η τελική τιµή πρέπει να 

είναι µικρότερη από την τελική τιµή. 

 
    Public Sub main() 
        Dim i As Integer 
        For i = 5 To 1 Step -1 
            Console.WriteLine(i) 
        Next 
    End Sub 

Παράδειγµα 4.10 

 

Στο παραπάνω παράδειγµα η µεταβλητή i µειώνεται από την αρχική τιµή 5 στην τελική τιµή 1. Το 

αποτέλεσµα της εκτέλεσης του προγράµµατος είναι: 

 
5 
4 
3 
2 
1 
 

Ο δείκτης των επαναλήψεων µπορεί να είναι οποιοδήποτε τύπος δεδοµένων για τον οποίο ορίζονται 

οι τελεστές <=, >=, + και -. Μπορεί για παράδειγµα να χρησιµοποιηθεί πραγµατικός τύπος 

µεταβλητής διπλής ακρίβειας µε βήµα 0.1. 

    Public Sub main() 
        For x As Double = 10 To 10.5 Step 0.1 
            Console.WriteLine(x) 
        Next 
    End sub 

Παράδειγµα 4.11 

 

Το αποτέλεσµα της εκτέλεσης του παραπάνω προγράµµατος είναι: 

 
10 
10,1 
10,2 
10,3 
10,4 
10,5 
 

Η δήλωση του µεταβλητής του δείκτη των επαναλήψεων µπορεί να γίνει προαιρετικά στην ίδια 

γραµµή µε την εντολή For.  

 
    Public Sub main() 
        For i As Integer = 1 To 5 
            Console.WriteLine(i) 
        Next 
    End Sub 

 Παράδειγµα 4.12 

 

Με την προαιρετική χρήση της εντολής Continue For σταµατά η εκτέλεση της τρέχουσας 

επανάληψης και οι επαναλήψεις συνεχίζονται µε την επόµενη τιµή του δείκτη των επαναλήψεων. 

 
    Public Sub main() 
        For i As Integer = 1 To 5 
            If i=3 then Continue For 
            Console.WriteLine(i) 
        Next 
    End Sub 

Παράδειγµα 4.13 

 

Στο παραπάνω παράδειγµα σταµατά η εκτέλεση της επανάληψης όταν ο δείκτης έχει την τιµή 3. Το 

αποτέλεσµα της εκτέλεσης του προγράµµατος είναι: 

 
1 
2 
4 
5 
 

Με την προαιρετική εντολή Exit For διακόπτεται η εντολή επανάληψης 

 
    Public Sub main() 
        For i As Integer = 1 To 5 
            If i=3 then Exit For 
            Console.WriteLine(i) 
        Next 
    End Sub 

Παράδειγµα 4.14 

 

Στο παραπάνω παράδειγµα διακόπτονται όλες οι επαναλήψεις όταν ο δείκτης έχει την τιµή 3. Το 

αποτέλεσµα της εκτέλεσης του προγράµµατος είναι: 

 
1 
2 

Ο δείκτης των επαναλήψεων µπορεί να χρησιµοποιηθεί σε διάφορους υπολογισµός όπως για 

παράδειγµα την εύρεση του αθροίσµατος ή του παραγοντικού ενός ακεραίου. 

 
 Public Sub main() 
        'Πρόγραµµα το οποίο υπολογίζει 
        'το άθροισµα των αριθµών από 1 έως 100 
        Dim sum As Integer = 0 
        For i As Integer = 1 To 100 
            sum += i'Πρόσθεση του i στο sum 
        Next 
        Console.WriteLine(sum) 
    End Sub 

Παράδειγµα 4.15 Εύρεση του αθροίσµατος των αριθµών από 1 έως 100 

 

Το  παραπάνω παράδειγµα χρησιµοποιεί µια µεταβλητή sum ως αθροιστή µε αρχική τιµή 0. Σε κάθε 

επανάληψη ο δείκτης επανάληψης  o οποίος παίρνει τιµές από 1 έως 100 προστίθεται στην µεταβλητή 

sum. To sum µετά το τέλος των επαναλήψεων είναι ίσο µε το άθροισµα των ακεραίων από 1 έως 100.  

 
    Public Sub main() 
        'Πρόγραµµα το οποίο υπολογίζει 
        'το παραγοντικό ενός αριθµού 
        'n = 1*2*3*..n 
        Dim n As Integer ' Ο ακέραιος αριθµός 
        Dim p As Integer = 1 ' Tο παραγοντικό 
        Console.Write("∆ώστε έναν ακέραιο: ") 
        n = Console.ReadLine() 
        For i As Integer = 2 To n 
            p *= i ' p = p * i 
        Next 
        Console.WriteLine("Το παραγοντικό του {0} είναι {1}", n, p) 
    End Sub 

Παράδειγµα 4.16 Εύρεση του παραγοντικού ενός ακέραιου 

 

Το  παραπάνω παράδειγµα χρησιµοποιεί την µεταβλητή p µε αρχική τιµή 1 για να υπολογίσει το 

παραγοντικό ενός ακέραιου n ο οποίος εισάγεται από τον χρήστη. Σε κάθε  επανάληψη το p γίνεται 

ίσο µε γινόµενο της προηγούµενης τιµής του επί τον δείκτη των επαναλήψεων i o οποίος παίρνει τις 

τιµές από 2 έως n.  

 

Επανάληψη Τιµή του n Τιµή του p 

1 2 1*2 

2 3 1*2*3 

3 4 1*2*3*4 

..   

n-1 n 1*2*3*4*..*n 

   

 

Το αποτέλεσµα της εκτέλεσης του προγράµµατος εάν δοθεί ο αριθµός 6 είναι: 

 
∆ώστε έναν ακέραιο: 6 
Το παραγοντικό του 6 είναι 720 

Ένθεση εντολών For 

 

Το τµήµα του προγράµµατος το οποίο επαναλαµβάνεται µε µία εντολή For µπορεί να εµπεριέχει 

κάποια άλλη εντολή For. Είναι δηλαδή δυνατή η ένθεση εντολών For και γενικότερα εντολών 

επανάληψης. 

    Public Sub main() 
        Dim i, j As Integer 
        For i = 1 To 3 
            For j = 1 To 2 
                Console.WriteLine("i={0} j={1}", i, j) 
            Next 
        Next 
    End sub 

Παράδειγµα 4.17 

 

Το παραπάνω πρόγραµµα ενθέτει µια εντολή επανάληψης µε δείκτη j ο οποίος παίρνει τις τιµές από 1 

έως 2 σε µια εντολή επανάληψης µε δείκτη i ο οποίος παίρνει τις τιµές από 1 έως 3. Η ένθεση έχει ως 

αποτέλεσµα η εντολή επανάληψης µε δείκτη j να εκτελείται για κάθε αλλαγή της τιµής του i. Το 

αποτέλεσµα της εκτέλεσης του παραπάνω προγράµµατος έχει σαν αποτέλεσµα: 

 
i=1 j=1 
i=1 j=2 
i=2 j=1 
i=2 j=2 
i=3 j=1 
i=3 j=2 
 

Το παρακάτω παράδειγµα χρησιµοποιεί ένθετες εντολές For για να εκτυπώσει ένα πλαίσιο 5 γραµµών 

και 6 στηλών µε τους χαρακτήρες «*». Στο τέλος της ένθετης εντολής επανάληψης µε δείκτη j η 

οποία εκτυπώνει 6 χαρακτήρες «*» στην ίδια γραµµή, εκτελείται η εντολή Console.WriteLine() µε 

την οποία γίνεται η αλλαγή γραµµής για κάθε διαφορετική τιµή του i. 

 
    Public Sub main() 
        Dim i, j As Integer 
        For i = 1 To 5 
            For j = 1 To 6 
                Console.Write("*") 
            Next 
            Console.WriteLine() 
        Next 
    End Sub 

Παράδειγµα 4.18 

 

Το αποτέλεσµα της εκτέλεσης του παραπάνω προγράµµατος είναι: 

 
****** 
****** 
****** 
****** 
****** 

4.6 Η εντολή επανάληψης Do Loop 

   

Η εντολή επανάληψης Do Loop επαναλαµβάνει διαρκώς την εκτέλεση ενός τµήµατος του 

προγράµµατος. Οι επαναλήψεις  συνεχίζουν ή σταµατούν ανάλογα µε την τιµή κάποιας λογικής 

συνθήκης. Εάν η λογική συνθήκη ελέγχεται στην αρχή τότε η εντολή έχει την µορφή: 

 
Do [ While ή Until <συνθήκη> ] 
 <εντολή> 
 [if <συνθήκη> then Exit  Do] 
 [if <συνθήκη> then Continue Do] 
 <εντολή> 
Loop  
 

Εάν η λογική συνθήκη ελέγχεται στην τέλος τότε η εντολή έχει την µορφή: 

Do 
 <εντολή> 
 [if <συνθήκη> then Exit  Do] 
 [if <συνθήκη> then Continue Do] 
 <εντολή> 
Loop [ While ή Until <συνθήκη> ] 
 

Εάν χρησιµοποιείται ή δήλωση While οι επαναλήψεις διαρκούν όσο η τιµή της λογικής συνθήκης 

είναι αληθής. Αντίθετα όταν χρησιµοποιείται ή δήλωση Until οι επαναλήψεις διακόπτονται όταν η 

τιµή της λογικής συνθήκης γίνει αληθής.  

 

Με την προαιρετική εντολή Exit Do διακόπτονται οι επαναλήψεις, ενώ µε την προαιρετική εντολή 

Continue Do δεν εκτελούνται οι επόµενες εντολές του τµήµατος του προγράµµατος που 

επαναλαµβάνεται. 

 

Η εντολή µπορεί επίσης να συνταχθεί χωρίς της χρήση λογικής συνθήκης στην αρχή ή το τέλος της 

εντολής. 

 
Do 
 <εντολή> 
 [if <συνθήκη> then Exit  Do] 
 [if <συνθήκη> then Continue Do] 
 <εντολή> 
Loop  
 

Στην περίπτωση αυτή οι επαναλήψεις θα πρέπει να διακόπτονται µε την εντολή Exit Do διαφορετικά 

οι επαναλήψεις θα συνεχίζουν διαρκώς. 

 

Οι διάφορες µορφές της εντολής Do Loop δίνουν µεγάλη ευελιξία στην σύνταξη τµηµάτων του 

προγράµµατος τα οποία επαναλαµβάνονται. Μπορούν να χρησιµοποιηθούν στην θέση της εντολής 

For, αλλά και για την επανάληψη τµηµάτων του προγράµµατος όταν δεν είναι γνωστός εκ των 

προτέρων ο αριθµός των επαναλήψεων.   

 

Τα παρακάτω παραδείγµατα είναι ισοδύναµα µε το παράδειγµα 4.9 όπου χρησιµοποιείται εντολή For 

για να εκτυπώσει του αριθµούς από ένα έως 5. 

 

 
    Public Sub main() 
        Dim i As Integer = 1      'Αρχική τιµή 
        Do While i <= 5           'Τελική τιµή 
            Console.WriteLine(i) 
            i += 1                'Βήµα 
        Loop 
    End Sub 

Παράδειγµα 4.19 Έλεγχος συνθήκης στην αρχή µε While 

 
    Public Sub main() 
        Dim i As Integer = 1      'Αρχική τιµή 
        Do Until i > 5            'Τελική τιµή 
            Console.WriteLine(i) 
            i += 1                'Βήµα 
        Loop 
    End Sub 

Παράδειγµα 4.20 Έλεγχος συνθήκης στην αρχή µε Until 

    Public Sub main() 
        Dim i As Integer = 1      'Αρχική τιµή 
        Do 
            Console.WriteLine(i) 
            i += 1                'Βήµα 
        Loop While i <= 5         'Τελική τιµή 
    End Sub 

Παράδειγµα 4.21 Έλεγχος συνθήκης στην τέλος µε While 

 
    Public Sub main() 
        Dim i As Integer = 1      'Αρχική τιµή 
        Do 
            Console.WriteLine(i) 
            i += 1                'Βήµα 
        Loop Until i > 5          'Τελική τιµή 
    End Sub 

Παράδειγµα 4.22 Έλεγχος συνθήκης στην τέλος µε Until 

 
    Public Sub main() 
        Dim i As Integer = 1      'Αρχική τιµή 
        Do 
            Console.WriteLine(i) 
            i += 1                'Βήµα 
            If i > 5 Then Exit Do 'Τελική τιµή 
        Loop 
    End Sub 

Παράδειγµα 4.23 Χρήση Do Loop χωρίς έλεγχο αρχικής ή τελικής συνθήκης 

 

Σε αρκετές περιπτώσεις ο αριθµός των επαναλήψεων δεν είναι εκ των προτέρων γνωστός. Στο 

παρακάτω παράδειγµα το πρόγραµµα διαβάζει διαδοχικά ακέραιους αριθµούς τους οποίους εισάγει ο 

χρήστης. Η εισαγωγή των αριθµών σταµατά όταν ο χρήστης πληκτρολογήσει τον αριθµό 0. Το 

πρόγραµµα υπολογίζει και εκτυπώνει τον αριθµό των αρνητικών αριθµών που δόθηκαν από τον 

χρήστη. 

 
    Public Sub main() 
        'Πρόγραµµα το οποίο µετρά τον αριθµό 
        'των αρνητικών αριθµών που εισάγει ο χρήστης 
        Dim counter As Integer = 0 'Μετρητής αρνητικών αριθµών 
        Dim i As Integer 'Αρχική τιµή 
        Do 
            Console.Write("∆ώστε ένα ακέραιο ή 0 για τέλος: ") 
            i = Console.ReadLine() 
            If i < 0 Then counter += 1 
        Loop While i <> 0 
        Console.WriteLine("∆ώσατε {0} αρνητικούς αριθµούς ", 
counter) 
    End Sub 

Παράδειγµα 4.24  

 

Το αποτέλεσµα της εκτέλεσης του προγράµµατος είναι: 
∆ώστε ένα ακέραιο ή 0 για τέλος: -2 
∆ώστε ένα ακέραιο ή 0 για τέλος: 4 
∆ώστε ένα ακέραιο ή 0 για τέλος: -9 
∆ώστε ένα ακέραιο ή 0 για τέλος: 6 
∆ώστε ένα ακέραιο ή 0 για τέλος: 0 
∆ώσατε 2 αρνητικούς αριθµούς 

4.7 Η εντολή επανάληψης While 

 

Η εντολή επανάληψης while είναι αντίστοιχη της εντολής Do While Loop. Η σύνταξη της είναι: 

 
While <συνθήκη> 
 <εντολή> 
 [if <συνθήκη> then Exit  While] 
 [if <συνθήκη> then Continue While] 
 <εντολή> 
End While  
 

και είναι αντίστοιχη της: 

 
Do While <συνθήκη> 
 <εντολή> 
 [if <συνθήκη> then Exit  Do] 
 [if <συνθήκη> then Continue Do] 
 <εντολή> 
Loop  
 

Στο παρακάτω παράδειγµα η εντολή επανάληψης While χρησιµοποιείται για να υπολογίσει και 

εκτυπώσει πόσα χρόνια χρειάζονται για τον διπλασιασµό ενός αρχικού κεφαλαίου το οποίο 

κατατίθεται στην τράπεζα µε σταθερό επιτόκιο και ανατοκισµό στο τέλος κάθε έτους.  

 
    Public Sub main() 
        'Πρόγραµµα το οποίο υπολογίζει τα χρόνια που 
        'χρειάζονται για το διπλασιασµό ενός κεφαλαίου 
        Dim arxikoKefalaio As Single 'To αρχικό κεφάλαιο 
        Dim kefalaio As Single 'To κεφάλαιο στο τέλος κάθε έτους 
        Dim epitokio As Single 'Το επιτόκιο 
        Console.Write("∆ώστε αρχικό κεφάλαιο: ") 
        arxikoKefalaio = Console.ReadLine() 
        Console.Write("∆ώστε επιτόκιο: ") 
        epitokio = Console.ReadLine() 
        Dim metritis As Integer = 0 'Μετρητής ετών 
        kefalaio = arxikoKefalaio 
        While kefalaio < (2 * arxikoKefalaio) 
            metritis += 1   'αύξηση έτους 
            'Πρόσθεση τόκου στο κεφάλαιο  
            kefalaio += kefalaio * epitokio / 100  
            Console.WriteLine( _  
             "Έτος {0} Κεφάλαιο :{1:N2}", metritis, kefalaio) 
        End While 
    End Sub 

Παράδειγµα 4.25  

 

Το αποτέλεσµα της εκτέλεσης του παραπάνω προγράµµατος µε αρχικό κεφάλαιο 1000 και επιτόκιο 

15 % είναι: 

 
∆ώστε αρχικό κεφάλαιο: 1000 
∆ώστε επιτόκιο: 15 
Έτος 1 Κεφάλαιο: 1.150,00 
Έτος 2 Κεφάλαιο: 1.322,50 
Έτος 3 Κεφάλαιο: 1.520,88 
Έτος 4 Κεφάλαιο: 1.749,01 
Έτος 5 Κεφάλαιο: 2.011,36 
Το κεφάλαιο διπλασιάζεται σε 5 έτη 

4.8 Η εντολή επανάληψης For Each 

 

Η εντολή επανάληψης For Each χρησιµοποιείται για να απαριθµήσει όλα τα µέλη ενός πίνακα ή 

άλλων συλλογών (Collections) δεδοµένων.  

 

Στο παρακάτω παράδειγµα η εντολή For Each χρησιµοποιείται για να εκτυπώσει τους χαρακτήρες µια 

αλφαριθµητικής µεταβλητής. Οι αλφαριθµητικές µεταβλητές για τις περισσότερες γλώσσες 

προγραµµατισµού  είναι πίνακες χαρακτήρων. 

 
    Public Sub main() 
        Dim onoma As String 
        Console.Write("∆ώστε ένα όνοµα: ") 
        onoma = Console.ReadLine() 
        Dim c As Char 
        Console.WriteLine("Οι χαρακτήρες του ονόµατος είναι:") 
        For Each c In onoma 
            Console.WriteLine(c) 
        Next 
    End Sub 
 

 
∆ώστε ένα όνοµα: Άρης 
Οι χαρακτήρες του ονόµατος είναι: 
Ά 
ρ 
η 
ς 
 

Περισσότερα παραδείγµατα χρήσης της εντολής αυτής θα δοθούν σε επόµενα κεφάλαια. 

 

4.9 Η εντολή goto 

 

Η εντολή goto αν και δεν είναι εντολή επανάληψης ή επιλογής, µπορεί να χρησιµοποιηθεί για την 

επιλογή ή επανάληψη υποπρογραµµάτων. Τα προγράµµατα που χρησιµοποιούν εντολές goto γίνονται 

µε δυσκολία κατανοητά και είναι ιδιαίτερα εύκολα να χαθεί ο έλεγχος της ροής του προγράµµατος. 

 

Η εντολή goto δίνει την εντολή στο πρόγραµµα να συνεχίσει την εκτέλεση του σε κάποιο άλλη 

γραµµή η σηµείο του προγράµµατος. Η εντολή goto παίρνει ως παράµετρο τον αριθµό γραµµής ή το 

όνοµα ετικέτας (label) του προγράµµατος στο οποίο θα γίνει η µετάβαση. 

 

Ετικέτες µπορούµε να ορίσουµε σε οποιοδήποτε σηµείο του προγράµµατος µε την χρήση του 

ονόµατος της ετικέτας και τον χαρακτήρα «:». 

 

Το παρακάτω πρόγραµµα εκτυπώνει τους αριθµούς από 1 έως 5 στην οθόνη. 

 
    Public Sub main() 
        Dim i As Integer = 1 
arxi: 
        Console.Writeline(i) 
        i += 1 
        If i <= 5 Then GoTo arxi 
    End Sub 
 

Η χρήση της εντολής αυτής δεν ενδείκνυται παρά µόνο σε ελάχιστες περιπτώσεις όπως όταν 

χρειάζεται να διακοπούν ένθετες εντολές επανάληψης και ο έλεγχος να δοθεί σε συγκεκριµένο σηµείο 

του προγράµµατος η για να αντιµετωπισθούν λάθη κατά την εκτέλεση του προγράµµατος.  

4.10 Ασκήσεις κεφαλαίου 4 

 

Άσκηση 4.1 

∆ηµιουργήστε εφαρµογή lab_4_ask_1 η οποία ζητά από τον χρήστη να εισάγει δύο 

πραγµατικούς αριθµούς x και y και στην συνέχεια: 

• εξετάζει και εκτυπώνει εάν ο αριθµός x ανήκει στο διάστηµα [1 .. 10]  

• εκτυπώνει τον µεγαλύτερο αριθµό. 

 

Άσκηση 4.2 

∆ηµιουργήστε εφαρµογή lab_4_ask_2 η οποία ζητά από τον χρήστη να εισάγει έναν ακέραιο 

από 0 έως 9999 και στην συνέχεια κάνοντας χρήση της εντολής επιλογής Select εκτυπώνει τον 

αριθµό των ψηφίων του αριθµού (µονοψήφιος, διψήφιος, τριψήφιος ή τετραψήφιος). 

 

Άσκηση 4.3 

∆ηµιουργήστε εφαρµογή lab_4_ask_3 η οποία ζητά από τον χρήστη να εισάγει τα συνολικό 

εισόδηµα του και στην συνέχεια υπολογίζει και εκτυπώνει το φόρο τον οποίο θα πρέπει να 

πληρώσει. 

  

Ο φορολογικός συντελεστής είναι: 

• Για τα πρώτα 8.400 Ευρώ  0% 

• Για τα επόµενα 5.000 Ευρώ 15% 

• Για τα επόµενα 10.000 Ευρώ 30% 

• Για επιπλέον εισόδηµα 40% 

 

Πχ.  

Για εισόδηµα 20.000 Έυρώ ο φόρος είναι 8400*0%+5000*15%+(20000-13400)*30%=2730  

 

Άσκηση 4.4 

∆ηµιουργήστε εφαρµογή lab_4_ask_4 η οποία ζητά από τον χρήστη να εισάγει τρεις 

πραγµατικούς αριθµούς και την συνέχεια τους εκτυπώνει στην οθόνη σε αύξουσα σειρά.  

 

Υπόδειξη: Χρησιµοποιήστε ένθετες εντολές If για να ελέγξετε όλες τις περιπτώσεις 

 

Άσκηση 4.5 

∆ηµιουργήστε εφαρµογή lab_4_ask_5 η οποία κάνοντας χρήση εντολών επανάληψης 

εκτυπώνει τις αντιστοιχία σε Ευρώ των 100,200,300 έως 1000 δραχµών 

 

Άσκηση 4.6 

∆ηµιουργήστε εφαρµογή lab_4_ask_6 η οποία υπολογίζει και εκτυπώνει το άθροισµα των  

περιττών ακεραίων από 1 έως 199. 

 

Άσκηση 4.7 

∆ηµιουργήστε εφαρµογή lab_4_ask_7 η οποία ζητά από τον χρήστη να εισάγει ένα ακέραιο 

αριθµό Ν και στην συνέχεια εκτυπώνει τις τιµές των παρακάτω παραστάσεων: 

• 1+2+3+..+Ν 

• 1
2
+2

2
+3

2
+..+Ν

2
 

• 1*2*3* .. *Ν 

 

Άσκηση 4.8 

∆ηµιουργήστε εφαρµογή lab_4_ask_8 η οποία µε την χρήση ένθετων εντολών επανάληψης 

εµφανίζει στην οθόνη το παρακάτω σχήµα: 

 

1 2 3 4 5 

2 3 4 5 6 

3 4 5 6 7 

4 5 6 7 8 

Άσκηση 4.9 

∆ηµιουργήστε εφαρµογή lab_4_ask_9 η οποία ζητά από τον χρήστη να εισάγει διαδοχικά 30 

ακεραίους αριθµούς και στην συνέχεια εκτυπώνει το άθροισµα τους, τον µέσο όρο τους, τον 

µεγαλύτερο αριθµό και το πλήθος των αρνητικών αριθµών.
 

 

Άσκηση 4.10 

∆ηµιουργήστε εφαρµογή lab_4_ask_10 η οποία κάνοντας χρήση εντολής επανάληψης Do 

Loop, υπολογίζει και εκτυπώνει τα ηµίτονα και συνηµίτονα των γωνιών από 10 έως 20 µοίρες 

µε βήµα 0.1  

 

Άσκηση 4.11 

∆ηµιουργήστε εφαρµογή lab_4_ask_11 η οποία διαβάζει διαδοχικά ακεραίους αριθµούς τους 

οποίους εισάγει ο χρήστης. Η εισαγωγή των αριθµών σταµατά όταν ο χρήστης 

πληκτρολογήσει τον αριθµό 0 και στην συνέχεια το πρόγραµµα εκτυπώνει στην οθόνη το 

άθροισµα τους, τον µικρότερο αριθµό, το πλήθος των περιττών αριθµών και το πλήθος των 

αριθµών οι οποίοι είναι µεγαλύτεροι του 20. 

 

Άσκηση 4.12 

∆ηµιουργήστε εφαρµογή lab_4_ask12 η υπολογίζει την µικρότερη τιµή του Ν για την οποία η 

παρακάτω παράσταση είναι µεγαλύτερη του 65000. 

   1
2
 + 2

2
 + 3

2
 + .. + Ν

2
 

 

Άσκηση 4.13 

Σφαίρα αφήνεται να πέσει από αρχικό ύψος x. Σε κάθε αναπήδηση η σφαίρα χάνει το 10% του 

προηγούµενου ύψους. Η σφαίρα σταµατά την κίνηση της όταν το ύψος της γίνεται <= 0.01 . 

Γράψτε εφαρµογή lab_4_ask_13 ζητά από τον χρήστη να δώσει το αρχικό ύψος x και στην 

συνέχεια υπολογίζει και εκτυπώνει τον αριθµό των αναπηδήσεων που θα κάνει η σφαίρα. 

 

Άσκηση 4.14 

∆ηµιουργήστε εφαρµογή lab_4_ask_14 η οποία ζητά από τον χρήστη να εισάγει δύο αριθµούς 

x και y και στην συνέχεια υπολογίζει το υπόλοιπο της διαίρεσης  x / y µε την χρήση 

διαδοχικών αφαιρέσεων του y από το x.  

 

Άσκηση 4.15 

∆ηµιουργήστε εφαρµογή lab_4_ask_15 η οποία εκτυπώνει τους Πυθαγόρειους αριθµούς από 

το 1 µέχρι το 100. Τρεις ακέραιοι αριθµοί α, β, γ λέγονται Πυθαγόρειοι αν ισχύει : α
2
+β

2
=γ

2 

 
Υπόδειξη: Χρησιµοποιήστε τρεις ένθετες εντολές επανάληψης µε µεταβλητές a,b και c. 

 

Άσκηση 4.16 

∆ηµιουργήστε εφαρµογή lab_4_ask_16 η οποία ζητά από τον χρήστη να εισάγει έναν ακέραιο 

αριθµό, εκτυπώνει τους διαιρέτες του αριθµού και εκτυπώνει εάν ο αριθµός αυτός είναι 

πρώτος ή όχι. 

 

Υπόδειξη: Πρώτος αριθµός είναι ο αριθµός ο οποίος διαιρείται ακριβώς µόνο µε την µονάδα 

και τον εαυτό του. Χρησιµοποιήστε εντολή επανάληψης για να εκτυπώσετε και να µετρήσετε 

τους διαιρέτες του αριθµού. Ανάλογα µε το αποτέλεσµα της µέτρησης εκτυπώστε εάν ο 

αριθµός είναι πρώτος ή όχι.    

5 Πίνακες δεδοµένων 

Η Visual Basic υποστηρίζει µονοδιάστατους, δισδιάστατους ή και πίνακες περισσοτέρων διαστάσεων. 

Τα στοιχεία ενός πίνακα ανήκουν στον ίδιο τύπο δεδοµένων.  

5.1 Μονοδιάστατοι πίνακες 

 

Ο ορισµός ενός µονοδιάστατου πίνακα τύπου δεδοµένων  γίνεται µε την δήλωση: 

 
Dim <όνοµα πίνακα> (<πλήθος στοιχείων – 1 >) as <τύπος δεδοµένων> 
 

π.χ.  η δήλωση 

 
Dim m(6) As Integer 
    

ορίζει ένα µονοδιάστατο πίνακα 7 ακεραίων µε δείκτες στοιχείων του πίνακα από 0 έως 6.  

 

Η αναφορά σε ένα στοιχείο του πίνακα (για αποθήκευση ή ανάγνωση της τιµής του) γίνεται µε την 

χρήση του ονόµατος του πίνακα ακολουθούµενο από τον δείκτη του στοιχείου εντός παρενθέσεων. 

 

Π.χ. 

 
m(0) = 45 ' Απόδοση τιµής στο πρώτο στοιχείο του πίνακα 
 
Console.WriteLine(m(0)) ' Εκτύπωση της τιµής του πρώτου στοιχείου 
 

Το πρώτο στοιχείο ενός µονοδιάστατου πίνακα έχει πάντα τον δείκτη 0 και το τελευταίο στοιχείο του 

πίνακα έχει ως δείκτη το πλήθος των στοιχείων ελαττωµένο κατά µία µονάδα. 

 

Ο ορισµός ενός πίνακα µπορεί να γίνει και µε τον ταυτόχρονο καθορισµό των τιµών των στοιχείων 

του πίνακα. Στην περίπτωση αυτή παραλείπεται ο ορισµός του πλήθους των στοιχείων του πίνακα. 

 

Π.χ. 

 
Dim m() As Integer = {10, 11, 45, 23, 45, 8} 
 

Η επεξεργασία των στοιχείων ενός πίνακα γίνεται συνήθως µε την χρήση εντολών επανάληψης. 

Εντολές επανάληψης χρησιµοποιούνται επίσης για την εισαγωγή των στοιχείων από το πληκτρολόγιο, 

στην περίπτωση κατά την οποία τα στοιχεία του πίνακα δεν ορίζονται µε εντολές του προγράµµατος 

αλλά είναι δυναµικά και εισάγονται από το πληκτρολόγιο κάθε φορά που γίνεται η εκτέλεση του 

προγράµµατος. 

 

Στο παρακάτω παράδειγµα 5.1 ορίζεται ένας πίνακας 6 ακεραίων , γίνεται η εισαγωγή των στοιχείων 

του από το πληκτρολόγιο, εκτυπώνονται τα στοιχεία του πίνακα και υπολογίζεται και εκτυπώνεται  ο 

µέσος όρος, το µικρότερο στοιχείο, το µεγαλύτερο στοιχείο και το πλήθος των µηδενικών στοιχείων 

του πίνακα. 

  
    Public Sub main() 
        'Πρόγραµµα το οποίο ορίζει πίνακα 6 ακεραίων, 
        'διαβάζει τα στοιχεία από το πληκτρολόγιο, 
        'εκτυπώνει τον πίνακα και υπολογίζει το µέσο όρο, 
        'το µικρότερο , το µεγαλύτερο και το πλήθος των 
        'µηδενικών στοιχείων του πίνακα 
 
        Dim m(6) As Integer 'Ορισµός πίνακα 
        Dim i As Integer '∆είκτης επανάληψης 
        Dim sum As Integer 'Άθροισµα στοιχείων πίνακα 
        Dim min As Integer 'Μικρότερο στοιχείο πίνακα 
        Dim max As Integer 'Μεγαλύτερο στοιχείο πίνακα 

        Dim n As Integer 'Πλήθος µηδενικών στοιχείων 
 
        'Εισαγωγή στοιχείων από το πληκτρολόγιο 
        Console.WriteLine("Εισαγωγή στοιχείων πίνακα") 
        Console.WriteLine("-----------------------------") 
        For i = 0 To 5 
            Console.Write("∆ώστε το στοιχείο  m({0}): ", i) 
            m(i) = Console.ReadLine() 
        Next 
 
        'Εκτύπωση στοιχείων πίνακα 
        Console.WriteLine() 
        Console.WriteLine("Τα στοιχεία του πίνακα είναι:") 
        Console.WriteLine("-----------------------------") 
        For i = 0 To 5 
            Console.WriteLine("  m({0})= {1,10} ", i, m(i)) 
        Next 
 
        Console.WriteLine() 
        Console.WriteLine("Επεξεργασία στοιχείων πίνακα") 
        Console.WriteLine("-----------------------------")        
 
       'Έυρεση µέσου όρου 
        sum = 0 
        For i = 0 To 5 
            sum += m(i) 
        Next 
        Console.WriteLine("Μέσος όρος         = {0,8:N2}", sum / 6) 
 
        'Έυρεση µικρότερου στοιχείου 
        min = m(0) 
        For i = 0 To 5 
            If m(i) < min Then min = m(i) 
        Next 
        Console.WriteLine("Μικρότερο στοιχείο = {0,8}", min) 
 
        'Έυρεση µεγαλύτερου στοιχείου 
        max = m(0) 
        For i = 0 To 5 
            If m(i) > max Then max = m(i) 
        Next 
        Console.WriteLine("Μεγαλύτερο στοιχείο= {0,8}", max) 
 
        'Πλήθος µηδενικών στοιχείων 
        n = 0 
        For i = 0 To 5 
            If m(i) = 0 Then n += 1 
        Next 
        Console.WriteLine("Πλήθος µηδενικών   = {0,8}", n) 
 
    End Sub 
 

Παράδειγµα 5.1 

 

Το αποτέλεσµα της εκτέλεσης του παραδείγµατος 5.1 είναι το παρακάτω: 

 
Εισαγωγή στοιχείων του πίνακα 
----------------------------- 
∆ώστε το στοιχείο  m(0): 9 
∆ώστε το στοιχείο  m(1): 0 
∆ώστε το στοιχείο  m(2): -23 

∆ώστε το στοιχείο  m(3): 12 
∆ώστε το στοιχείο  m(4): 0 
∆ώστε το στοιχείο  m(5): 45 
 
Τα στοιχεία του πίνακα είναι: 
----------------------------- 
  m(0)=          9 
  m(1)=          0 
  m(2)=        -23 
  m(3)=         12 
  m(4)=          0 
  m(5)=         45 
 
Επεξεργασία στοιχείων πίνακα 
----------------------------- 
Μέσος όρος         =     7,17 
Μικρότερο στοιχείο =      -23 
Μεγαλύτερο στοιχείο=       45 
Πλήθος µηδενικών   =        2 

5.2 ∆ισδιάστατοι πίνακες 

 

Ο ορισµός ενός δισδιάστατου πίνακα τύπου δεδοµένων  γίνεται µε την δήλωση: 

 
Dim <όνοµα πίνακα> (<γραµµές – 1 > <στήλες –1>) as <τύπος δεδοµένων> 
 

π.χ.  η δήλωση 

 
Dim m(2, 2) As Double 
    

ορίζει ένα δισδιάστατο πίνακα 3x3 ακεραίων.  

 

Πίνακες περισσοτέρων διαστάσεων ορίζονται µε την δήλωση των δεικτών της τρίτης, της τέταρτης 

διάστασης κ.ο.κ.  

 

Π.χ. η δήλωση: 

 
Dim m(2, 2, 2) As Double 
    

ορίζει ένα τρισδιάστατο πίνακα 3x3x3 ακεραίων  

 

Η αναφορά στα στοιχεία ενός δισδιάστατου του πίνακα γίνεται µε την χρήση δύο δεικτών, ο πρώτος 

καθορίζει την γραµµή και ο δεύτερος την στήλη του πίνακα. Η αρίθµηση των γραµµών και των 

στηλών ξεκινά µε την τιµή 0. 

 

Π.χ. 

 
m(0, 0) = 45 ' Απόδοση τιµής στο στοιχείο m00 του πίνακα 
 
Console.WriteLine(m(0, 0)) ' Εκτύπωση της τιµής του στοιχείου m00 

 

Η απόδοση τιµών στα στοιχεία ενός δισδιάστατου πίνακα µπορεί να γίνει ταυτόχρονα µε τον ορισµό 

του όπως φαίνεται στο παρακάτω παράδειγµα. 

 

Π.χ. 

 
        Dim m(,) As Double = { {1, 12, 3}, _ 
                               {4,  5, 6}, _ 
                               {7, 18, 9}} 

Στο παρακάτω παράδειγµα 5.2 ορίζεται ένας πίνακας 3x3 ακεραίων  και όπως και στο παράδειγµα 5.1 

για µονοδιάστατους πίνακες,  γίνεται η εισαγωγή των στοιχείων του από το πληκτρολόγιο, 

εκτυπώνονται τα στοιχεία του πίνακα και υπολογίζεται και εκτυπώνεται  ο µέσος όρος, το µικρότερο 

στοιχείο, το µεγαλύτερο στοιχείο και το πλήθος των µηδενικών στοιχείων του πίνακα. 

 

 
    Public Sub main() 
        'Πρόγραµµα το οποίο ορίζει πίνακα 3X3 πραγµατικών, 
        'διαβάζει τα στοιχεία από το πληκτρολόγιο, 
        'εκτυπώνει τον πίνακα και υπολογίζει το µέσο όρο, 
        'το µικρότερο, το µεγαλύτερο και το πλήθος των 
        'µηδενικών στοιχείων του πίνακα 
 
        Dim m(2, 2) As Double 'Ορισµός πίνακα 
        Dim i As Integer '∆είκτης επανάληψης γραµµών 
        Dim j As Integer '∆είκτης επανάληψης στηλών 
        Dim sum As Double 'Άθροισµα στοιχείων πίνακα 
        Dim min As Double 'Μικρότερο στοιχείο πίνακα 
        Dim max As Double 'Μεγαλύτερο στοιχείο πίνακα 
        Dim n As Integer 'Πλήθος µηδενικών στοιχείων 
 
        'Εισαγωγή στοιχείων από το πληκτρολόγιο 
        Console.WriteLine("Εισαγωγή στοιχείων πίνακα") 
        Console.WriteLine("-----------------------------") 
        For i = 0 To 2 
            For j = 0 To 2 
                Console.Write("∆ώστε το στοιχείο m({0},{1}): ", i,j) 
                m(i, j) = Console.ReadLine() 
            Next 
        Next 
 
        'Εκτύπωση στοιχείων πίνακα 
        Console.WriteLine() 
        Console.WriteLine("Τα στοιχεία του πίνακα είναι:") 
        Console.WriteLine("-----------------------------") 
        For i = 0 To 2 
            For j = 0 To 2 
                Console.Write("{0,8:N2} ", m(i, j)) 
            Next 
            Console.WriteLine() 
        Next 
 
        Console.WriteLine() 
        Console.WriteLine("Επεξεργασία στοιχείων πίνακα") 
        Console.WriteLine("-----------------------------") 
 
        'Έυρεση µέσου όρου 
        sum = 0 
        For i = 0 To 2 
            For j = 0 To 2 
                sum += m(i, j) 
            Next 
        Next 
        Console.WriteLine("Μέσος όρος         = {0,8:N2}", sum / 9) 
 
        'Έυρεση µικρότερου στοιχείου 
        min = m(0, 0) 
        For i = 0 To 2 
            For j = 0 To 2 
                If m(i, j) < min Then min = m(i, j) 
            Next 

        Next 
        Console.WriteLine("Μικρότερο στοιχείο = {0,8:N2}", min) 
 
        'Έυρεση µεγαλύτερου στοιχείου 
        max = m(0, 0) 
        For i = 0 To 2 
            For j = 0 To 2 
                If m(i, j) > max Then max = m(i, j) 
            Next 
        Next 
        Console.WriteLine("Μεγαλύτερο στοιχείο= {0,8:N2}", max) 
 
        'Πλήθος µηδενικών στοιχείων 
        n = 0 
        For i = 0 To 2 
            For j = 0 To 2 
                If m(i, j) = 0 Then n += 1 
            Next 
        Next 
        Console.WriteLine("Πλήθος µηδενικών   = {0,8}", n) 
 
    End Sub 
 

Παράδειγµα 5.2 

 

Το αποτέλεσµα της εκτέλεσης του παραπάνω προγράµµατος είναι: 

 
Εισαγωγή στοιχείων πίνακα 
----------------------------- 
∆ώστε το στοιχείο  m(0,0): 1,23 
∆ώστε το στοιχείο  m(0,1): -4,876 
∆ώστε το στοιχείο  m(0,2): 2 
∆ώστε το στοιχείο  m(1,0): -98,2 
∆ώστε το στοιχείο  m(1,1): 1 
∆ώστε το στοιχείο  m(1,2): 0 
∆ώστε το στοιχείο  m(2,0): 2,4 
∆ώστε το στοιχείο  m(2,1): 45 
∆ώστε το στοιχείο  m(2,2): 0 
 
Τα στοιχεία του πίνακα είναι: 
----------------------------- 
    1,23    -4,88     2,00 
  -98,20     1,00     0,00 
    2,40    45,00     0,00 
 
Επεξεργασία στοιχείων πίνακα 
----------------------------- 
Μέσος όρος         =    -5,72 
Μικρότερο στοιχείο =   -98,20 
Μεγαλύτερο στοιχείο=    45,00 
Πλήθος µηδενικών   =        2 
 

Συχνά είναι αναγκαίος ο ορισµός και η απόδοση τιµών σε ένα πίνακα µε την χρήση κάποιων σχέσεων 

ή ιδιοτήτων. Στο παρακάτω παράδειγµα δηµιουργείται ένας τριγωνικός πίνακας 8x8 και εκτυπώνεται. 

Το πρόγραµµα δίνει την τιµή 1 στο στοιχεία του πίνακα τα οποία βρίσκονται πάνω ή µετά την κύρια 

διαγώνιο του πίνακα και την τιµή 0 στα υπόλοιπα.  

    Public Sub main() 
        'Πρόγραµµα το οποίο δηµιουργεί ένα τριγωνικό πίνακα 
        'και τον εκτυπώνει 
        Dim m(7, 7) As Integer 'Ορισµός πίνακα 8Χ8 
        Dim i, j As Integer '∆είκτες πίνακα 
        'Απόδοση τιµής στα στοιχεία του πίνακα 
        For i = 0 To 7 
            For j = 0 To 7 
                'Έλεγχος εάν το στοιχείο είναι 
                'πάνω η µετά την κύρια διαγώνιο 
                If j >= i Then 
                    m(i, j) = 1 
                Else 
                    m(i, j) = 0 
                End If 
            Next 
        Next 
        'Εκτύπωση πίνακα 
        For i = 0 To 7 
            For j = 0 To 7 
                Console.Write("{0,2}", m(i, j)) 
            Next 
            Console.WriteLine() 
        Next 
    End Sub 
 

 
 1 1 1 1 1 1 1 1 
 0 1 1 1 1 1 1 1 
 0 0 1 1 1 1 1 1 
 0 0 0 1 1 1 1 1 
 0 0 0 0 1 1 1 1 
 0 0 0 0 0 1 1 1 
 0 0 0 0 0 0 1 1 
 0 0 0 0 0 0 0 1 

5.3 Ταξινόµηση, και αναζήτηση στοιχείων πινάκων 

 

Η Visual Basic παρέχει διάφορες χρήσιµες µεθόδους και συναρτήσεις για τον χειρισµό, και 

επεξεργασία πινάκων.  

 

Η µέθοδος Length επιστρέφει το πλήθος των στοιχείων ενός πίνακα (µίας η περισσοτέρων 

διαστάσεων).  

 

Οι µέθοδοι GetLowerBound και GetUpperBound επιστρέφουν τον αρχικό και τον τελικό δείκτη της 

διάστασης η οποία δίνεται ως παράµετρος εισόδου  ( 0 για την πρώτη διάσταση, 1 για την δεύτερη 

διάσταση κ.ο.κ.). 

 

Π.χ. αν ο m είναι ένας δισδιάστατος πίνακας, η παρακάτω εντολή εκτυπώνει τον τελικό δείκτη της 

δεύτερης διάστασης του πίνακα (των στηλών). 

 
Console.WriteLine( m.GetUpperBound(1)) 
 

Εναλλακτικά, αντί των µεθόδων GetLowerBound και GetUpperBound µπορούν να χρησιµοποιηθούν 

οι συναρτήσεις LBound and Ubound  αντίστοιχα. Οι συναρτήσεις αυτές δέχονται ως πρώτη 

παράµετρο τον πίνακα και ως δεύτερη παράµετρο την διάσταση (1 για την πρώτη διάσταση, 2 για την 

δεύτερη διάσταση κ.ο.κ.). 

 

Πχ. η παρακάτω εντολή εκτυπώνει τον τελικό δείκτη της δεύτερης διάστασης του δισδιάστατου 

πίνακα m. 

Console.WriteLine( Ubound(m,2))  
 

Ειδικά για τους µονοδιάστατους πίνακες υπάρχουν και οι µέθοδοι IndexOf και Sort της κλάσης 

αντικειµένων Array.  

 

Η µέθοδος IndexOf επιστρέφει τον δείκτη ενός στοιχείου του πίνακα ή τον αριθµό  –1 εάν το στοιχείο 

δεν ανήκει στον πίνακα. Π.χ. εάν m είναι µονοδιάστατος πίνακας  ακεραίων, η παρακάτω εντολή 

εκτυπώνει τον δείκτη του στοιχείου µε τιµή 2 

 
Console.WriteLine(Array.IndexOf(m, 2)) 
 

Η µέθοδος Sort ταξινοµεί τα στοιχεία ενός µονοδιάστατου πίνακα.  Π.χ. εάν m είναι µονοδιάστατος 

πίνακας, η παρακάτω εντολή ταξινοµεί τα στοιχεία του. 

 
Array.Sort(m) 
 

Στο παρακάτω παράδειγµα δίνονται παράδειγµα χρήσης των µεθόδων αυτών για ένα µονοδιάστατο 

πίνακα m1 και για ένα δισδιάστατο πίνακα  m2. 

  
    Public Sub main() 
 
        'Παράδειγµα µεθόδων µονοδιάστατου πίνακα 
        Dim m1() As Double = {12, 22.3, 4.1, 6.7, 9, 12} 
        Dim i As Integer 
 
        Console.WriteLine("Πλήθος στοιχείων m1 = {0}", m1.Length) 
        Console.WriteLine( _  
          "Άρχικός δείκτης στοιχείων m1 = {0}", m1.GetLowerBound(0)) 
        Console.WriteLine( _  
          "Τελικός δείκτης στοιχείων m1 = {0}", m1.GetUpperBound(0)) 
        Console.WriteLine("") 
 
        'Εύρεση δείκτη στοιχείου πίνακα 
        Console.WriteLine( _  
          "Το 4.1 είναι στην θέση = {0} ", Array.IndexOf(m1, 4.1)) 
        Console.WriteLine("") 
 
        'Tαξινόµιση µονοδιάστατου πίνακα 
        Console.WriteLine("Τα στοιχεία πριν την ταξινόµιση είναι:") 
        For i = 0 To m1.GetUpperBound(0) 
            Console.Write("{0,6:N2} ", m1(i)) 
        Next 
        Console.WriteLine("") 
        Array.Sort(m1) 
        Console.WriteLine("Τα στοιχεία µετά την ταξινόµιση είναι:") 
        For i = 0 To m1.GetUpperBound(0) 
            Console.Write("{0,6:N2} ", m1(i)) 
        Next 
        Console.WriteLine("") 
        Console.WriteLine("") 
 
        'Παράδειγµα µεθόδων δισδιάστατου πίνακα 
        Dim m2(,) As Double = {{1, 12}, _ 
                               {4, 51}, _ 
                               {7, 18}} 
 
        Console.WriteLine( _  
         "∆είκτης πρώτης γραµµής m2 = {0}", m2.GetLowerBound(0)) 
        Console.WriteLine( _  
         "∆είκτης τελευταίας γραµµής m2 = {0}", m2.GetUpperBound(0)) 

        Console.WriteLine( _  
          "∆είκτης πρώτης στήλης m2 = {0}", m2.GetLowerBound(1)) 
        Console.WriteLine( _  
          "∆είκτης τελευταίας στήλης m2 = {0}", m2.GetUpperBound(1)) 
 
        Console.WriteLine("Πλήθος στοιχείων m2 = {0}", m2.Length) 
 
    End Sub 

Παράδειγµα 5.3 

 

Το αποτέλεσµα της εκτέλεσης του παραπάνω προγράµµατος είναι: 

 
Πλήθος στοιχείων m1 = 6 
Άρχικός δείκτης στοιχείων m1 = 0 
Τελικός δείκτης στοιχείων m1 = 5 
 
Το 4.1 είναι στην θέση = 2 
 
Τα στοιχεία πριν την ταξινόµιση είναι: 
 12,00  22,30   4,10   6,70   9,00  12,00 
Τα στοιχεία πριν την ταξινόµιση είναι: 
  4,10   6,70   9,00  12,00  12,00  22,30 
  
∆είκτης πρώτης γραµµής m2 = 0 
∆είκτης τελευταίας γραµµής m2 = 2 
∆είκτης πρώτης στήλης m2 = 0 
∆είκτης τελευταίας στήλης m2 = 1 
Πλήθος στοιχείων m2 = 6 
 

5.4 ∆υναµική τροποποίηση του µεγέθους ενός πίνακα 

 

Το πλήθος των στοιχείων ενός πίνακα είναι σταθερό. Εάν όµως χρειάζεται να τροποποιηθεί δυναµικά 

(κατά την εκτέλεση ενός προγράµµατος) το πλήθος των στοιχείων ενός πίνακα, τότε χρησιµοποιείται 

η  δήλωση Redim ή η δήλωση Redim Preserve. Με την δήλωση Redim Preserve οι τιµές των 

στοιχείων του πίνακα διατηρούνται µετά την  αλλαγή του πλήθους των στοιχείων. 

 

Στο παρακάτω παράδειγµα ο χρήστης ορίζει δυναµικά το πλήθος των στοιχείων ενός πίνακα και στην 

συνέχεια εισάγει τα στοιχεία του πίνακα. 

 
    Public Sub main() 
        'Παράδειγµα δυναµικής τροποποίησης του µεγέθους ενός πίνακα 
        Dim m() As Double 
        Dim i, n As Integer 
        Console.Write("∆ώσε πλήθος στοιχείων: ") 
        n = Console.ReadLine() 
 
        'Αλλαγή µεγέθους πίνακα 
        ReDim Preserve m(n - 1) 
 
        'Εισαγωγή στοιχείων 
        For i = 0 To m.GetUpperBound(0) 
            Console.Write("∆ώσε στοιχείο m({0}): ", i) 
            m(i) = Console.ReadLine() 
        Next 
 
        'Εκτύπωση των στοιχείων µε την εντολή for each 
        Console.Write("Τα στοιχεία που δώσατε είναι:") 
        For Each d As Double In m 
            Console.Write("{0,6:N2} ", d) 

        Next 
        Console.WriteLine() 
 
    End Sub 

Παράδειγµα 5.4 

 

Το αποτέλεσµα της εκτέλεσης του παραπάνω προγράµµατος είναι: 

 
∆ώσε πλήθος στοιχείων: 3 
∆ώσε στοιχείο m(0): 1 
∆ώσε στοιχείο m(1): 2,345 
∆ώσε στοιχείο m(2): 1,23 
Τα στοιχεία που δώσατε είναι:  1,00   2,35   1,23 

5.5 Ασκήσεις Κεφαλαίου 5 

 

Άσκηση 5.1 

∆ηµιουργήστε εφαρµογή lab_5_ask_1 η οποία ορίζει µονοδιάστατο πίνακα 8 πραγµατικών 

αριθµών, ζητά από τον χρήστη να εισάγει τα στοιχεία του πίνακα, και εκτυπώνει: 

• τον πίνακα 

• το άθροισµα,  τον µέσο όρο, το µικρότερο και το µεγαλύτερο στοιχείο του πίνακα 

• το πλήθος των αρνητικών του στοιχείων του πίνακα 

 

Άσκηση 5.2 

∆ηµιουργήστε εφαρµογή lab_5_ask_2 η οποία ορίζει µονοδιάστατο πίνακα 6 ακεραίων, ζητά 

από τον χρήση να εισάγει τα στοιχεία του πίνακα, ταξινοµεί τον πίνακα, εκτυπώνει τον 

ταξινοµηµένο πίνακα, και στην περίπτωση που ο πίνακας περιέχει τον αριθµό 25 εκτυπώνει 

την θέση του (τον δείκτη του στοιχείου). 

 

Άσκηση 5.3 

∆ηµιουργήστε εφαρµογή lab_5_ask_3 ζητά από τον χρήστη να εισάγει σε πίνακα  

πραγµατικών αριθµών τις ελάχιστες θερµοκρασίες 7 διαδοχικών ηµερών  µιας εβδοµάδος και 

εκτυπώνει: 

• Το µέσο όρο της ελάχιστης θερµοκρασίας 

• Το πλήθος των ηµερών µε ελάχιστη θερµοκρασία άνω των 4 βαθµών 

• Το πλήθος των ηµερών µε ελάχιστη θερµοκρασία κάτω του µηδενός 

• Τον µέσο όρο της ελάχιστης θερµοκρασίας των ηµερών µε θερµοκρασία κάτω του 

µηδενός 

 

Άσκηση 5.4 

∆ηµιουργήστε εφαρµογή lab_5_ask_4 η οποία ορίζει µονοδιάστατο πίνακα 7 χαρακτήρων, 

ζητά από τον χρήση να εισάγει τα στοιχεία του πίνακα, εκτυπώνει τον πίνακα  και στην 

συνέχεια εξετάζει εάν όλοι οι χαρακτήρες του  πίνακα είναι κεφαλαίοι λατινικοί χαρακτήρες. 

 

Υπόδειξη: Μετρήστε πόσους κεφαλαίους λατινικούς χαρακτήρες περιέχει ο πίνακας. 

 

Άσκηση 5.5 

∆ηµιουργήστε εφαρµογή lab_5_ask_5 η οποία ορίζει έναν πίνακα πραγµατικών αριθµών 2Χ3, 

ζητά από τον χρήστη να εισάγει τα στοιχεία του πίνακα και στην συνέχεια εκτυπώνει: 

• τον πίνακα 

• το µικρότερο, το µεγαλύτερο και τον µέσο όρο των στοιχείων του πίνακα 

 

Άσκηση 5.6 

∆ηµιουργήστε εφαρµογή lab_5_ask_6 η οποία αποθηκεύει µε την χρήση ένθετων εντολών 

επανάληψης σε πίνακα 5Χ5 ακεραίων τις παρακάτω τιµές: 
     1  2  3  4  5 

     2  4  6  8 10 

     3  6  9 12 15 

     4  8 12 16 20 

     5 10 15 20 25  

 και εκτυπώνει τον πίνακα στην οθόνη του Η/Υ 

 

Άσκηση 5.7 

∆ηµιουργήστε εφαρµογή lab_5_ask_7 η οποία ορίζει έναν δισδιάστατο πίνακα ακεραίων 4Χ4, 

ζητά από τον χρήστη να εισάγει τα στοιχεία του πίνακα, εξετάζει εάν τα στοιχεία του πίνακα 

είναι όλα αρνητικοί αριθµοί και εκτυπώνει µε την χρήση εντολών επανάληψης τα αθροίσµατα 

των γραµµών του πίνακα 

Άσκηση 5.8 

∆ηµιουργήστε εφαρµογή lab_5_ask_8 η οποία ορίζει ένα τετραγωνικό δισδιάστατο πίνακα 

NxN (αριθµός γραµµών ίσος µε τον αριθµό των στηλών του πίνακα) χαρακτήρων, δίνει την 

τιµή ‘*’ στα στοιχεία της κύριας και δευτερεύουσας διαγωνίου του πίνακα, την τιµή ‘ ‘ (κενός 

χαρακτήρας) στα υπόλοιπα στοιχεία του πίνακα και εκτυπώνει τον πίνακα στην οθόνη. Η 

διάσταση N του πίνακα δίνεται από τον χρήστη και το µέγεθος του πίνακα καθορίζεται 

δυναµικά µε την δήλωση ReDim. 

6 Παραθυρικές εφαρµογές 

6.1 ∆ηµιουργία παραθυρικών εφαρµογών  

 

Η δηµιουργία µιας παραθυρικής εφαρµογής γίνεται µε την επιλογή «New Project» του µενού «File». 

Στη συνέχεια εµφανίζεται ένα παράθυρο διαλόγου µε τίτλο «New Project» (εικόνα 6.1) όπου πρέπει 

να επιλεγεί ο τύπος εφαρµογής  «Windows Application» και να καθορισθεί το όνοµα της εφαρµογής. 

Η δηµιουργία της εφαρµογής ολοκληρώνεται µε το πάτηµα του πλήκτρο «OK». 

 

 
 

Εικόνα 6.1 ∆ηµιουργία παραθυρικής εφαρµογής  

 

Η παραθυρική εφαρµογή (εικόνα 6.2) αρχικά  αποτελείται από ένα κενό παράθυρο χρήστη µε όνοµα 

και τίτλο Form1. 

 

Με την χρήση του παραθύρου ιδιοτήτων (κάτω δεξιό τµήµα του περιβάλλοντος ανάπτυξης) µπορεί να 

γίνει η αλλαγή των ιδιοτήτων του παραθύρου. Το παράθυρο ιδιοτήτων µπορεί να εµφανισθεί µε την 

επιλογή «Properties Window» από το µενού  «View» ή το πλήκτρο συντόµευσης F4.  

 

Βασικές ιδιότητες, µέθοδοι και συµβάντα του παραθύρου χρήστη είναι: 

 

Ιδιότητα Περιγραφή Παράδειγµα 

Νame Το όνοµα του παραθύρου FrmFirstApp 

Text Ο τίτλος του παραθύρου Η πρώτη εφαρµογή 

µου 

BackColor Το χρώµα υπόβαθρου  

Size Οι διαστάσεις (πλάτος και ύψος) του παράθυρου 474;303 

Μέθοδος Περιγραφή Παράδειγµα 

Close Κλείσιµο  του παραθύρου frmFirstApp.Close() 

Hide Απόκρυψη του παραθύρου frmFirstApp.Hide() 

Show Εµφάνιση του παραθύρου frmFirstApp.Show() 

Συµβάν Περιγραφή Παράδειγµα 

Load Φόρτωση του παραθύρου. Το συµβάν αυτό 

ενεργοποιείται κατά την φόρτωση του παράθυρου. 

FrmFirstApp_Load 

  

Οι διαστάσεις του παραθύρου µπορούν να αλλάξουν και µε την χρήση του ποντικιού. Η αλλαγή του 

µεγέθους του παραθύρου γίνεται µε την τοποθέτηση του δείκτη του ποντικιού στην κάτω αριστερή 

γωνία του παραθύρου οπότε και η εµφάνιση (εικόνα) του δείκτη  του ποντικιού αλλάζει σε διπλό 

βέλος. Με το σταθερό πάτηµα του αριστερού πλήκτρου του ποντικιού και την ταυτόχρονη 

µετακίνηση του ποντικιού, µπορεί να µεταβληθεί το µέγεθος του παραθύρου. 

 

 
 

Εικόνα 6.2 Αρχική παραθυρική εφαρµογή  

 

Οι ιδιότητες ενός παραθύρου, εκτός από τον ορισµός τους κατά την διάρκεια σχεδίασης της 

εφαρµογής, µπορούν να ορισθούν δυναµικά κατά την διάρκεια της εκτέλεσης της εφαρµογής µε την 

χρήση εντολών. Για την απόδοση τιµής σε κάποια ιδιότητα ενός παραθύρου χρησιµοποιείται η λέξη 

«Me» για να δηλώσει ότι η ιδιότητα αυτή ανήκει  στο παράθυρο (και όχι σε κάποιο άλλο στοιχείο 

ελέγχου ή αντικείµενο του παραθύρου) και το όνοµα της ιδιότητας. Με την παρακάτω εντολή 

ορίζεται δυναµικά ο τίτλος ενός παραθύρου δίνοντας την τιµή «Η πρώτη εφαρµογή µου» στην 

ιδιότητα Text του παραθύρου. 

 
        Me.Text = "Η πρώτη εφαρµογή µου" 

6.2 Σχεδιασµός του παραθύρου χρήστη 

 

Ο σχεδιασµός ενός παραθύρου γίνεται µε την προσθήκη στοιχείων ελέγχου στο παράθυρο, τον 

καθορισµό των ιδιοτήτων τους και την διασύνδεση των διαφόρων συµβάντων του παραθύρου και των 

στοιχείων ελέγχου του  µε µεθόδους οι οποίες εκτελούνται για το κάθε συµβάν.  

 

Τα βασικά στοιχεία ελέγχου ενός παραθύρου είναι: 

 

Στοιχείο Ελέγχου Περιγραφή 

Ετικέτα  

(Label) 

Στοιχείο ελέγχου το οποίο απεικονίζει κείµενο (χωρίς 

δυνατότητα αλλαγής από τον χρήστη) ή εικόνα 

Πλαίσιο κειµένου 

(TextBox) 

Στοιχείο ελέγχου το οποίο χρησιµοποιείται για την εισαγωγή 

κειµένου από τον χρήστη 

Πλήκτρο επιλογής 

(Radio Button) 

Πλήκτρο µε το οποίο ο χρήστης µπορεί να κάνει κάποια επιλογή  

Πλαίσιο επιλογής 

(CheckBox) 

Πλήκτρο µε το οποίο ο χρήστης µπορεί να κάνει κάποια επιλογή 

(χρησιµοποιείται συνήθως σε ερωτήσεις τύπου ΝΑΙ/ΟΧΙ)  

Κατάλογος Επιλογής  

(List Box) 

Στοιχείο ελέγχου το οποίο απεικονίζει έναν κατάλογο 

αντικείµενων. Ο χρήστης µπορεί να επιλέξει ένα η περισσότερα 

αντικείµενα. 

Αναδιπλούµενος Κατάλογος 

Επιλογής (ComboBox) 

Αναδιπλούµενος κατάλογος επιλογής.  

Πλαίσιο στοιχείων ελέγχου 

(Panel) 

Πλαίσιο στο οποίο  µπορούν να τοποθετηθούν και να 

οµαδοποιηθούν άλλα στοιχεία ελέγχου 

Πλήκτρο 

(Button) 

Πλήκτρο το οποίο χρησιµοποιείται για την κλήση κάποιου 

υποπρογράµµατος ή µεθόδου 

 

 

 
 

 

Εικόνα 6.3 Εργαλειοθήκη – Στοιχεία Ελέγχου 

 

Η επιλογή στοιχείων ελέγχου και η τοποθέτηση τους στο παράθυρο χρήστη γίνεται µε την χρήση της 

εργαλειοθήκης (Μενού «View» επιλογή «ToolBox» ή συντόµευση Ctrl-Alt-X). Τα συνηθέστερα 

στοιχεία ελέγχου µπορούν να επιλεχθούν από την κατηγορία Common Controls.  

Ετικέτα 

 

Η ετικέτα απεικονίζει κείµενο (χωρίς δυνατότητα αλλαγής από τον χρήστη) ή εικόνα. Η ετικέτα 

χρησιµοποιείται κυρίως σε συνδυασµό µε άλλα στοιχεία ελέγχου για να δηλώσει τον τίτλο τους ή να 

επεξηγήσει το περιεχόµενο τους ή την λειτουργία τους. 

Στο παρακάτω παράδειγµα χρησιµοποιούνται ετικέτες σε συνδυασµό µε πλαίσια κειµένου για την  

εισαγωγή του ονόµατος , του επιθέτου, της διεύθυνσης ηλεκτρονικού ταχυδροµείου και της 

διεύθυνσης  ενός σπουδαστή.  

 

 
 

Εικόνα 6.4 Ετικέτες και πλαίσια κειµένου  

 

Τα κυριότερα χαρακτηριστικά µίας ετικέτας είναι:  

 

Ιδιότητα Περιγραφή Παράδειγµα 

Νame Το όνοµα της ετικέτας LabelSurname 

Text Κείµενο το οποίο εµφανίζεται (Τίτλος) Επίθετο 

TextAlign Στοίχιση του κείµενου Left, Right ή Center 

Font Γραµµατοσειρά Microsoft Sans Serif 8.25pt 

BackColor Το χρώµα υπόβαθρου  

BorderStyle Τύπος πλαισίου Fixed3D, Fixed Single ή None 

Image Εικόνα η οποία εµφανίζεται   

Visible Ιδιότητα η οποία καθορίζει εάν το στοιχείο 

ελέγχου εµφανίζεται στο παράθυρο 

True ή False 

(ορατό ή κρυφό) 

Πλαίσιο κειµένου 

 

Το πλαίσιο κειµένου χρησιµοποιείται για την εισαγωγή κειµένου από τον χρήστη. Οι κυριότερες 

ιδιότητες ενός πλαισίου κειµένου είναι: 

 

Ιδιότητα Περιγραφή Παράδειγµα 

Νame Το όνοµα του πλαισίου κειµένου TextBoxSurname 

Text Κείµενο το οποίο εµφανίζεται ή έχει εισαχθεί 

από τον χρήστη 

 

TextAlign Στοίχιση του κείµενου Left, Right ή Center 

BackColor Το χρώµα υπόβαθρου  

BorderStyle Τύπος πλαισίου Fixed3D, Fixed Single ή None 

Font Γραµµατοσειρά Microsoft Sans Serif 8.25pt 

Visible Ιδιότητα η οποία καθορίζει εάν το στοιχείο 

ελέγχου εµφανίζεται στο παράθυρο 

True ή False 

(ορατό ή κρυφό) 

Multiline Καθορίζει εάν το κείµενο εκτείνεται σε 

περισσότερες από µία γραµµές 

True ή False 

ScrollBars Καθορίζει εάν το πλαίσιο κειµένου έχει 

οριζόντιες ή κάθετες µπάρες κύλισης 

None ή 

Horizontal ή 

Vertical ή 

Both  

Συµβάν Περιγραφή Παράδειγµα 

TextChanged Αλλαγή του κειµένου TextBoxSurname_TextChanged 

Με την χρήση της ιδιότητας Text µπορούµε να τροποποιήσουµε το κείµενο ενός πλαισίου κειµένου η 

να διαβάσουµε το κείµενο το οποίο εισήγαγε ο χρήστης. Για παράδειγµα, η εντολή  

 
        TextBoxSurName.Text = "" 
  

καθαρίζει το κείµενο του πλαίσιο κειµένου του επιθέτου του σπουδαστή. Το διάβασµα του επιθέτου 

το οποίο πληκτρολογήθηκε από τον χρήστη και η αποθήκευση του σε µεταβλητή µε το όνοµα 

strSurname γίνεται µε τις  παρακάτω εντολές 

 
        Dim strSurname As String 
        strSurname = TextBoxSurname.Text 
 

Πλήκτρα και πλαίσια επιλογής 

 

Το πλήκτρα επιλογής (RadioButtons) χρησιµοποιούνται σε επιλογές µεταξύ εναλλακτικών 

απαντήσεων. Στο παρακάτω παράδειγµα χρησιµοποιούνται δύο πλήκτρα επιλογής για την επιλογή 

του φύλο του σπουδαστή.  

 

Τα πλαίσια επιλογής (CheckBoxes) χρησιµοποιούνται συνήθως σε ερωτήσεις τύπου NAI/ΟΧΙ. Στο 

παρακάτω παράδειγµα το πλήκτρο επιλογής χρησιµοποιείται για την εισαγωγή της πληροφορίας  εάν 

ο σπουδαστής επιθυµεί να ενηµερώνεται µε ηλεκτρονικό ταχυδροµείο ή όχι. 

 

 
 

Εικόνα 6.5 Πλήκτρα και πλαίσια επιλογής  

 

Κύριες ιδιότητες ενός πλήκτρου επιλογής είναι: 

 

Ιδιότητα Περιγραφή Παράδειγµα 

Νame Το όνοµα του πλήκτρου επιλογής RadioButtonMale 

Text Κείµενο το οποίο εµφανίζεται Α 

Font Γραµµατοσειρά Microsoft Sans Serif 8.25pt 

BackColor Το χρώµα υπόβαθρου  

RightToLeft Καθορίζει εάν το κείµενο 

εµφανίζεται δεξιά του πλήκτρο 

True ή False 

Checked Ιδιότητα η οποία καθορίζει εάν το 

πλήκτρο επιλογής είναι επιλεγµένο 

True ή False 

Συµβάν Περιγραφή Παράδειγµα 

CheckedChanged Αλλαγή επιλογής RadioButtonMale _CheckedChanged 

Η βασικότερη ιδιότητα το πλήκτρου επιλογής είναι η ιδιότητα Checked. Εάν κατά την σχεδίαση του 

παραθύρου δώσουµε την τιµή True στην ιδιότητα Checked, τότε η το πλήκτρο αυτό γίνεται η 

προεπιλεγµένη επιλογή. Κατά την διάρκεια της εκτέλεσης του προγράµµατος µε τον έλεγχο της τιµής 

της ιδιότητα αυτής προσδιορίζεται εάν o χρήστης επέλεξε το πλήκτρο αυτό.   

 

Στην περίπτωση όπου χρησιµοποιούνται περισσότερες από µία οµάδες επιλογών στο ίδιο παράθυρο 

(π.χ. επιλογή φύλου και επιλογή αγαπηµένου αθλήµατος) τότε τα πλήκτρα επιλογής πρέπει να 

τοποθετηθούν πάνω σε διαφορετικά πλαίσια στοιχείων ελέγχου. 

 

Κύριες ιδιότητες ενός πλαισίου επιλογής είναι: 

 

Ιδιότητα Περιγραφή Παράδειγµα 

Νame Το όνοµα του πλήκτρου επιλογής CheckBoxMale 

Text Κείµενο το οποίο εµφανίζεται Α 

Font Γραµµατοσειρά Microsoft Sans Serif 8.25pt 

BackColor Το χρώµα υπόβαθρου  

RightToLeft Καθορίζει εάν το κείµενο εµφανίζεται 

δεξιά του πλασίου 

True ή False 

Checked Ιδιότητα η οποία καθορίζει εάν το πλαίσιο 

επιλογής είναι επιλεγµένο 

True ή False 

(Ναι/Όχι) 

ThreeState Ιδιότητα η οποία καθορίζει εάν το πλαίσιο 

επιλογής έχει τρεις καταστάσεις: 

επιλεγµένο, µη επιλεγµένο και µη 

καθορισµένο 

True ή False 

CheckedState Η ιδιότητα αυτή καθορίζει την κατάσταση 

του πλαισίου επιλογής στην περίπτωση που 

έχουν επιλεχθεί τρεις καταστάσεις για το 

πλαίσιο επιλογής (ThreeState=True)  

Checked ή 

Unchecked ή 

Indeterminate 

(Ναι/Όχι/Άγνωστο) 

Συµβάν Περιγραφή Παράδειγµα 

CheckedChanged Αλλαγή επιλογής CheckBoxMale_CheckedChanged 

 

Το πλαίσιο επιλογής έχει συνήθως δύο καταστάσεις: επιλεγµένο (Checked=true) και µη επιλεγµένο 

(Checked = false). Με την χρήση της ιδιότητας Checked κατά την διάρκεια της σχεδίασης 

καθορίζεται η προεπιλεγµένη κατάσταση και κατά την διάρκεια της εκτέλεσης ελέγχεται η επιλογή 

του χρήστη. 

 

Με την χρήση της ιδιότητας ThreeState είναι δυνατό να ορισθεί µια επιπλέον ενδιάµεση κατάσταση 

όπου ο χρήστης δεν έχει κάνει καµία επιλογή. Για την χρήση και της τρίτης κατάστασης η τιµή της 

ιδιότητας ThreeState πρέπει να είναι true. Ο έλεγχος της προεπιλεγµένης ή τρέχουσας κατάστασης 

γίνεται µε την ιδιότητα CheckedState η οποία έχει ως δυνατές τιµές checked, unchecked και 

indeterminate. 

Κατάλογος επιλογής και αναδιπλούµενος κατάλογος 

 

Με τον κατάλογο επιλογής (ListBox) και το αναδιπλούµενο κατάλογο επιλογής (ComboBox) δίνεται 

η δυνατότητα στον χρήστη της επιλογής ενός ή περισσοτέρων αντικειµένων από έναν κατάλογο. 

 

Στο παρακάτω παράδειγµα χρησιµοποιείται ένας κατάλογος επιλογής για τον καθορισµό του 

τµήµατος του σπουδαστή και ένας αναδιπλούµενος κατάλογος επιλογής για τον καθορισµό του 

εξαµήνου σπουδών του. 

Εικόνα 6.6 Κατάλογος επιλογής και αναδιπλούµενος κατάλογος  

 

 

Οι βασικότερες ιδιότητες του καταλόγου επιλογής είναι: 

 

Ιδιότητα Περιγραφή Παράδειγµα 

Νame Το όνοµα του καταλόγου επιλογής ListBoxDepartment 

Font Γραµµατοσειρά Microsoft Sans Serif 8.25pt 

BackColor Το χρώµα υπόβαθρου  

BorderStyle Τύπος πλαισίου Fixed3D, Fixed Single ή None 

RightToLeft Καθορίζει την στοίχιση των 

αντικειµένων 

True ή False 

SelectionMode Ιδιότητα η οποία καθορίζει τον 

αριθµό των επιλέξιµων 

αντικειµένων 

None ή 

One ή 

MultiSimple ή 

MultiExtended 

Items Συλλογή (Collection) η οποία 

περιέχει τα αντικείµενα του 

καταλόγου 

Ηλεκτρονικής 

Ηλεκτρολογίας 

Πληροφορικής 

Sorted Η ιδιότητα αυτή καθορίζει εάν τα 

αντικείµενα ταξινοµούνται 

αλφαβητικά  

True ή False 

SelectedIndex Ιδιότητα ή οποία καθορίζει ή 

επιστρέφει τον δείκτη του 

επιλεγµένου αντικείµενο. Η 

ιδιότητα αυτή δεν είναι διαθέσιµη 

κατά την σχεδίαση της εφαρµογής. 

ListBoxDepartment.SelectedIndex 

SelectedItem Επιστρέφει το επιλεγµένο 

αντικείµενο. Η ιδιότητα αυτή δεν 

είναι διαθέσιµη κατά την σχεδίαση 

της εφαρµογής. 

ListBoxDepartment.SelectedItem 

SelectedItems Επιστρέφει µια συλλογή 

(Collection) µε τα επιλεγµένα 

αντικείµενα του καταλόγου  

ListBoxDepartment.SelectedItems 

SelectedIndexes Επιστρέφει µια συλλογή 

(Collection) µε τους επιλεγµένους 

δείκτες των αντικειµένων του 

ListBoxDepartment.SelectedIndexes 

καταλόγου  

Μέθοδοι Περιγραφή Παράδειγµα 

GetSelected ∆έχεται ως παράµετρο εισόδου τον 

δείκτη του αντικειµένου και 

επιστρέφει True ή False ανάλογα 

εάν το αντικείµενο είναι 

επιλεγµένο ή όχι  

GetSelected(0) 

 

SetSelected ∆έχεται ως πρώτη παράµετρο τον 

δείκτη του αντικειµένου και ως 

δεύτερη Τrue ή False ή οποία 

καθορίζει εάν το αντικέιµενο είναι 

επιλεγµένο ή µη όχι  

SetSelected(0,False) 

(Επιλέγει το πρώτο αντικείµενο του 

καταλόγου) 

Συµβάν Περιγραφή Παράδειγµα 

SelectedIndexChanged Αλλαγή επιλογής ListBoxDepartment 

_SelectedIndexChanged 

 

Στην περίπτωση που η ιδιότητα SelectionMode έχει την τιµή None, ο χρήστης δεν έχει την 

δυνατότητα επιλογής αντικειµένων. Όταν έχει την τιµή One, µπορεί να επιλέξει µόνο ένα αντικείµενο, 

όταν έχει την τιµή MultiSimple µπορεί να επιλέξει περισσότερα από ένα αντικείµενα και όταν έχει 

την τιµή MultiExtended µπορεί µε το πάτηµα του πλήκτρου Shift και την ταυτόχρονη χρήση των 

πλήκτρων  µετακίνησης του πληκτρολογίου(πάνω ή κάτω βέλος) ή την χρήση του ποντικιού να 

επιλέξει περιοχές αντικειµένων. 

 

Οι ιδιότητες SelectedIndex και SelectedItem καθορίζουν ή επιστρέφουν τον δείκτη του επιλεγµένου 

αντικειµένου και το επιλεγµένο αντικείµενο του καταλόγου αντίστοιχα. Οι ιδιότητες αυτές δεν είναι 

διαθέσιµες κατά την σχεδίαση της εφαρµογής και βρίσκουν χρήση σε επιλογές ενός µόνο 

αντικειµένου (SelectionMode=One). 

 

Για παράδειγµα η παρακάτω εντολή προεπιλέγει το πρώτο αντικείµενο του καταλόγου 

 
     Me.ListBoxDepartment.SelectedIndex = 0 
 

και η παρακάτω εντολή αποθηκεύει το επιλεγµένο αντικείµενο στην µεταβλητή strDepartment. 

 
     Dim strDepartment As String = Me.ListBoxDepartment.SelectedItem 
 

Εάν δεν έχεο επιλεγεί κανένα αντικείµενο του καταλόγου επιλογής, τότε η τιµή της ιδιότητας 

SelectedIndex είναι ίση µε –1.  Η παρακάτω εντολή ελέγχει εάν έχει επιλεγεί κάποιο αντικείµενο από 

τον κατάλογο επιλογής: 

 
     If Me.ListBoxDepartment.SelectedIndex = -1 then .. 
 

Για την επιλογή πολλαπλών αντικειµένων χρησιµοποιούνται οι ιδιότητες SelectedItems και 

SelectedIndexes οι οποίες επιστρέφουν µια συλλογή (Collection) αντικειµένων και δεικτών 

αντικειµένων αντίστοιχα. Μία συλλογή είναι ένας πίνακα αντικειµένων ο οποίος έχει την δυνατότητα 

δυναµικής προσθήκης και διαγραφής αντικειµένων.  

 

Οι κυριότερες µέθοδοι και ιδιότητες συλλογών αντικειµένων είναι : 

• Add µε την οποία µπορούµε να προσθέσουµε ένα νέο αντικείµενο στην συλλογή 

• Clear µε την οποία µπορούµε να διαγράψουµε όλα τα στοιχεία της συλλογής 

• Remove µε την οποία µπορούµε να αφαιρέσουµε από την συλλογή ένα συγκεκριµένο 

αντικείµενο το οποίο δίνεται ως παράµετρος της µεθόδου.  
         πχ    Me.ListBoxDepartment.Items.Remove("Ηλεκτρολογίας") 

• Count Ιδιότητα η οποία επιστρέφει τον αριθµό των στοιχείων της συλλογής 

 

Η πρόσβαση στα στοιχεία µιας συλλογής µπορεί να γίνει µε την χρήση δεικτών όπως γίνεται και µε 

τους πίνακες. 

 

Τα στοιχεία ενός καταλόγου επιλογής µπορούν να καθορισθούν κατά την σχεδίαση της εφαρµογής ή 

να καθορισθούν δυναµικά κατά την διάρκεια εκτέλεσης της εφαρµογής µε την  χρήση της ιδιότητας 

Items. Το παρακάτω παράδειγµα απαλείφει όλα τα στοιχεία ενός καταλόγου, προσθέτει δυναµικά τρία 

αντικείµενα στον κατάλογο και προεπιλέγει το  πρώτο αντικείµενο. 

 
        'Καθαρισµός καταλόγου 
        Me.ListBoxDepartment.Items.Clear() 
        '∆υναµική προσθήκη στοιχείων στον κατάλογο 
        Me.ListBoxDepartment.Items.Add("Ηλεκτρονικής") 
        Me.ListBoxDepartment.Items.Add("Ηλεκτρολογίας") 
        Me.ListBoxDepartment.Items.Add("Πληροφορικής") 
        'Προεπιλόγή τµήµατος Ηλεκτρονικής 
        Me.ListBoxDepartment.SetSelected(0, True) 
 

Παράδειγµα 6.1 ∆υναµικός καθορισµός των στοιχείων καταλόγου επιλογής  

 

Οι βασικότερες ιδιότητες του αναδιπλούµενου καταλόγου επιλογής είναι: 

 

Ιδιότητα Περιγραφή Παράδειγµα 

Νame Το όνοµα του αναδιπλούµενου 

καταλόγου επιλογής 

ComboBoxSemester 

Font Γραµµατοσειρά Microsoft Sans Serif 8.25pt 

BackColor Το χρώµα υπόβαθρου  

RightToLeft Καθορίζει την στοίχιση των 

αντικειµένων 

True ή False 

DropDownStyle Ιδιότητα η οποία καθορίζει τον 

τρόπο µε τον οποίο ο χρήστης 

επιλέγει αντικείµενα 

Simple ή 

DropDown ή 

DropDownList 

Items Τα αντικείµενα του καταλόγου Α Β Γ ∆ Ε ΣΤ Η και Ζ 

(γραµµένα το καθένα σε διαφορετική 

γραµµή) 

Sorted Η ιδιότητα αυτή καθορίζει εάν 

τα αντικείµενα ταξινοµούνται 

αλφαβητικά  

True ή False 

SelectedIndex Ιδιότητα ή οποία καθορίζει ή 

επιστρέφει το δείκτη του 

επιλεγµένου αντικείµενο. Η 

ιδιότητα αυτή δεν είναι 

διαθέσιµη κατά την σχεδίαση 

της εφαρµογής. 

ComboBoxSemester.SelectedIndex() 

SelectedItem Επιστρέφει το επιλεγµένο 

αντικείµενο. Η ιδιότητα αυτή 

δεν είναι διαθέσιµη κατά την 

σχεδίαση της εφαρµογής. 

ComboBoxSemester.SelectedItem() 

Text Καθορίζει ή επιστρέφει το 

κείµενο το οποίο εµφανίζεται 

στον αναδιπλούµενο κατάλογο 

ComboBoxSemester.Text 

Συµβάν Περιγραφή Παράδειγµα 

SelectedIndexChanged Αλλαγή επιλογής ComboBoxSemester 

_SelectedIndexChanged 

 

Ο αναδιπλούµενος κατάλογος χρησιµοποιείται για την επιλογή ενός µόνο αντικειµένου από τον 

χρήστη. Εάν δεν έχει επιλεγεί  κάποιο αντικείµενο του καταλόγου τότε το εµφανιζόµενο κείµενο 

(ιδιότητα Text) είναι κενό. Εάν προεπιλεγεί κάποια αντικείµενο ή ο χρήστης επιλέξει κάποιο 

αντικείµενο από τον κατάλογο, τότε το αντικείµενο αυτό εµφανίζεται ως το κείµενο του 

αναδιπλούµενου κατάλογου. 

 

Ανάλογα µε την τιµή της ιδιότητας DropDownStyle διαφέρει και ο τρόπος επιλογής αντικειµένων. 

Εάν έχει επιλεγεί ως DropDownStyle η τιµή SimpleCombo, ο χρήστης επιλέγει κάποιο αντικείµενο µε 

την εισαγωγή κειµένου ή µε την χρήση των πλήκτρων (πάνω και κάτω βέλος του πληκτρολογίου). 

Εάν έχει επιλεγεί η τιµή DropDown, ο χρήστης µπορεί να αναπτύξει και να αναδιπλώσει τον 

κατάλογο αντικειµένων και να επιλέξει κάποιο στοιχείο του καταλόγου µε την χρήση του ποντικιού . 

Τέλος, εάν έχει επιλεγεί η τιµή DropDownList, ο χρήστης δεν έχει την δυνατότητα ελεύθερης 

εισαγωγής κειµένου αλλά πρέπει αναγκαστικά να επιλέξει κάποιο από τα υπάρχοντα αντικείµενα του 

καταλόγου. 

 

Οι ιδιότητες SelectedIndex και SelectedItem είναι αντίστοιχες µε τις ιδιότητες του καταλόγου 

επιλογής. Για παράδειγµα η παρακάτω εντολή προεπιλέγει το πρώτο αντικείµενο ως τιµή του 

αναδιπλούµενου καταλόγου 

 
       Me.ComboBoxSemester.SelectedIndex = 0 
 

και η παρακάτω εντολή αποθηκεύει το επιλεγµένο αντικείµενο στην µεταβλητή strSemester. 

 
      Dim strSemester As String = Me.ComboBoxSemester.SelectedItem 
 

Εναλλακτικά µπορεί να χρησιµοποιηθεί η ιδιότητα Text. 

  
      Dim strSemester As String = Me.ComboBoxSemester.Text 
 

Για την αποεπιλογή του επιλεγµένου αντικειµένου µπορεί να χρησιµοποιηθεί  η εντολή:  

 
       Me.ComboBoxSemester.SelectedIndex = -1 
 

Πλαίσιο στοιχείων ελέγχου 

 

Το πλαίσιο στοιχείων ελέγχου (Panel) µπορεί να επιλεγεί από την κατηγορία Containers της 

εργαλειοθήκης. Το πλαίσιο στοιχείων ελέγχου µπορεί να χρησιµοποιηθεί για την τοποθέτηση και 

οµαδοποίηση άλλων στοιχείων ελέγχου εντός του πλαισίου. Η οµαδοποίηση αυτή διευκολύνει τον 

σχεδιασµό ενός παραθύρου, την οµαδική µετακίνηση των στοιχείων ελέγχου που περιέχει και την 

οµαδοποίηση πλήκτρων επιλογής όταν απαιτείται να διαχωριστούν σε  διαφορετικές οµάδες επιλογής. 

 

Στο παρακάτω παράδειγµα χρησιµοποιούνται  δύο πλαίσια στοιχείων ελέγχου, το πρώτο για να 

οµαδοποιήσει όλα τα στοιχεία ελέγχου τα οποία είναι σχετικά µε την εισαγωγή των στοιχείων ενός 

σπουδαστή και το δεύτερο για να οµαδοποιήσει τα πλήκτρα επιλογής για τον καθορισµό του φύλου 

του σπουδαστή. 

 

 
 

Εικόνα 6.7 Πλαίσια στοιχείων ελέγχου  

 

Τα κυριότερα χαρακτηριστικά ενός πλαισίου στοιχείων ελέγχου είναι:  

Ιδιότητα Περιγραφή Παράδειγµα 

Νame Το όνοµα του πλαισίου στοιχείων ελέγχου panelStudentData 

BackColor Το χρώµα υπόβαθρου  

BorderStyle Τύπος πλαισίου Fixed3D, Fixed Single ή 

None 

Visible Ιδιότητα η οποία καθορίζει εάν το στοιχείο ελέγχου 

εµφανίζεται στο παράθυρο 

True ή False 

Πλήκτρο 

 

Τα πλήκτρα (Buttons) δίνουν την δυνατότητα στον χρήστη επιλογής ενεργειών οι οποίες εκτελούν 

κάποιο υποπρόγραµµα, τροποποιούν ιδιότητες αντικειµένων ή καλούν µεθόδους αντικειµένων. 

 

Στο παράδειγµα 6.8 χρησιµοποιούνται τρία πλήκτρα τα οποία δίνουν την δυνατότητα στο χρήστη να 

καθαρίσει τα δεδοµένα του παραθύρου, να εµφανίσει τα στοιχεία τα οποία έχει εισάγει και να κλείσει 

το παράθυρο της εφαρµογής. 

 

 
 

Εικόνα 6.8 Πλήκτρα  

 

Τα κυριότερα χαρακτηριστικά ενός πλήκτρου είναι:  

 

Ιδιότητα Περιγραφή Παράδειγµα 

Νame Το όνοµα του πλήκτρου ButtonClear 

BackColor Το χρώµα υπόβαθρου  

Text Ο τίτλος του πλήκτρου Καθαρισµός 

Visible Ιδιότητα η οποία καθορίζει εάν το πλήκτρο 

εµφανίζεται στο παράθυρο 

True ή False 

(ορατό ή κρυφό) 

Enabled Ενεργοποιεί ή απενεργοποιεί (δεν είναι δυνατή η 

επιλογή του από τον χρήστη) ένα πλήκτρο 

True ή False 

Συµβάν Περιγραφή Παράδειγµα 

Click Πάτηµα του πλήκτρου από το χρήστη ButtonClear_Click 

 

Βασικό χαρακτηριστικό ενός πλήκτρου είναι το συµβάν Click το οποίο προκαλείται όταν πατηθεί το 

πλήκτρο από τον χρήστη. 

6.3 ∆ιασύνδεση συµβάντων ενός παραθύρου 

 

Η υλοποίηση της λειτουργικότητας ενός παραθύρου γίνεται µε την διασύνδεση συµβάντων του 

παραθύρου ή των στοιχείων ελέγχου του µε την κλήση µεθόδων του παραθύρου. 

 

Σε εφαρµογές του τερµατικού (Console Application) η εκτέλεση του προγράµµατος ξεκινά µε την κλήση 

της διαδικασίας main. Αντίθετα σε µία παραθυρική εφαρµογή, η εκτέλεση του προγράµµατος συνήθως 

ξεκινά µε την εµφάνιση ενός αρχικού παράθυρου της εφαρµογής. Τα παράθυρο εκκίνησης καθορίζεται 

µε την χρήση του παραθύρου ιδιοτήτων της εφαρµογής.  

 

Σε µία παραθυρική εφαρµογή, η επικοινωνία µε τον χρήστη γίνεται µε το πάτηµα πλήκτρων της 

εφαρµογής,  την επιλογή εντολών από τα µενού της εφαρµογής και την αλληλεπίδραση γενικότερα µε τα 

παράθυρα και τα στοιχεία ελέγχου της εφαρµογής. Οι ενέργειες του χρήστη προκαλούν συµβάντα τα 

οποία µε την σειρά τους προκαλούν την εκτέλεση µεθόδων οι οποίες είναι συνδεδεµένες 

προγραµµατιστικά µε τα συµβάντα αυτά. 

 

Το παράθυρο του παραδείγµατος 6.8 χρησιµοποιεί τα παρακάτω συµβάντα: 

• Load για το παράθυρο της εφαρµογής FormStudentData 

• Click για το πλήκτρο ButtonClear  (Καθαρισµός) 

• Click για το πλήκτρο ButtonShow (Εµφάνιση) 

• Click για το πλήκτρο ButtonΟΚ µε (ΟΚ) 

 

 

 
  

Εικόνα 6.9 Επιλογή συµβάντων στο παράθυρο ιδιοτήτων  

 

Η αντιστοίχιση συµβάντων µε µεθόδους γίνεται µε την χρήση του παραθύρου ιδιοτήτων. Με το 

πάτηµα του εικονιδίου συµβάντα (δεξιό τµήµα της εικόνας 6.9) εµφανίζονται τα διαθέσιµα συµβάντα 

ενός αντικειµένου της εφαρµογής. Η επιστροφή στις ιδιότητες ενός αντικειµένου γίνεται µε το 

εικονίδιο ιδιότητες. 

 

Εικονίδιο 

Συµβάντα 
Εικονίδιο 

Ιδιότητες 

Η διασύνδεση ενός συµβάντος µε µία µέθοδο γίνεται µε το διπλό πάτηµα του αριστερού πλήκτρου 

του ποντικιού πάνω σε κάποια ιδιότητα. Εναλλακτικά, για τις βασικότερα συµβάντα ενός στοιχείου 

ελέγχου ή παραθύρου, η διασύνδεση µπορεί να γίνει µε το διπλό πάτηµα του αριστερού πλήκτρου του 

ποντικιού πάνω στο στοιχείο ελέγχου ή το παράθυρο της εφαρµογής. Όταν επιλεγεί κάποια συµβάν 

για διασύνδεση, το περιβάλλον της VisualBasic εµφανίζει το αρχείο κώδικα της VisualBasic το οποίο 

αντιστοιχεί στο παράθυρο της εφαρµογής και εισάγει αυτόµατα τον ορισµό της µεθόδου η οποία 

διασυνδέεται  µε το συµβάν (εικόνα 6.10).  

 

Η µέθοδος αρχικά είναι κενή. Ο προγραµµατιστής πρέπει να καθορίσει τις εντολές οι οποίες 

εκτελούνται για κάθε συµβάν. Για το συγκεκριµένο παράδειγµα ορίζονται οι παρακάτω µέθοδοι για 

τα συµβάντα:  

  

 
 

Εικόνα 6.10 Επιλογή συµβάντων στο παράθυρο ιδιοτήτων  

 

Συµβάν FormStudentData Load 
 
    Private Sub FormStudentData_Load(ByVal sender As System.Object, _ 
                     ByVal e As System.EventArgs) Handles MyBase.Load 
        'Καθορισµός τίτλου παραθύρου 
        Me.Text = "Παράδειγµα Στοιχείων Ελέγχου" 
    End Sub 

Παράδειγµα 6.2 

 

Συµβάν ButtonClear Click 

 
    Private Sub ButtonClear_Click(ByVal sender As System.Object, _ 
                     ByVal e As System.EventArgs) Handles ButtonClear.Click 
        'Καθαρισµός πλαισίων κειµένου 
        Me.TextBoxName.Text = "" 
        Me.TextBoxSurname.Text = "" 
        Me.TextBoxEmail.Text = "" 
        Me.TextBoxAddress.Text = "" 
 
        'Καθαρισµός πλήκτρων επιλογής 
        Me.RadioButtonFemale.Checked = False 
        Me.RadioButtonMale.Checked = False 

        'Καθαρισµός πλασίου επιλογής 
        Me.CheckBoxMail.Checked = False 
 
        'Καθαρισµός καταλόγων επιλογής 
        Me.ListBoxDepartment.SelectedIndex = -1 
        Me.ComboBoxSemester.SelectedIndex = -1 
 
    End Sub 

Παράδειγµα 6.3 

 

Συµβάν ButtonOk Click 

 
    Private Sub ButtonOK_Click(ByVal sender As System.Object, _ 
                    ByVal e As System.EventArgs) Handles ButtonOK.Click 
        'Τερµατισµός εφαρµογής 
        End 
    End Sub 

Παράδειγµα 6.4 

 

6.4 Χρήση πολλαπλών παραθύρων σε µία εφαρµογή 

 

Μια παραθυρική εφαρµογή µπορεί να χρησιµοποιεί περισσότερα του ενός παράθυρα. Η πρόσθεση 

ενός νέου παράθυρου σε µία εφαρµογή γίνεται µε την επιλογή «Add Windows Form» από το µενού 

«Project». Στο παράθυρο διαλόγου το οποίο εµφανίζεται (εικόνα 6.11) επιλέγεται ως τύπος 

αντικειµένου «WindowsForm» και στο πλαίσιο κειµένου εισάγεται το όνοµα του παράθυρου. Η 

διαδικασία δηµιουργίας τελειώνει µε την πάτηµα του πλήκτρου «Add». 

 

 

 
 

Εικόνα 6.11 ∆ηµιουργία νέου παραθύρου  

 

Παράδειγµα πρόσθεσης ενός νέου παραθύρου σε µία εφαρµογή είναι το παράθυρο µε όνοµα 

FormShowData (εικόνα 6.12). Το παράθυρο αυτό χρησιµοποιείται για την εµφάνιση των στοιχείων 

ενός σπουδαστή και αποτελείται από ένα πλαίσιο κειµένου µε όνοµα TextBoxData και ένα πλήκτρο 

µε όνοµα ButtonOK. Το πλαίσιο κειµένου εµφανίζει τα στοιχεία του σπουδαστή τα οποία εισήγαγε ο 

χρήστης και το πλήκτρο ButtonOK χρησιµοποιείται για το κλείσιµο του παραθύρου.   

Εικόνα 6.12 Το παράθυρο εµφάνισης των στοιχείων του σπουδαστή  

 

Η ιδιότητα Multiline του πλαισίου κειµένου TextBoxData έχει την τιµή True και η ιδιότητα του Font 

έχει την τιµή  Courier New; 9,75pt.  

 

Το πλήκτρο ButtonOK ορίζει το συµβάν Click το οποίο διασυνδέεται µε την παρακάτω µέθοδο η 

οποία κλείνει το παράθυρο FormShowData. 

 
    Private Sub ButtonOK_Click(ByVal sender As System.Object, _  
                       ByVal e As System.EventArgs) Handles ButtonOK.Click 
        Me.Close() 
    End Sub  

Παράδειγµα 6.5 

 

Το παράθυρο FormShowData εµφανίζεται µε το πάτηµα του πλήκτρου ButtonShowData του 

παράθυρου FormStudentData. Το πλήκτο διασυνδέεται µε την παρακάτω µέθοδο η οποία αρχικά 

εµφανίζει το παράθυρο FormShowData, αποθηκεύει τα στοιχεία του σπουδαστή σε µια µεταβλητή µε 

όνοµα strData και τέλος εµφανίζει τα δεδοµένα αυτά στο πλαίσιο TextBoxData του παραθύρου 

FormShowData. 

 
    Private Sub ButtonShow_Click(ByVal sender As System.Object, _ 
                       ByVal e As System.EventArgs) Handles ButtonShow.Click 
        'Εµφάνιση παραθύρου στοιχείων σπουδαστή 
        FormShowData.Show() 
 
        'Στοιχεία πλαισίων κειµένου 
        Dim strName As String = Me.TextBoxName.Text 
        Dim strSurname As String = Me.TextBoxSurname.Text 
        Dim strEmail As String = Me.TextBoxEmail.Text 
 
        'Στοιχεία πλήκτρων επιλογής 
        Dim strSex As String 
        If Me.RadioButtonMale.Checked Then 
            strSex = "A" 
        ElseIf Me.RadioButtonFemale.Checked Then 
            strSex = "Θ" 
        Else 
            strSex = "" 
        End If 
 
        'Στοιχεία πλαίσιου επιλογής 
        Dim strSendMail As String 
        If Me.CheckBoxMail.Checked Then 
            strSendMail = "ΝΑΙ" 
        Else 
            strSendMail = "OXI" 
        End If 
 
        'Στοιχεία καταλόγων επιλογής 
        Dim strDepartment As String = Me.ListBoxDepartment.SelectedItem 
        Dim strSemester As String = Me.ComboBoxSemester.SelectedItem 
 
        'Μορφοποίηση στοιχείων 
        Dim strData As String = "" 

        strData &= "Όνοµα       " + strName + vbCrLf 
        strData &= "Επώνυµο     " + strSurname + vbCrLf 
        strData &= "Email       " + strEmail + vbCrLf 
        strData &= "Φύλο        " + strSex + vbCrLf 
        strData &= "Ενηµέρωση   " + strSendMail + vbCrLf 
        strData &= "Τµήµα       " + strDepartment + vbCrLf 
        strData &= "Εξάµηνο     " + strSemester + vbCrLf 
 
        'Εµφάνιση στοιχείων 
        FormShowData.TextBoxStudentData.Text = strData 
 
 
    End Sub 
 

Παράδειγµα 6.6 

 

Εµφάνιση πολλαπλών παραθύρων του ιδίου τύπου 

 

Με την προσθήκη ενός νέου τύπου παραθύρου σε µία εφαρµογή δηµιουργείται ταυτόχρονα και µία 

νέα κλάση αντικειµένων µε το ίδιο όνοµα η οποία χρησιµοποιείται για τον ορισµό των ιδιοτήτων, 

µεταβλητών, µεθόδων και συµβάντων του παραθύρου. Εάν για παράδειγµα δηµιουργηθεί ένα νέο 

παράθυρο µε όνοµα FormShowData, τότε δηµιουργείται µία νέα κλάση αντικειµένων µε το όνοµα 

FormShowData (αρχείο FormShowData.Vb της εικόνας 6.10). 

 

Η εµφάνιση ενός παραθύρου γίνεται µε την  µέθοδο Show. Στο παράδειγµα 6.6 το άνοιγµα του 

παραθύρου FormShowData γίνεται µε την εντολή: 

 
        FormShowData.Show() 
 

Η εντολή αυτή εµφανίζει το παράθυρο το οποίο αντιστοιχεί στην κλάση αντικειµένων 

FormShowData. 

 

Μία κλάση αντικειµένων µπορεί να έχει αντικείµενα τα οποίαν ανήκουν στην κλάση αυτή. Τα 

αντικείµενα αυτά έχουν τα ίδια χαρακτηριστικά, ιδιότητες, µεθόδους και συµβάντα µε την  κλάση 

αντικειµένων  που ανήκουν. Η δηµιουργία νέων αντικειµένων µίας κλάσης γίνεται µε την εντολή 

new. 

 

Στο παρακάτω παράδειγµα δηµιουργείται ένα  νέο αντικείµενο µε όνοµα newForm το οποίο ανήκει 

στην κλάση αντικειµένων FormShowData και εµφανίζεται στην οθόνη µε την εντολή Show. 

 
        Dim newForm As FormShowData  
        newForm = New FormShowData 
        newForm.Show() 

Παράδειγµα 6.7 

 

Με την χρήση νέων αντικειµένων τα οποία ανήκουν στην κλάση ενός παραθύρου µπορούµε να 

εµφανίσουµε πολλαπλά παράθυρα του ιδίου τύπου σε µία εφαρµογή. Τα παράθυρα αυτά είναι 

πανοµοιότυπα αλλά το καθένα έχει τα δικά του δεδοµένα και συµπεριφορά. 

6.5 Είσοδος δεδοµένων µε την µέθοδο  InputBox 

 

Η µέθοδος InputBox µπορεί να χρησιµοποιηθεί για την είσοδο δεδοµένων σε παραθυρικές 

εφαρµογές. Η µέθοδος αυτή εµφανίζει ένα παράθυρο το οποίο ζητά από τον χρήση να εισάγει 

δεδοµένα. Για παράδειγµα οι εντολές 

 
        Dim strName As String 
        strName = InputBox("∆ώστε ένα όνοµα") 
 

εµφανίζουν το παρακάτω παράθυρο το οποίο ζητά από τον χρήση να εισάγει ένα όνοµα. 

Εικόνα 6.13 Παράθυρο εισόδου δεδοµένων  

 

Η εισαγωγή τελειώνει µε το είτε µε το πάτηµα του πλήκτρου ΟΚ. Η µέθοδος InputBox µε το πάτηµα 

του πλήκτρου OK επιστρέφει το κείµενο το οποίο πληκτρολόγησε ο χρήστης  και το αποθηκεύει στο 

συγκεκριµένο παράδειγµα στην µεταβλητή strName. Εάν πατηθεί το πλήκτρο Cancel, τότε η 

εισαγωγή στοιχείων ακυρώνεται και επιστρέφεται κείµενο µηδενικού µήκος (strName = ""). 

 

Η πρώτη παράµετρος της µεθόδου InputBox ορίζει το µήνυµα το οποίο εµφανίζεται στον χρήστη 

κατά την εισαγωγή. Προαιρετικά µπορεί να δοθεί δεύτερη παράµετρος η οποία ορίζει τον τίτλο του 

παραθύρου και τρίτη παράµετρος η οποία εµφανίζει µία αρχική τιµή στο πλαίσιο κειµένου. 

 

Για παράδειγµα η εντολή  
 
strName = InputBox("∆ώστε ένα όνοµα", "Εισαγωγή δεδοµένων", "George") 
 

εµφανίζει το παρακάτω παράθυρο 

 

 
 

Εικόνα 6.14 Παράθυρο εισόδου δεδοµένων  

 

Η µέθοδος InputBox µε την χρήση κατάλληλων συναρτήσεων µετατροπής µπορεί να χρησιµοποιηθεί 

για την είσοδο αριθµητικών δεδοµένων. 

 

Πχ. 

 
        Dim x As Double 
        x = CDbl(InputBox("∆ώστε πραγµατικό αριθµό")) 
 

 

Η συνάρτησης µετατροπής δεν είναι απαραίτητη εάν έχει ορισθεί Option Strict Off 

 

6.6 Εµφάνιση µηνυµάτων στον χρήστη 

 

Η εµφάνιση µηνυµάτων στον χρήστη γίνεται µε την χρήση της µεθόδου Show της κλάσης 

αντικειµένων  MessageBox.  

 

Στην απλούστερη µορφή της η µέθοδος Show χρησιµοποιείται µε παράµετρο το µήνυµα το οποίο 

εµφανίζεται στον χρήστη.  

Π.χ. η εντολή  

 
        MessageBox.Show("Σφάλµα Εφαρµογής") 
 

εµφανίζει το παρακάτω παράθυρο µηνυµάτων στον χρήστη 

 

 
 

Εικόνα 6.15 Παράθυρο µηνύµατος  

 

Προαιρετικά µπορεί να καθορισθεί ο τίτλος του παραθύρου µηνύµατος µε την εντολή: 

 
        MessageBox.Show("Σφάλµα Εφαρµογής", "Λάθος") 
 

Μπορεί επίσης να καθορισθεί ο τύπος των πλήκτρων και το εικονίδιο του παραθύρου. 

 

Πχ. Η εντολή 
        MessageBox.Show("Σφάλµα Εφαρµογής", _ 
                         "Λάθος", _ 
                         MessageBoxButtons.OK, _ 
                         MessageBoxIcon.Exclamation) 
 

εµφανίζει το παρακάτω παράθυρο: 

 

 
 

Εικόνα 6.16 Παράθυρο µηνύµατος 

 

Το παράθυρο µηνυµάτων στον χρήστη, όταν χρησιµοποιείται µε περισσότερα του ενός πλήκτρα 

(AbortRetryIgnore, OkCancel, RetryCancel, YesNo και YesNoCancel), µπορεί να χρησιµοποιηθεί για 

την λήψη αποφάσεων αναλόγως της προτίµησης του χρήστη. 

 

Το παρακάτω παράδειγµα ζητά επιβεβαίωση για τον τερµατισµό της εφαρµογής. Η µεταβλητή result 

χρησιµοποιείται για την αποθήκευση της προτίµησης του χρήστη και σε περίπτωση που έχει πατηθεί 

το πλήκτρο Ok, τότε η εκτέλεση της εφαρµογής σταµατά. 

 
        Dim Result as Integer 
        Result = MessageBox.Show("Τερµατισµός Εφαρµογής",  "", _ 
                                  MessageBoxButtons.OKCancel) 
        If Result = Windows.Forms.DialogResult.OK Then 
            End 
        End If 

Παράδειγµα 6.8 

6.7 Παράδειγµα παραθυρικής εφαρµογής 

 

Η παρακάτω εφαρµογή µε όνοµα Calculator δίνει την δυνατότητα στον χρήστη να κάνει βασικές 

αριθµητικές πράξεις. H εφαρµογή αποτελείται από το παράθυρο FormCalculator το οποίο φαίνεται 

στο σχήµα 6.17.  

 

 

 

 
 

 

Εικόνα 6.17 Η εφαρµογή της αριθµοµηχανής  

 

Το παράθυρο έχει τίτλο «Αριθµοµηχανή» και περιέχει τα παρακάτω στοιχείο ελέγχου: 

 

Όνοµα στοιχείο ελέγχου Περιγραφή Χαρακτηριστικά 

LabelNumber1 Ετικέτα πρώτου αριθµού Text = "Αριθµός 1" 

LabelNumber2 Ετικέτα δεύτερου αριθµού Text = "Αριθµός 2" 

LabelOperation Ετικέττα πράξης Text = «Πράξη» 

LabelResult Ετικέτα αποτελέσµατος Text = "Αποτέλεσµα" 

TextBoxNumber1 Πλαίσιο κειµένου για την 

εισαγωγή του πρώτου 

πραγµατικού αριθµού 

Text = "" 
TextAlign =Right 

TextBoxNumber2 Πλαίσιο κειµένου για την 

εισαγωγή του δεύτερου 

πραγµατικού αριθµού 

Text = "" 
TextAlign =Right 

TextBoxResult Πλαίσιο κειµένου για την  

εµφάνιση του αποτελέσµατος 

Text = "" 
TextAlign =Right 

ButtonAdd Πλήκτρο πρόσθεσης Event Click 

ButtonSubtract Πλήκτρο αφαίρεσης Event Click 

ButtonMultiply Πλήκτρο πολλαπλασιασµού Event Click 

ButtonDivide Πλήκτρο διαίρεσης Event Click 

ButtonOk Πλήκτρος τέλους εφαρµογής Event Click 

 

Οι ιδιότητες των στοιχείων ελέγχου φαίνονται στην στήλη του παραπάνω πίνακα µε τίτλο 

«Χαρακτηριστικά». Επιπλέον έχουν καθορισθεί τα συµβάντα Click για όλα τα πλήκτρα της 

εφαρµογής.  

 

Η µέθοδος η οποία αντιστοιχεί στο πάτηµα του πλήκτρου της πρόσθεσης είναι: 

 
    Private Sub ButtonAdd_Click(ByVal sender As System.Object, _ 
              ByVal e As System.EventArgs) Handles ButtonAdd.Click 
 
        'Ορισµός πραγµατικών µεταβλητών  
        Dim num1 As Double = 0 
        Dim num2 As Double = 0 
 
        'Μετατροπή και αποθήκευση πρώτης τιµής 

        If Me.TextBoxNumber1.Text <> "" Then 
            num1 = CDbl(Me.TextBoxNumber1.Text) 
        End If 
 
        'Μετατροπή και αποθήκευση δεύτερης τιµής 
        If Me.TextBoxNumber1.Text <> "" Then 
            num2 = CDbl(Me.TextBoxNumber2.Text) 
        End If 
 
        'Υπολογισµός και εµφάνιση αποτελέσµατος 
        Me.TextBoxResult.Text = num1 + num2 
 
    End Sub 

Παράδειγµα 6.9 

 

Σε περίπτωση που το αποτέλεσµα πρέπει να εµφανισθεί µε συγκεκριµένη µορφή (πχ δεκαδικά 

στοιχεία), µπορεί να χρησιµοποιηθεί η µέθοδος Parse (Κεφάλαιο 2). Π.χ. η εντολή 

 
        Me.TextBoxResult.Text = (num1 + num2).ToString("N4")  
 

εκτυπώνει το αποτέλεσµα της πρόσθεσης  µε ακρίβεια 4 δεκαδικών ψηφίων. 

 

Αντίστοιχες µε την µέθοδο της πρόσθεσης είναι και οι µέθοδοι για την υλοποίηση των άλλων 

αριθµητικών πράξεων. 

 

Για τον τερµατισµό της εφαρµογής χρησιµοποιείται η παρακάτω µέθοδος:  

 
    Private Sub ButtonOk_Click(ByVal sender As System.Object, _ 
                ByVal e As System.EventArgs) Handles ButtonOk.Click 
         
         End 
    End Sub 

Παράδειγµα 6.10 

6.8 Ασκήσεις Κεφαλαίου 6 

 

Άσκηση 6.1 

 

∆ηµιουργήστε παραθυρική εφαρµογή lab_6_ask_1 η οποία εµφανίζει ένα παράθυρο µε τίτλο 

Άσκηση 6.1. Στο παράθυρο της εφαρµογής εµφανίζονται τρία πλήκτρα µε τίτλους (ιδιότητα 

text) «1», «2» και «3» αντίστοιχα. Με το πάτηµα του αντίστοιχου πλήκτρου εµφανίζεται τα 

µηνύµατα «1», «2» και «3» µε την χρήση της µεθόδου show της κλάσης αντικειµένων 

Message Box. 

 

Προσθέστε πλήκτρο για τον τερµατισµό της εφαρµογής (παράδειγµα 6.8) 

 

                 
 

Άσκηση 6.2 

 

∆ηµιουργήστε παραθυρική εφαρµογή lab_6_ask_2 η οποία µε το πάτηµα ενός πλήκτρου µε 

τίτλο «Υπολογισµός εµβαδού κύκλου» ζητά από τον χρήστη να εισάγει την ακτίνα ενός 

κύκλου (µε την µέθοδο InputBox) και εκτυπώνει µε την χρήση της µεθόδου show της κλάσης 

αντικειµένων MessageBox το εµβαδόν του κύκλου. 

 

Χρησιµοποιήστε την επιλογή Option Strict On και τις κατάλληλες συναρτήσεις µετατροπής 

τύπων µεταβλητών. 

 

Άσκηση 6.3 

 
∆ηµιουργήστε παραθυρική εφαρµογή lab_6_ask_3 η οποία εµφανίζει το παρακάτω παράθυρο. 

 

 
 

∆ώστε  το όνοµα FormLab6Ask4 στο παράθυρο της εφαρµογής, και εισάγετε τα παρακάτω 

στοιχεία ελέγχου. 

 

• Ετικέτα µε όνοµα labelRadius και ιδιότητα Text = «Ακτίνα Κύκλου» 

• Ετικέτα µε όνοµα labelPerimeter και ιδιότητα Text = «Περίµετρος Κύκλου» 

• Πλαίσιο κειµένου µε όνοµα textBoxRadius 

• Πλαίσιο κειµένου µε όνοµα textBoxPerimeter 

• Πλήκτρο µε όνοµα buttonCalculatePerimeter και ιδιότητα Text = «Υπολογισµός 

Περιµέτρου» 

Ο χρήστης εισάγει την ακτίνα του κύκλου στο πλαίσιο κειµένου textBoxRadius και µε το πάτηµα 

του πλήκτρου buttonCalculatePerimeter το πρόγραµµα υπολογίζει την περίµετρο του κύκλου και 

την εµφανίζει στο πλαίσιο κειµένου textBoxPerimeter. 

 

Άσκηση 6.4 

 

Τροποποιήστε την  εφαρµογή lab_6_ask_3 προσθέτοντας πλήκτρο µε το όνοµα ButtonAbout και 

τίτλο «Σχετικά» το οποίο εµφανίζει ένα νέο παράθυρο όνοµα FormAbout και τίτλο «Σχετικά». 

Το παράθυρο αυτό εµφανίζει ένα πλαίσιο κειµένου µε το όνοµα της εφαρµογής, την ηµεροµηνία 

που δηµιουργήθηκε και το όνοµα σας. 

 

Πως µπορείτε να εµφανίζεται ταυτόχρονα πολλαπλά παράθυρα FormAbout ; 

 

Άσκηση 6.5 

 

∆ηµιουργήστε παραθυρική εφαρµογή lab_6_ask_5 η οποία εµφανίζει ένα παράθυρο µε τίτλο 

«Άσκηση 6.5» το οποίο περιέχει ένα αναδιπλούµενο κατάλογο επιλογής (comboBox) µε όνοµα 

comboBoxSchool και ένα κατάλογο επιλογής (listBox) µε όνοµα listBoxDepartment. Ο 

αναδιπλούµενος κατάλογος επιλογής εµφανίζει τις σχολές «Σ.Τ.ΕΦ.» και «Σ.Ε.Υ.Π». Όταν ο 

χρήστης επιλέξει κάποια από τις δύο σχολές, τότε εµφανίζονται τα τµήµατα της επιλεγµένης 

σχολής στον κατάλογο επιλογής listBoxDepartment (π.χ. εάν επιλεγεί η σχολή Σ.Τ.Ε.Φ τότε 

εµφανίζονται τα τµήµατα Ηλεκτρονικής, Ηλεκτρολογίας και Πληροφορικής). 

 

Άσκηση 6.6 

 
∆ηµιουργήστε παραθυρική εφαρµογή lab_6_ask_6 η οποία εµφανίζει ένα παράθυρο µε τίτλο 

«Άσκηση 6.6» το οποίο περιέχει έναν κατάλογο επιλογής (listBox) µε όνοµα listBoxNames 

και τρία πλήκτρα µε τίτλους «Προσθήκη Ονόµατος», «∆ιαγραφή Ονόµατος» και «Πλήθος 

ονοµάτων ». Με το πάτηµα του πλήκτρου «Προσθήκη Ονόµατος» η εφαρµογή ζητά από τον 

χρήστη να εισάγει έναν όνοµα µε την µέθοδο InputBox και τον προσθέτει στον κατάλογο 

επιλογής.  Με το πάτηµα του πλήκτρου «∆ιαγραφή Ονόµατος» διαγράφεται το επιλεγµένο 

όνοµα. Με το πάτηµα του πλήκτρου «Πλήθος ονοµάτων» εµφανίζεται µήνυµα µε το πλήθος 

των στοιχείων του πίνακα επιλογής. 

 

Άσκηση 6.7 

 

∆ηµιουργήστε εφαρµογή µε όνοµα lab_6_ask_7 ή οποία απεικονίζει σε παράθυρο µε όνοµα 

FormListBox έναν κατάλογο επιλογής  µε όνοµα ListBoxNumbers. Προσθέστε πλήκτρα 

ButtonAdd, ButtonSum και ButtonOK µε τίτλους «Πρόσθεση στοιχείων», «Άθροισµα 

στοιχείων» και «Τέλος» αντίστοιχα. Το πλήκτρο ButtonAdd ζητά από τον χρήστη να εισάγει 

διαδοχικά 10 αριθµούς (µε την χρήση εντολής επανάληψης και της µεθόδου InputBox) και 

τους προσθέτει στον κατάλογο επιλογής, το πλήκτρο ButtonSum υπολογίζει το άθροισµα των 

στοιχείων του καταλόγου και εµφανίζει σχετικό µήνυµα στον χρήστη και το πλήκτρο 

ButtonOK σταµατά την εκτέλεση της εφαρµογής. 

 

Άσκηση 6.8 

 

∆ηµιουργήστε εφαρµογή µε όνοµα lab_6_ask_8 ή οποία λειτουργεί ως ηλεκτρονική κλειδαριά. 

Το παράθυρο της εφαρµογής έχει 10 πλήκτρα µε τους αριθµούς από 0 έως 10. Ο χρήστης 

πρέπει να πληκτρολογήσει τον κατάλληλο αριθµό (πχ 456741) ώστε να εµφανισθεί το µήνυµα 

ότι δόθηκε ο σωστός αριθµός. Σε κάθε λάθος πάτηµα πλήκτρου εµφανίζεται µήνυµα λάθους. 

 

Υπόδειξη: Χρησιµοποιήστε ένα κρυφό πλαίσιο κειµένου (ιδιότητα Visible = False) για την 

αποθήκευση  των αριθµών που εισάγει ο χρήστης. Ελέγξτε σε κάθε πάτηµα κάποιου πλήκτρου 

εάν το κείµενο του πλαισίου κειµένου ταιριάζει µε το αριθµό κλειδί ‘456741’  

Άσκηση 6.9 

 
∆ηµιουργήστε παραθυρική εφαρµογή lab_6_ask_9. ∆ώστε  το όνοµα FormStudentData στο 

αρχικό παράθυρο της εφαρµογής και εισάγετε τα στοιχεία ελέγχου τα οποία φαίνονται στο 

εικόνα 6.6. 

 

∆ώστε τα ονόµατα TextBoxName, TextBoxSurname, TextBoxEmail και TextBoxAddress στα 

πλαίσια κειµένου του παραθύρου που αντιστοιχούν στα στοιχεία του ονόµατος, επωνύµου, 

email και διεύθυνσης του σπουδαστή.  

 

Ορίστε την ιδιότητα Multiline =True για το πλαίσιο κειµένου  TextBoxAddress. 

 

∆ώστε τα ονόµατα RadioButtonMale και RadioButtonFemale στα πλήκτρα επιλογής του 

φύλου του σπουδαστή και δώστε στην  ιδιότητα Text τις τιµές «Α» και «Θ» αντίστοιχα. ∆ώστε 

στην ιδιότητα RightToLeft την τιµή No και για τα δύο πλήκτρα επιλογής. 

 

∆ώστε το όνοµα  CheckBoxMail και τον  τίτλο «Επιθυµεί να ενηµερώνεται µε email» στο 

πλαίσιο επιλογής.  

 

∆ώστε το όνοµα ListBoxDepartment στον κατάλογο επιλογής και εισάγετε στην ιδιότητα 

Items τις τιµές που φαίνονται στην εικόνα 6.6. 

 

∆ώστε το όνοµα ComboBoxSemester στον αναδιπλούµενο κατάλογο επιλογής και εισάγετε 

στην ιδιότητα Items τα εξάµηνα από Α έως Η. 

 

Άσκηση 6.10 

 

Τοποθετήστε στο παράθυρο FormStudentData  της εφαρµογής lab_6_ask_9 πλήκτρα µε 

ονόµατα ButtonClear, ButtonShow και ButtonOk (εικόνα 6.8) και ορίστε τον τίτλο των 

πλήκτρων σε «Καθαρισµός», «Εµφάνιση» και «ΟΚ» αντίστοιχα. 

 

Προσθέστε ένα νέο παράθυρο στην εφαρµογή lab_6_ask_9 µε όνοµα FormDataShow (Εικόνα 

6.12). ∆ώστε το όνοµα TextBoxData  στο πλαίσιο κειµένου και το όνοµα ButtonOK στο 

πλήκτρο του παραθύρου. ∆ώστε την ιδιότητα Multiline=True στο πλαίσιο κειµένου 

TextBoxData και ορίστε τον τίτλο του πλήκτρου σε «ΟΚ». 

 

Ορίστε τα παρακάτω συµβάντα και τις µεθόδους των (αντιγράψτε τα παραδείγµατα 6.2, 6.3, 

6.4, 6.5 και 6.6): 

• Load για το παράθυρο FormStudentData 

• Click για το πλήκτρο ΒuttonClear του παράθυρου FormStudentData 

• Click για το πλήκτρο ΒuttonShow του παράθυρου FormStudentData 

• Click για το πλήκτρο ΒuttonOk του παράθυρου FormStudentData 

• Click για το πλήκτρο ΒuttonOk του παράθυρου FormShowData 

 

Εκτελέστε το πρόγραµµα και ελέγξτε την λειτουργικότητα των παραθύρων που 

δηµιουργήσατε. 

 

Άσκηση 6.11 

 

Τροποποιήστε την εφαρµογή lab_6_ask_9: 

 

• Εισάγετε τις κατάλληλες εντολές στην µέθοδο FormStudentData_Load ώστε ο 

καθορισµός των αντικειµένων του κατάλογου επιλογής ListBoxDepartment και του 

αναδιπλούµενου κατάλογου ComboSemester να γίνεται δυναµικά κατά το άνοιγµα 

του παραθύρου FormStudentData. (Παράδειγµα 6.1). Ζητήστε επίσης µε την χρήση 

της µεθόδου InputBοx ένα όνοµα από τον χρήστη και ορίστε το όνοµα αυτό ως τον 

τίτλο του παραθύρου FormStudentData. Ο τίτλος του παραθύρου δεν πρέπει να 

αλλάζει εάν ακυρωθεί η εισαγωγή του ονόµατος. 

• Τροποποιήστε την µέθοδο ButtonShow_Click η οποία εµφανίζει τα στοιχεία ενός 

σπουδαστή ώστε να εµφανίζει µε κάθε πάτηµα του πλήκτρου ένα νέο παράθυρο µε τα 

στοιχεία του σπουδαστή (Παράδειγµα 6.7) 

 

Άσκηση 6.12 

 

∆ηµιουργήστε εφαρµογή µε όνοµα Calculator όπως περιγράφεται στο υποκεφάλαιο 6.7 και 

ελέγξτε την λειτουργικότητα της. 

 

Στην περίπτωση που η διαίρεση δεν είναι δυνατή (Ο δεύτερος αριθµός είναι µηδενικός) 

εµφανίστε µήνυµα το οποίο ενηµερώνει τον χρήστη ότι η διαίρεση δεν είναι δυνατή. 

 

Προσθέστε πλήκτρο ButtonClear µε τίτλο «Καθαρισµός» το οποίο καθαρίζει όλα τα πλαίσια 

κειµένου του παραθύρου (τους αριθµούς και το αποτέλεσµα της πράξης) . 

 

Τροποποιήστε την µέθοδο ButtonOk_Click ώστε να ζητά την επιβεβαίωση από τον χρήστη για 

το κλείσιµο της εφαρµογής.  

 

Τοποθετήστε ετικέτα µε  τίτλο «Ακρίβεια ∆εκαδικών» και πλαίσιο κειµένου µε όνοµα 

TextBoxPrecision και αρχική τιµή «4». Η τιµή του πλαισίου κειµένου καθορίζει τον αριθµό 

των δεκαδικών ψηφίων του αποτελέσµατος.  

(Υπόδειξη: Χρησιµοποιήστε για την µορφοποίηση του αποτελέσµατος: 

        Me.TextBoxResult.Text = (num1 + num2).ToString("N"&TextBoxPrecision.Text) 

 

Άσκηση 6.13 

 
∆ηµιουργήστε εφαρµογή µε όνοµα lab_6_ask_13 ή οποία εµφανίζει παράθυροo µε δύο 

πλαίσια κειµένου textBoxNumber και textBoxDay. Όταν ο χρήστης εισάγει ένα αριθµό από το 

1 έως το 7 στο πλαίσιο κειµένου textBoxNumber, τότε το πρόγραµµα εµφανίζει στο πλαίσιο 

κειµένου  textBoxDay την αντίστοιχη ηµέρα της εβδοµάδος (1 – ∆ευτέρα, 2 – Τρίτη, .. , 7 – 

Κυριακή) 

 

Άσκηση 6.14 

 

∆ηµιουργήστε εφαρµογή µε όνοµα lab_6_ask_14 ή οποία εµφανίζει παράθυρο µε πλαίσιο 

κειµένου textBoxPassword και πλήκτρο ButtonOK. O χρήστης πρέπει να εισάγει τον 

κατάλληλο κωδικό (π.χ. 4567 ) στο πλαίσιο κειµένου και να πατήσει το πλήκτρο ButtonOK για 

να γίνει ο έλεγχος του κωδικού από πρόγραµµα. Σε περίπτωση λάθους κωδικού εµφανίζεται 

µήνυµα λάθους στον χρήστη. Σε τρεις διαδοχικές προσπάθειες µε λάθος κωδικό το πρόγραµµα 

τερµατίζει την λειτουργία του.  (Χρησιµοποιήστε ένα κρυφό πλαίσιο κειµένου µε όνοµα 

textBoxCounter για να αποθηκεύετε τον αριθµό των λάθος προσπαθειών)  

 

Άσκηση 6.15 

 

∆ηµιουργήστε εφαρµογή µε όνοµα lab_6_ask_15 ή οποία ζητά από τον χρήστη να απαντήσει 

σε 4 ερωτήσεις πολλαπλής επιλογής και αξιολογεί την απάντηση του. Για την κάθε ερώτηση 

υπάρχουν 4 δυνατές επιλογές οι οποίες υλοποιούνται µε οµαδοποιηµένα πλήκτρα επιλογής σε 

πλαίσιο στοιχείων ελέγχου (RadioButtons τοποθετηµένα σε διαφορετικά Panels). Για την κάθε 

ερώτηση υπάρχει µία µόνο σωστή απάντηση η οποία βαθµολογείται µε µία µονάδα. Οι λάθος 

απαντήσεις δεν βαθµολογούνται. 

 

 Άσκηση 6.16 

 

∆ηµιουργήστε εφαρµογή µε όνοµα lab_6_ask_16 ή οποία ζητά από τον χρήστη να απαντήσει 

σε 6 ερωτήσεις ΝΑΙ/ΟΧΙ οι οποίες υλοποιούνται µε πλαίσια επιλογής και τον αξιολογεί. Η 

κάθε ερώτηση έχει τρεις καταστάσεις (Ναι, Οχι και Αναπάντητη οι οποίες αντιστοιχούν στις 

καταστάσεις checked, unchecked και indeterminate του πλαισίου επιλογής). Για κάθε σωστή 

απάντηση δίνεται ένας βαθµός και για κάθε λάθος απάντηση αφαιρείται ένας βαθµός. Οι 

αναπάντητες ερωτήσεις δεν βαθµολογούνται. 

7 Υποπρογράµµατα  και µέθοδοι 

7.1 Υποπρογράµµατα σε λειτουργικές µονάδες 

 

Μια εφαρµογή τύπου Console Application αποτελείται τουλάχιστον από µία λειτουργική µονάδα η 

οποία περιέχει ένα υποπρόγραµµα µε την ονοµασία main.  

 
Module Module1 
 
    Sub Main() 
        Console.WriteLine("Hello") 
    End Sub 
 
End Module 

Παράδειγµα 7.1 

 

Υπάρχουν δύο τύποι υποπρογραµµάτων: οι διαδικασίες και οι συναρτήσεις. Μία λειτουργική µονάδα 

µπορεί να περιέχει εκτός του υποπρογράµµατος Main και άλλες διαδικασίες ή συναρτήσεις.  

∆ιαδικασίες 

 

Οι δηλώσεις των διαδικασιών αρχίζουν µε την λέξη Sub (συντοµογραφία του Subprocedure) και 

τελειώνουν µε την δήλωση End Sub. Στο παρακάτω παράδειγµα ορίζεται µια διαδικασία µε το όνοµα  

PrintLine η οποία εκτυπώνει µια γραµµή κειµένου στην οθόνη τερµατικού.  

 
Module Module1 
 
    Sub Main() 
        Printline() 
        Console.WriteLine("Τµήµα Ηλεκτρονικής") 
        Printline() 
    End Sub 
 
    Sub Printline() 
        Console.WriteLine("******************") 
    End Sub 
 
End Module 

Παράδειγµα 7.2 

 

Η διαδικασία PrintLine καλείται δύο φορές από την διαδικασία Main. Σε κάθε κλήση της διαδικασίας 

PrintLine, η εκτέλεση της διαδικασίας Main διακόπτεται προσωρινά µέχρι να ολοκληρωθεί η 

εκτέλεση της διαδικασίας PrintLine. To αποτέλεσµα της εκτέλεσης του παραδείγµατος 7.2 είναι το 

παρακάτω: 

 
****************** 
Τµήµα Ηλεκτρονικής 
****************** 
 

Ένα υποπρόγραµµα έχει την δυνατότητα να ορίζει τις δικές του τοπικές µεταβλητές όπως φαίνεται 

στο παράδειγµα 7.3. Οι µεταβλητές αυτές είναι διαθέσιµες µόνο κατά την διάρκεια εκτέλεσης του 

υποπρογράµµατος. Με το τέλος του υποπρογράµµατος αποδεσµεύεται η µνήµη του υπολογιστή που 

χρησιµοποιήθηκε για τις µεταβλητές αυτές. Σε κάθε νέα κλήση του υποπρογράµµατος οι µεταβλητές 

αυτές δηµιουργούνται εκ νέου. 

    Sub Printline() 
        Dim i As Integer 
        For i = 1 To 18 
            Console.Write("*") 
        Next 
        Console.WriteLine() 
    End Sub 

 

Παράδειγµα 7.3 Η διαδικασία PrintLine µε τοπικές µεταβλητές και εντολή επανάληψης 

Παράµετροι υποπρογραµµάτων 

 

Η κλήση ενός υποπρογράµµατος µπορεί να γίνει και µε την χρήση παραµέτρων. Για παράδειγµα 

µπορεί να καθορισθεί το πλήθος και ο χαρακτήρας τον οποίο τυπώνει η διαδικασία PrintLine 

(Παράδειγµα 7.4).  

 
Module Module1 
 
    Sub Main() 
        Printline(18, ".") 
        Console.WriteLine("Τµήµα Ηλεκτρονικής") 
        Printline(18, ".") 
    End Sub 
 
    Sub Printline(ByVal n As Integer, ByVal c As Char) 
        Dim i As Integer 
        For i = 1 To n 
            Console.Write(c) 
        Next 
        Console.WriteLine() 
    End Sub 
 
End Module 

 

Παράδειγµα 7.4 Η παραµετροποιηµένη διαδικασία PrintLine 

 

Οι παράµετροι ενός υποπρογράµµατος δηλώνονται εντός παρενθέσεων µετά την δήλωση του 

ονόµατος του υποπρογράµµατος. Οι παράµετροι ενός υποπρογράµµατος µπορούν να θεωρηθούν ως 

ειδικού τύπου µεταβλητές στις οποίες δίνεται τιµή κατά την κλήση του υποπρογράµµατος. 

 

Στο παραπάνω παράδειγµα κατά την κλήση του υποπρογράµµατος µε την εντολή PrintLine(18, "." ) 

δίνεται η τιµή 18 στην παράµετρο n και η τιµή "." στην παράµετρο c. Η παράµετρος n καθορίζει το 

πλήθος των χαρακτήρων που θα εκτυπωθούν και η παράµετρος c τον χαρακτήρα ο οποίος θα 

εκτυπωθεί. 

 

Οι παράµετροι, όπως και οι τοπικές µεταβλητές ενός υποπρογράµµατος είναι προσβάσιµες µόνον από 

το υποπρόγραµµα και είναι διαθέσιµες µόνο κατή την εκτέλεση του. 

 

Όταν µια παράµετρος χρησιµοποιείται µόνο για την είσοδο δεδοµένων σε ένα υποπρόγραµµα (όπως 

στο παράδειγµα 7.4), τότε δηλώνεται µε την λέξη ByVal (ByValue). Στην περίπτωση αυτή, ακόµη κι 

αν το καλούµενο υποπρόγραµµα  τροποποιήσει την τιµή της παραµέτρου, η αλλαγή δεν θα επηρεάσει 

τιµές των µεταβλητών του υποπρογράµµατος που έκανε την κλήση. 

 

Εάν η παράµετρος ενός υποπρογράµµατος ορισθεί µε την λέξη ByRef τότε η παράµετρος αυτή µπορεί 

να χρησιµοποιείται τόσο για την  είσοδο όσο και για την έξοδο δεδοµένων. Στο παράδειγµα 7.5 η 

διαδικασία IncNumber αυξάνει την τιµή της µεταβλητής number1 η οποία δίνεται ως παράµετρος µε 

την εντολή IncNumber(number1). Κατά την κλήση της διαδικασίας IncNumber, αντί να δοθεί η τιµή 

της µεταβλητής number1 στην παράµετρο n όπως θα γινόταν µε την δήλωση ByVal,  η µεταβλητή 

number1 και η παράµετρος n αποκτούν την ίδια διεύθυνση µνήµης και γίνονται πρακτικά ίδιες 

µεταβλητές η οποίες αναφέρονται µε διαφορετικό όνοµα από το υποπρόγραµµα που κάνει την κλήση 

και το υποπρόγραµµα το οποίο καλείται. 

 
Module Module1 
 
    Sub Main() 
        Dim number1 As Integer = 0 
        IncNumber(number1) 
        Console.WriteLine(number1) 
    End Sub 
 
    Sub IncNumber(ByRef n As Integer) 
        n = n + 1 
    End Sub 
 
End Module 

 

Παράδειγµα 7.5 

  

Το αποτέλεσµα της εκτέλεσης του παραπάνω προγράµµατος είναι ο αριθµός 1 

   

 

Ένα υποπρόγραµµα µπορεί να έχει προαιρετικές παραµέτρους όπως φαίνεται στο παράδειγµα 7.6. Οι 

παράµετροι n c δηλώνονται ως προαιρετικές και δίνονται σε αυτές αρχικές τιµές 18 και "." 

αντίστοιχα.   

 
    Sub Printline(Optional ByVal n As Integer = 18, _ 
                  Optional ByVal c As Char = "*") 
        Dim i As Integer 
        For i = 1 To n 
            Console.Write(c) 
        Next 
        Console.WriteLine() 
    End Sub 

 

Παράδειγµα 7.6 

Συναρτήσεις 

 

Οι δηλώσεις των συναρτήσεων αρχίζουν µε την λέξη Function και τελειώνουν µε την λέξη End 

Function. Οι συναρτήσεις σε αντίθεση µε τις διαδικασίες µετά το τέλος της εκτέλεσης τους 

επιστρέφουν κάποια τιµή. Ο τύπος δεδοµένων ή του αντικειµένου το οποίο επιστρέφει µία συνάρτηση  

ορίζεται µετά την δήλωση του ονόµατος της συνάρτησης. Η τιµή µιας συνάρτησης επιστρέφεται µε 

την εντολή Return. 

 

Στο παράδειγµα 7.7 ορίζεται η συνάρτηση DegreesToRad η οποία µετατρέπει την τιµή µιας γωνίας 

από µοίρες σε ακτίνια. Η διαδικασία Main, η οποία υπολογίζει το ηµίτονου µιας γωνίας η οποία 

δίνεται από τον χρήστη, καλεί την συνάρτησης DegreesToRad για να µετατρέψει την γωνία σε 

ακτίνια. 

 
Module Module1 
 
    Sub main() 
        Dim x As Double 
        Console.Write("∆ώστε γωνία σε µοίρες : ") 
        x = Console.ReadLine() 
        Console.WriteLine("Το ηµίτονο της γωνίας είναι {0}", _ 
                           Math.Sin(DegreesToRad(x))) 
    End Sub 

    Function DegreesToRad(ByVal angle As Double) As Double 
        Return angle * Math.PI / 180 
    End Function 
 
End Module 

 

Παράδειγµα 7.7 Συνάρτηση µετατροπής γωνίας σε ακτίνια 

 

 

Ένα άλλο παράδειγµα συνάρτησης για τον υπολογισµό του παραγοντικού ενός αριθµού δίνεται από 

το παρακάτω πρόγραµµα: 

 
Module Module1 
 
    Sub main() 
        Dim num As Integer 
        Console.Write("∆ώστε ακέραιο : ") 
        num = Console.ReadLine() 
        Console.WriteLine("Το παραγοντικό του είναι {0}", _ 
                            Paragontiko(num)) 
    End Sub 
 
    Function Paragontiko(ByVal n As Integer) 
        Dim par As Integer = 1 
        Dim i As Integer 
        For i = 2 To n 
            par = par * i 
        Next 
        Return par 
    End Function 
 
End Module 
 

Παράδειγµα 7.8 Συνάρτηση υπολογισµού του παραγοντικού ενός αριθµού 

Αναδροµικές συναρτήσεις 

 

Στην περίπτωση που κάποια συνάρτηση είναι αναδροµική (όπως το παραγοντικό ενός αριθµού) τότε 

για την υλοποίηση της µπορεί να χρησιµοποιηθεί αναδροµή, δηλαδή η τιµή της συνάρτησης να 

υπολογισθεί µε την χρήση των τιµών προηγούµενων όρων της συνάρτησης.  

 

Στο παρακάτω παράδειγµα 7.9 χρησιµοποιείται αναδροµή (αντί της εντολής επανάληψης του 

παραδείγµατος 7.8) για να υπολογισθεί το παραγοντικό ενός αριθµού. Το παραγοντικό ενός αριθµού 

Ν, εάν ο αριθµός είναι µηδέν είναι ο αριθµός ένα, διαφορετικό είναι το γινόµενο του αριθµού µε το 

παραγοντικό του Ν - 1. 

 
    Function Paragontiko(ByVal n As Integer) 
        If n > 1 Then 
            Return n * Paragontiko(n - 1) 
        Else 
            Return 1 
        End If 
    End Function 
 

Παράδειγµα 7.9 Αναδροµική συνάρτηση 

 

Κατά τον υπολογισµό του παραγοντικού ενός αριθµού, για παράδειγµα του αριθµού 4 (εικόνα 7.1), η 

συνάρτηση θα καλέσει τον εαυτό της για τον υπολογισµό του παραγοντικού του 3, στην συνέχεια θα 

κληθεί νέα συνάρτηση για τον του παραγοντικού του αριθµού 2, κ.ο.κ. έως ότου η αναδροµή 

σταµατήσει στον υπολογισµό του παραγοντικού του αριθµού 0. Στην συνέχεια το αποτέλεσµα του 

υπολογισµού του παραγοντικού του αριθµού 0 θα χρησιµοποιηθεί για τον υπολογισµό του 

παραγοντικού του αριθµού 1, κατόπιν για τον υπολογισµό του παραγοντικού του αριθµού 2 κ.ο.κ. 

µέχρις ότου υπολογισθεί το παραγοντικό του αρχικού αριθµού και τελειώσει η εκτέλεση όλων των 

συναρτήσεων  που χρησιµοποιήθηκαν. 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

Εικόνα 7.1 ∆ιαδοχικές κλήσεις αναδροµικών συναρτήσεων 

 

Στατικές µεταβλητές υποπρογραµµάτων 

 

Οι µεταβλητές ενός προγράµµατος δηµιουργούνται κατά την κλήση του υποπρογράµµατος και 

αποδεσµεύονται µε το τέλος του υποπρογράµµατος (οπότε και δεν διατηρούν τις τιµές τους). Με την 

χρήση της λέξης Static για την δήλωση µιας µεταβλητής µπορεί να διατηρηθεί η τιµή της µεταβλητής 

µεταξύ διαδοχικών κλήσεων του υποπρογράµµατος. 

 

Στο παράδειγµα 7.7 χρησιµοποιείται η δήλωση µιας στατικής µεταβλητής µε το όνοµα num η οποία 

διατηρεί την τιµή της και την αυξάνει κάθε φορά που καλείται η συνάρτηση counter η οποία 

χρησιµοποιείται ως µετρητής. 

 
Module Module1 
 
    Public Sub main() 
        Console.WriteLine(counter()) 
        Console.WriteLine(counter()) 
        Console.WriteLine(counter()) 
        Console.WriteLine(counter()) 
        Console.WriteLine(counter()) 
        Console.ReadLine() 
    End Sub 
 
    Public Function counter() As Integer 
        Static num As Integer 
        num = num + 1 
        Return num 
    End Function 
 
End Module 
 

Παράδειγµα 7.10 Στατική µεταβλητή 

 

6 

Paragontiko(3) = 3 * paragontiko(2) 

2 

Paragontiko(2) = 2 * paragontiko(1) 

1 

Paragontiko(1) = 1 * paragontiko(0) 

1 

Paragontiko(0) = 1 

24 

Paragontiko(4) = 4 * paragontiko(3) 

Το αποτέλεσµα της εκτέλεσης του παραπάνω προγράµµατος είναι 

 
1 
2 
3 
4 
5 
 

7.2 Οργάνωση υποπρογραµµάτων και δεδοµένων σε λειτουργικές µονάδες 

 

Μια εφαρµογή µπορεί να διαθέτει αρκετές λειτουργικές µονάδες οι οποίες οργανώνουν τα 

υποπρογράµµατα σε λογικές ενότητες. 

 

Η προσθήκη µιας νέας λειτουργικής µονάδας σε µία εφαρµογή γίνεται µε την επιλογή «Add New 

Module» του µενού «Project» και τον ορισµό του ονόµατος της λειτουργικής µονάδας στο παράθυρο 

διαλόγου το οποίο εµφανίζεται στην συνέχεια. 

 

 
 

Εικόνα 7.2 Πρόσθεση λειτουργικής µονάδας 

 

Έστω για παράδειγµα η παρακάτω εφαρµογή η οποία αποτελείται από δύο λειτουργικές µονάδες µε 

ονοµασίες Exercises και Trigonemetric H λειτουργική µονάδα Exercices είναι η αρχική µονάδα 

εκκίνησης της εφαρµογής και η λειτουργική µονάδα Trigonemetric χρησιµοποιείται ως βοηθητική 

µονάδα για την οµαδοποίηση τριγωνοµετρικών συναρτήσεων και ορισµών. 

 

Η λειτουργική µονάδα Exercises (Παράδειγµα 7.8) αποτελείται από την διαδικασία Main και τις 

διαδικασίες Askisi1, Askisi2, Askisi3 και Askisi4 οι οποίες αντιστοιχούν στις τέσσερις ασκήσεις ενός 

εργαστηρίου. Η διαδικασία Main εµφανίζει ένα µενού επιλογών από όπου ο χρήστης µπορεί είτε να 

επιλέξει την άσκηση την οποία θέλει να εκτελέσει ή να να τελειώσει την εκτέλεση της εφαρµογής. 

 

Η άσκηση 1 ζητά από τον χρήστη να εισάγει µία γωνία σε µοίρες και υπολογίζει το ηµίτονο της 

γωνίας. Για τον υπολογισµό του ηµίτονου της γωνίας  χρησιµοποιείται η βοηθητική συνάρτηση 

«Ηµίτονο» η οποία ορίζεται στην λειτουργική µονάδα Trigonemetric (Παράδειγµα 7.9). 

Module exercises 
 
    Sub main() 
        Dim selection As String 
        Do 
            'Εκτύπωση µενού επιλογών χρήστη 
            Console.WriteLine() 
            Console.WriteLine("Μενού Επιλογών") 
            Console.WriteLine() 
            Console.WriteLine("1. Άσκηση 1") 
            Console.WriteLine("2. Άσκηση 2") 
            Console.WriteLine("3. Άσκηση 3") 
            Console.WriteLine("4. Άσκηση 4") 
            Console.WriteLine("5. Τέλος") 
            Console.WriteLine() 
            Console.Write("∆ώστε επιλογή : ") 
            selection = Console.ReadLine() 
            Console.WriteLine() 
 
            'Εκτέλεση της ανάλογης άσκησης 
            Select Case selection 
                Case "1" : askisi1() 
                Case "2" : askisi2() 
                Case "3" : askisi3() 
                Case "4" : askisi4() 
            End Select 
 
        Loop Until selection = "5" 
    End Sub 
 
    Sub askisi1() 
        'Άσκηση 1 
        Dim x As Double 
        Console.Write("∆ώστε γωνία σε µοίρες : ") 
        x = Console.ReadLine() 
        Console.WriteLine("Το ηµίτονο της γωνίας είναι {0}", _ 
                           Trigonometric.Ηµίτονο(x)) 
    End Sub 
 
    Sub askisi2() 
        'Ορισµός υποπρογράµµατος άσκησης 2 
       
    End Sub 
    Sub askisi3() 
        'Ορισµός υποπρογράµµατος άσκησης 3 
 
    End Sub 
    Sub askisi4() 
        'Ορισµός υποπρογράµµατος άσκησης 4 
 
    End Sub 
 
End Module 
 

Παράδειγµα 7.11 Η λειτουργική µονάδα Exercises 

 

 

 

 

 

 



Κεφάλαιο 8 Σύνθετες δοµές δεδοµένων - Αντικείµενα Σελίδα  89 

 

Σηµειώσεις Προγραµµατισµού  Γρηγόρης Τζιάλλας Τµήµα Ηλεκτρονικής Τ.Ε.Ι. Λαµίας  

Module trigonometric 
 
    'Ορισµός σταθεράς π 
    Public Const π As Double = Math.PI 
 
    Public Function Ηµίτονο(ByVal angle As Double) As Double 
        'Επιστρέφει το ηµίτονο γωνίας η οποία δίνεται σε µοίρες 
        Return Math.Sin(angle * π / 180) 
    End Function 
 
End Module 
 
 

Παράδειγµα 7.12 Η λειτουργική µονάδα Trigonometric 

 

Το αποτέλεσµα της εκτέλεσης της εφαρµογής εάν ο χρήστης επιλέξει την  πρώτη άσκηση και 

πληκτρολογήσει γωνία 30 µοιρών είναι το παρακάτω: 

 
Μενού Επιλογών 
 
1. Άσκηση 1 
2. Άσκηση 2 
3. Άσκηση 3 
4. Άσκηση 4 
5. Τέλος 
 
∆ώστε επιλογή : 1 
 
∆ώστε γωνία σε µοίρες : 30 
Το ηµίτονο της γωνίας είναι 0,5 
 
Μενού Επιλογών 
 
1. Άσκηση 1 
2. Άσκηση 2 
3. Άσκηση 3 
4. Άσκηση 4 
5. Τέλος 
 
∆ώστε επιλογή : 
 
 

Μετά το τέλος της εκτέλεσης του υποπρογράµµατος της άσκησης 1 το µενού επιλογών 

επανεµφανίζεται ώστε ο χρήστης να επιλέξει κάποια άλλη άσκηση ή τον  τερµατισµό της εφαρµογής. 

 

Μία λειτουργική µονάδα µπορεί να ορίζει µεταβλητές ή σταθερές η οποίες είναι κοινές για όλα τα 

υποπρογράµµατα της. Οι µεταβλητές αυτές ή οι σταθερές δηµιουργούνται και λαµβάνουν τιµές µε 

την εκκίνηση της εφαρµογής και είναι διαθέσιµες σε όλη την  διάρκεια εκτέλεσης της εφαρµογής. Για 

παράδειγµα η λειτουργική µονάδα Trigonemetric ορίζει την σταθερά π. 

 

 Οι µεταβλητές, σταθερές και τα υποπρογράµµατα µιας λειτουργική µονάδας για να είναι διαθέσιµα 

στις υπόλοιπες λειτουργικές µονάδες της εφαρµογής θα πρέπει να δηλωθούν ως Public (∆ηµόσια). 

Για παράδειγµα η σταθερά π και η συνάρτηση ηµίτονο δηλώνονται ως Public. Όταν χρησιµοποιείται 

η δήλωση Public ή Private για το ορισµό µιας µεταβλητής η λέξη Dim παραλείπεται. 

 

Π.χ. 
Public myProjectName as String 
 

∆εδοµένα και υποπρογράµµατα τα οποία δεν χρειάζεται να είναι διαθέσιµα σε άλλες µονάδες ή είναι 

απαραίτητο να προστατευθούν από την υπόλοιπη εφαρµογή δηλώνονται ως Private (Ιδιωτικά).  

Εάν κατά τον ορισµό δεδοµένων ή υποπρογραµµάτων δεν χρησιµοποιηθεί η δήλωση Public η Private, 

τότε οι µεν σταθερές και µεταβλητές είναι ιδιωτικές και τα δε υποπρογράµµατα δηµόσια. 

 

Η αναφορά στα δεδοµένα ή σε κάποιο υποπρόγραµµα µιας λειτουργικής µονάδας από κάποια άλλη 

λειτουργική µονάδα (αν και µπορεί να παραλειφθεί όταν το όνοµα της µεταβλητής, σταθεράς ή του 

υποπρογράµµατος είναι µοναδικό) πρέπει να γίνεται µε την ταυτόχρονη δήλωση του ονόµατος της 

λειτουργικής µονάδας. 

 

Π.χ. 
Trigonometric.Ηµίτονο 
Trigonometric.π 
 

7.3 Μέθοδοι κλάσεων αντικειµένων 

 

Τα υποπρογράµµατα κλάσεων αντικειµένων ονοµάζονται µέθοδοι. Για παράδειγµα οι µέθοδοι µε τις 

οποίες διασυνδέονται τα συµβάντα µιας παραθυρικής εφαρµογής είναι υποπρογράµµατα. Μία κλάση 

αντικειµένων µπορεί επίσης να περιλαµβάνει δηλώσεις µεταβλητών και σταθερών. 

 

Στο παρακάτω παράδειγµα δίνεται ένας εναλλακτικός τρόπος υλοποίησης της αριθµοµηχανής του 

υποκεφαλαίου 6.7. Με την χρήση των µεθόδων ReadNumbers και ShowResult και των µεταβλητών 

num1 και num2 της κλάσης ελαχιστοποιείται και οργανώνεται καλύτερα ο κώδικας της εφαρµογής. 

 

 
Public Class FormCalculator 
    Dim num1, num2, result As Double 
    Private Sub ReadNumbers() 
        'Μετατροπή και αποθήκευση πρώτης τιµής 
        If Me.TextBoxNumber1.Text <> "" Then 
            num1 = CDbl(Me.TextBoxNumber1.Text) 
        Else 
            num1 = 0 
        End If 
 
        'Μετατροπή και αποθήκευση δεύτερης τιµής 
        If Me.TextBoxNumber1.Text <> "" Then 
            num2 = CDbl(Me.TextBoxNumber2.Text) 
        Else 
            num2 = 0 
        End If 
    End Sub 
 
    Private Sub ShowResult() 
        Me.TextBoxResult.Text = result.ToString 
    End Sub 
 
    Private Sub ButtonAdd_Click(ByVal sender As System.Object, _  
                         ByVal e As System.EventArgs) Handles ButtonAdd.Click 
        ReadNumbers() 
        result = num1 + num2 
        showResult() 
    End Sub 
 
    Private Sub ButtonSubtract_Click(ByVal sender As System.Object, _ 
                         ByVal e As System.EventArgs) Handles ButtonSubtract.Click 
        ReadNumbers() 
        result = num1 - num2 
        ShowResult() 
    End Sub 
 
    Private Sub ButtonMultiply_Click(ByVal sender As System.Object, _ 
                         ByVal e As System.EventArgs) Handles ButtonMultiply.Click 
        ReadNumbers() 
        result = num1 * num2 
        ShowResult() 
    End Sub 
 
    Private Sub ButtonDivide_Click(ByVal sender As System.Object, _  
                           ByVal e As System.EventArgs) Handles ButtonDivide.Click 

        ReadNumbers() 
        result = num1 * num2 
        ShowResult() 
    End Sub 
 
    Private Sub ButtonOk_Click(ByVal sender As System.Object, _  
                           ByVal e As System.EventArgs) Handles ButtonOk.Click 
        End 
    End Sub 
End Class 

 

Παράδειγµα 7.13 Εναλλακτική υλοποίηση της αριθµοµηχανής 

7.4 Ασκήσεις Κεφαλαίου 7 

 

Άσκηση 7.1 

 

∆ηµιουργήστε εφαρµογή lab_7_ask_1 µε τις λειτουργικές µονάδες Exercises και 

Trigonemetric του υποκεφαλαίου 7.2 (Παραδείγµατα 7.11 και 7.12).  

 

Άσκηση 7.2 

 

Τροποποιήστε την εφαρµογή lab_7_ask_1 ώστε µε την επιλογή της άσκησης 2 να 

υπολογίζεται και να εκτυπώνεται η περίµετρος και το εµβαδόν ενός κύκλου. Η ακτίνα του 

κύκλου δίνεται από τον χρήστη και ο υπολογισµός γίνεται µε την χρήση των συναρτήσεων 

ΕµβαδόνΚύκλου και ΠερίµετροςΚύκλου της οποίες θα ορίσετε στην λειτουργική µονάδα 

Trigonemetric.  

 

Άσκηση 7.3 

 

Τροποποιήστε την εφαρµογή lab_7_ask_1 ώστε µε την επιλογή της άσκησης 3 να 

υπολογίζονται και να εκτυπώνονται  µε εντολή επανάληψης οι αντιστοιχίες σε δραχµές των 10, 

20, 30, .. 100 Ευρώ. Η µετατροπή γίνεται µε χρήση συνάρτησης ΕυρώΣε∆ραχµές την οποία θα 

ορίσετε στην λειτουργική µονάδα Trigonemetric. 

 

Άσκηση 7.4 

 
Τροποποιήστε την εφαρµογή lab_7_ask_1 ώστε µε την επιλογή της άσκησης 4 να 

υπολογίζεται και να εκτυπώνεται η τιµή της συνάρτησης Fibonacci για έναν ακέραιο αριθµό n 

ο οποίος δίνεται από τον χρήστη. O αναδροµικός ορισµός της συνάρτησης είναι: 

  

 Fibonacci(n) = Fibonacci(n-1) + Fibonacci (n-2)    n>2 

 Fibonacci(1) = 1   n = 1 ή n =2 

  

Άσκηση 7.5 

 

∆ηµιουργήστε παραθυρική εφαρµογή lab_7_ask_2 µε παράθυρο εκκίνησης FormMain όπως 

φαίνεται στην παρακάτω εικόνα. ∆ώστε τις ονοµασίες ButtonAskisi1, ButtonAskisi2, 

ButtonAskisi3, ButtonAskisi4 και ButtonOK στα πλήκτρα του παραθύρου και προσθέστε στην 

εφαρµογή την λειτουργική µονάδα Trigonemetric του υποκεφαλαίου 7.2. Με το πάτηµα το 

πλήκτρου ButtonAskisi1 εµφανίζεται ένα πλαίσιο διαλόγου (InputBox) για την εισαγωγή µιας 

γωνίας σε µοίρες και εµφανίζεται ένα µήνυµα στον χρήστη µε το αποτέλεσµα του 

υπολογισµού του ηµίτονου της γωνίας.    

Άσκηση 7.6 

 

Τροποποιήστε την παραθυρική εφαρµογή 

lab_7_ask_2 ώστε µε το πάτηµα το πλήκτρου 

ButtonAskisi2 να εµφανίζεται  το παρακάτω 

παράθυρο για τον υπολογισµό της περιµέτρου 

και του εµβαδού της ακτίνας που εισάγει ο 

χρήστης σε πλαίσιο κειµένου µε όνοµα 

textBoxRadius. Η περίµετρος εµφανίζεται σε 

πλαίσιο κειµένου µε όνοµα textBoxPerimeter 

και το εµβαδόν σε πλαίσιο κειµένου µε όνοµα 

textBoxEmbadon. Ο υπολογισµός της 

περιµέτρου και του εµβαδού γίνεται µε το 

πάτηµα του πλήκτρου µε όνοµα 

ButtonCalculate. O υπολογισµός χρησιµοποιεί τις συναρτήσεις ΕµβαδόνΚύκλου και 

ΠερίµετροςΚύκλου της οποίες θα ορίσετε στην λειτουργική µονάδα Trigonemetric. Η 

περίµετρος και το εµβαδόν να εµφανίζονται µε ακρίβεια 4 ψηφίων. 

 

 

Άσκηση 7.7 

 
Τροποποιήστε την παραθυρική εφαρµογή 

lab_7_ask_2 ώστε µε το πάτηµα το πλήκτρου 

ButtonAskisi3 να εµφανίζεται παράθυρο µε δύο 

καταλόγους επιλογής (ListBoxes) µε ονόµατα 

ListBoxEuros και ListBoxDrachmas. Το 

περιεχόµενο του καταλόγου επιλογής 

ListBoxEuros είναι 10, 20, .. 100 (σε Ευρώ) και το 

περιεχόµενο του καταλόγου επιλογής 

ListBoxDrachmas είναι οι αντιστοιχίες των Ευρώ 

σε δραχµές. Τα περιεχόµενα των καταλόγων 

προστίθενται δυναµικά  µε εντολή επανάληψης µε 

το πάτηµα του πλήκτρου ButtonCalculate. Η 

µετατροπή γίνεται µε χρήση συνάρτησης 

ΕυρώΣε∆ραχµές την οποία θα ορίσετε στην 

λειτουργική µονάδα Trigonemetric. 

 

 

Άσκηση 7.8 

 

 

Τροποποιήστε την παραθυρική εφαρµογή lab_7_ask_2 

ώστε µε το πάτηµα το πλήκτρου ButtonAskisi4 να 

εµφανίζεται παράθυρο µε κατάλογο επιλογής (ListBox) µε 

όνοµα ListBoxProtos. Με το πάτηµα του πλήκτρου 

ButtonCalculate στο περιεχόµενο του καταλόγου επίλογής 

προστίθονται δυναµικά οι πρώτοι αριθµοί από τον αριθµό 3 

έως και τον αριθµό 35. Ο υπολογισµός των πρώτων 

αριθµών γίνεται µε την χρήση εντολής επανάληψης και τη 

χρήση συνάρτησης isProtos την οποία θα ορίσετε στην 

λειτουργική µονάδα Trigonemetric. Η συνάρτηση δέχεται 

ως παράµετρο έναν ακέραιο αριθµό και επιστρέφει την τιµή 

true εάν ο αριθµός είναι πρώτος και την  τιµή false εάν ο 

αριθµός δεν είναι πρώτος. 

 

 

 

 

 


